

# The Isabelle Programming Tutorial (draft)

by Christian Urban with contributions from:

Stefan Berghofer Jasmin Blanchette Sascha Böhme Jeremy Dawson Alexander Krauss Christian Sternagel

August 13, 2009

# **Contents**

Contents					
1	Intr	oduction	1		
	1.1	Intended Audience and Prior Knowledge	1		
	1.2	Existing Documentation	1		
	1.3	Typographic Conventions	2		
	1.4	Aaaaargh! My Code Does not Work Anymore	3		
	1.5	Some Naming Conventions in the Isabelle Sources	4		
	1.6	Acknowledgements	4		
2	First Steps 7				
	2.1	Including ML-Code	7		
	2.2	Debugging and Printing	8		
	2.3	Combinators	12		
	2.4	Antiquotations	16		
	2.5	Terms and Types	19		
	2.6	Constructing Terms and Types Manually	22		
	2.7	Type-Checking	28		
	2.8	Theorems	30		
	2.9	Setups (TBD)	31		
	2.10	Theorem Attributes	32		
	2.11	Theories, Contexts and Local Theories (TBD)	38		
	2.12	Storing Theorems (TBD)	38		
	2.13	Pretty-Printing	39		
	2.14	Misc (TBD)	44		
	2.15	Name Space Issues (TBD)	44		
3	Parsing 45				
	3.1	Building Generic Parsers	45		
	3.2	Parsing Theory Syntax	51		
	3.3	Context Parser (TBD)	55		

ii CONTENTS

Ribliography 14					
C	Con	Comments for Authors 1			
В	Solutions to Most Exercises				
	A.9	User Space Type-Systems (TBD)	. 140		
	A.8	SAT Solvers	. 138		
	A.7	Writing an Oracle (TBD)			
	A.6	Executing an External Application (TBD)			
	A.5	Storing Data (TBD)			
	A.4	Configuration Options			
		Measuring Time			
		Restricting the Runtime of a Function			
А		Useful Document Antiquotations	-		
٨	Recipes 129				
	5.6	Definitional Packages	. 128		
	5.5	Extensions of the Package (TBD)			
	5.4	The Gory Details			
	5.3	The Code in a Nutshell			
	5.2	Parsing and Typing the Specification			
	5.1	Preliminaries			
5	Hov	v to Write a Definitional Package	99		
	4.8	Structured Proofs (TBD)	. 98		
	4.7	Declarations (TBD)	. 98		
	4.6	Conversions	. 92		
	4.5	Simprocs	. 87		
	4.4	Simplifier Tactics	. 81		
	4.3	Tactic Combinators	. 76		
	4.2	Simple Tactics			
	4.1	Basics of Reasoning with Tactics	. 65		
4	Tact	ical Reasoning	65		
	3.8	Methods (TBD)	. 63		
	3.7	New Commands and Keyword Files	. 59		
	3.6	Parsing Specifications	. 56		
	3.5	Parsing Inner Syntax	. 55		
	3.4	Argument and Attribute Parsers (TBD)	. 55		

CONTENTS	ii
0011121110	

Index 150

iv CONTENTS

# Chapter 1

# Introduction

"My thesis is that programming is not at the bottom of the intellectual pyramid, but at the top. It's creative design of the highest order. It isn't monkey or donkey work; rather, as Edsger Dijkstra famously claimed, it's amongst the hardest intellectual tasks ever attempted."

Richard Bornat, In Defence of Programming [1]

If your next project requires you to program on the ML-level of Isabelle, then this tutorial is for you. It will guide you through the first steps of Isabelle programming, and also explain tricks of the trade. The best way to get to know the ML-level of Isabelle is by experimenting with the many code examples included in the tutorial. The code is as far as possible checked against the Isabelle distribution. If something does not work, then please let us know. It is impossible for us to know every environment, operating system or editor in which Isabelle is used. If you have comments, criticism or like to add to the tutorial, please feel free—you are most welcome! The tutorial is meant to be gentle and comprehensive. To achieve this we need your feedback.

## 1.1 Intended Audience and Prior Knowledge

This tutorial targets readers who already know how to use Isabelle for writing theories and proofs. We also assume that readers are familiar with the functional programming language ML, the language in which most of Isabelle is implemented. If you are unfamiliar with either of these two subjects, you should first work through the Isabelle/HOL tutorial [3] or Paulson's book on ML [5].

## 1.2 Existing Documentation

The following documentation about Isabelle programming already exists (and is part of the distribution of Isabelle):

<sup>&</sup>lt;sup>1</sup>Isabelle repository snapshot d00238af17b6 (31-Jul-2009)

The Isabelle/Isar Implementation Manual describes Isabelle from a high-level perspective, documenting both the underlying concepts and some of the interfaces.

The Isabelle Reference Manual is an older document that used to be the main reference of Isabelle at a time when all proof scripts were written on the ML-level. Many parts of this manual are outdated now, but some parts, particularly the chapters on tactics, are still useful.

The Isar Reference Manual provides specification material (like grammars, examples and so on) about Isar and its implementation.

Then of course there are:

The Isabelle sources. They are the ultimate reference for how things really work. Therefore you should not hesitate to look at the way things are actually implemented. More importantly, it is often good to look at code that does similar things as you want to do and learn from it. The UNIX command *grep* −*R* is often your best friend while programming with Isabelle, or hypersearch if you program using jEdit under MacOSX. To understand the sources, it is often also necessary to track the change history of a file or files. The Mercurial repository<sup>2</sup> for Isabelle provides convenient interfaces to query the history of files and "change sets".

### 1.3 Typographic Conventions

All ML-code in this tutorial is typeset in shaded boxes, like the following ML-expression:

```
ML {*
    3 + 4
*}
```

These boxes correspond to how code can be processed inside the interactive environment of Isabelle. It is therefore easy to experiment with what is displayed. However, for better readability we will drop the enclosing ML {\* ... \*} and just write:

```
3 + 4
```

Whenever appropriate we also show the response the code generates when evaluated. This response is prefixed with a ">", like:

```
3 + 4
> 7
```

<sup>&</sup>lt;sup>2</sup>http://isabelle.in.tum.de/repos/isabelle/

The user-level commands of Isabelle (i.e., the non-ML code) are written in **bold face** (e.g., **lemma**, **apply**, **foobar** and so on). We use \$ ... to indicate that a command needs to be run in a UNIX-shell, for example:

\$ grep -R ThyOutput \*

Pointers to further information and Isabelle files are typeset in *italic* and highlighted as follows:

**Read More**Further information or pointers to files.

The pointers to Isabelle files are hyperlinked to the tip of the Mercurial repository of Isabelle at http://isabelle.in.tum.de/repos/isabelle/.

A few exercises are scattered around the text. Their solutions are given in Appendix B. Of course, you learn most, if you first try to solve the exercises on your own, and then look at the solutions.

### 1.4 Aaaaargh! My Code Does not Work Anymore

One unpleasant aspect of any code development inside a larger system is that one has to aim at a "moving target". The Isabelle system is no exception. Every update lets potentially all hell break loose, because other developers have changed code you are relying on. Cursing is somewhat helpful in such situations, but taking the view that incompatible code changes are a fact of life might be more gratifying. Isabelle is a research project. In most circumstances it is just impossible to make research backward compatible (imagine Darwin attempting to make the Theory of Evolution backward compatible).

However, there are a few steps you can take to mitigate unwanted interferences with code changes from other developers. First, you can base your code on the latest stable release of Isabelle (it is aimed to have one such release at least once every year). This might cut you off from the latest feature implemented in Isabelle, but at least you do not have to track side-steps or dead-ends in the Isabelle development. Of course this means also you have to synchronise your code at the next stable release. Another possibility is to get your code into the Isabelle distribution. For this you have to convince other developers that your code or project is of general interest. If you managed to do this, then the problem of the moving target goes away, because when checking in new code, developers are strongly urged to test it against Isabelle's code base. If your project is part of that code base, then maintenance is done by others. Unfortunately, this might not be a helpful advice for all types of projects. A lower threshold for inclusion has the Archive of Formalised Proofs, short AFP.<sup>3</sup> This archive has been created mainly for formalisations that are interesting but not necessarily of general interest. If you have ML-code as part of a formalisation, then this might be the right place for you. There is no problem with updating your code after submission. At the moment developers are not as diligent with checking

<sup>&</sup>lt;sup>3</sup>http://afp.sourceforge.net/

their code against the AFP than with checking agains the distribution, but generally problems will be caught and the developer, who caused them, is expected to fix them. So also in this case code maintenance is done for you.

### 1.5 Some Naming Conventions in the Isabelle Sources

There are a few naming conventions in the Isabelle code that might aid reading and writing code. (Remember that code is written once, but read many times.) The most important conventions are:

- t, u, trm for (raw) terms; ML-type: term
- ct, cu for certified terms; ML-type: cterm
- ty, T, U for (raw) types; ML-type: typ
- th, thm for theorems; ML-type: thm
- foo\_tac for tactics; ML-type: tactic
- thy for theories; ML-type: theory
- ctxt for proof contexts; ML-type: Proof.context
- 1thy for local theories; ML-type: local\_theory
- context for generic contexts; ML-type Context.generic
- mx for mixfix syntax annotations; ML-type mixfix
- prt for pretty printing; ML-type Pretty. T

## 1.6 Acknowledgements

Financial support for this tutorial was provided by the German Research Council (DFG) under grant number URB 165/5-1. The following people contributed to the text:

- Stefan Berghofer wrote nearly all of the ML-code of the simple\_inductive-package and the code for the chunk-antiquotation. He also wrote the first version of this chapter describing the package and has been helpful beyond measure with answering questions about Isabelle.
- **Jasmin Blanchette** helped greatly with section 2.13.
- **Sascha Böhme** contributed the recipes in A.2, A.4, A.5, A.6 and A.7. He also wrote section 4.6 and helped with recipe A.3.
- Jeremy Dawson wrote the first version of the chapter about parsing.

5

- **Armin Heller** helped with recipe A.8.
- Alexander Krauss wrote the first version of the "first-steps" chapter and also contributed the material on Named\_Thms.
- **Christian Sternagel** proofread the tutorial and made many improvemets to the text.

Please let me know of any omissions. Responsibility for any remaining errors lies with me.

This document is still in the process of being written! All of the text is still under construction. Sections and chapters that are under heavy construction are marked with TBD.

# Chapter 2

# First Steps

Isabelle programming is done in ML. Just like lemmas and proofs, ML-code for Isabelle must be part of a theory. If you want to follow the code given in this chapter, we assume you are working inside the theory starting with

```
theory FirstSteps
imports Main
begin
```

We also generally assume you are working with HOL. The given examples might need to be adapted if you work in a different logic.

## 2.1 Including ML-Code

The easiest and quickest way to include code in a theory is by using the **ML**-command. For example:

```
ML {*
    3 + 4
*}
> 7
```

Like normal Isabelle scripts, **ML**-commands can be evaluated by using the advance and undo buttons of your Isabelle environment. The code inside the **ML**-command can also contain value and function bindings, for example

```
ML {*
    val r = ref 0
    fun f n = n + 1
*}
```

and even those can be undone when the proof script is retracted. As mentioned in the Introduction, we will drop the ML {\* ... \*} scaffolding whenever we show

code. The lines prefixed with ">" are not part of the code, rather they indicate what the response is when the code is evaluated. There are also the commands ML\_val and ML\_prf for including ML-code. The first evaluates the given code, but any effect on the theory, in which the code is embedded, is suppressed. The second needs to be used if ML-code is defined inside a proof. For example

```
lemma test:
shows "True"
ML_prf {* writeln "Trivial!" *}
oops
```

However, both commands will only play minor roles in this tutorial (we will always arrange that the ML-code is defined outside of proofs, for example).

Once a portion of code is relatively stable, you usually want to export it to a separate ML-file. Such files can then be included somewhere inside a theory by using the command **use**. For example

```
theory FirstSteps
imports Main
uses ("file_to_be_included.ML") ...
begin
...
use "file_to_be_included.ML"
...
```

The **uses**-command in the header of the theory is needed in order to indicate the dependency of the theory on the ML-file. Alternatively, the file can be included by just writing in the header

```
theory FirstSteps
imports Main
uses "file_to_be_included.ML" ...
begin
...
```

Note that no parentheses are given this time. Note also that the included ML-file should not contain any **use** itself. Otherwise Isabelle is unable to record all file dependencies, which is a nuisance if you have to track down errors.

## 2.2 Debugging and Printing

During development you might find it necessary to inspect some data in your code. This can be done in a "quick-and-dirty" fashion using the function writeln. For example

```
writeln "any string"
> "any string"
```

will print out "any string" inside the response buffer of Isabelle. This function expects a string as argument. If you develop under PolyML, then there is a convenient, though again "quick-and-dirty", method for converting values into strings, namely the function PolyML.makestring:

```
writeln (PolyML.makestring 1)
> "1"
```

However, makestring only works if the type of what is converted is monomorphic and not a function.

The function writeln should only be used for testing purposes, because any output this function generates will be overwritten as soon as an error is raised. For printing anything more serious and elaborate, the function tracing is more appropriate. This function writes all output into a separate tracing buffer. For example:

```
tracing "foo" > "foo"
```

It is also possible to redirect the "channel" where the string *foo* is printed to a separate file, e.g., to prevent ProofGeneral from choking on massive amounts of trace output. This redirection can be achieved with the code:

Calling now

```
redirect_tracing (TextIO.openOut "foo.bar")
```

will cause that all tracing information is printed into the file *foo.bar*. You can print out error messages with the function *error*; for example:<sup>1</sup>

<sup>&</sup>lt;sup>1</sup>FIXME: unwanted pagebreak

```
if 0=1 then true else (error "foo")
> Exception- ERROR "foo" raised
> At command "ML".
```

This function raises the exception *ERROR*, which will then be displayed by the infrastructure.

(FIXME Mention how to work with debug and profiling.)

Most often you want to inspect data of type term, cterm or thm. Isabelle contains elaborate pretty-printing functions for printing them (see Section 2.13), but for quick-and-dirty solutions they are far too unwieldy. One way to transform a term into a string is to use the function string\_of\_term.

```
 Syntax.string\_of\_term @\{context\}  @\{term "1::nat"\} \\ > "\^F\rconst\^Fname=HOL.one\_class.one\^E1\^F\^E\^F\^E" \\
```

This produces a string with some additional information encoded in it. The string can be properly printed by using either the function writeln or tracing.

```
writeln (Syntax.string_of_term @{context} @{term "1::nat"})
> "1"
```

A cterm can be transformed into a string by the following function.

```
fun string_of_cterm ctxt t =
   Syntax.string_of_term ctxt (term_of t)
```

In this example the function term\_of extracts the term from a cterm. If there are more than one cterms to be printed, you can use the function commas to separate them.

```
fun string_of_cterms ctxt ts =
  commas (map (string_of_cterm ctxt) ts)
```

The easiest way to get the string of a theorem is to transform it into a cterm using the function crep\_thm.

```
fun string_of_thm ctxt thm =
  string_of_cterm ctxt (#prop (crep_thm thm))
```

Theorems also include schematic variables, such as ?P, ?Q and so on.

```
tracing (string_of_thm @{context} @{thm conjI}) > [P; P] \implies P \land P
```

In order to improve the readability of theorems we convert these schematic variables into free variables using the function *import*.

```
fun no_vars ctxt thm =
let
  val ((_, [thm']), _) = Variable.import true [thm] ctxt
in
  thm'
end

fun string_of_thm_no_vars ctxt thm =
  string_of_cterm ctxt (#prop (crep_thm (no_vars ctxt thm)))
```

Theorem *conjI* is now printed as follows:

```
tracing (string_of_thm_no_vars @{context} @{thm conjI}) > \llbracket P; \ Q \rrbracket \implies P \ \land \ Q
```

Again the function *commas* helps with printing more than one theorem.

```
fun string_of_thms ctxt thms =
  commas (map (string_of_thm ctxt) thms)

fun string_of_thms_no_vars ctxt thms =
  commas (map (string_of_thm_no_vars ctxt) thms)
```

When printing out several parcels of information that semantically belong together, like a warning message consisting for example of a term and a type, you should try to keep this information together in a single string. So do not print out information as

```
tracing "First half,";
tracing "and second half."
> First half,
> and second half.
```

but as a single string with appropriate formatting. For example

```
tracing ("First half," ^ "\n" ^ "and second half.")
> First half,
> and second half.
```

To ease this kind of string manipulations, there are a number of library functions. For example, the function <code>cat\_lines</code> concatenates a list of strings and inserts newlines.

```
cat_lines ["foo", "bar"]
> "foo\nbar"
```

Section 2.13 will also explain infrastructure that helps with more elaborate pretty printing.

#### 2.3 Combinators

For beginners perhaps the most puzzling parts in the existing code of Isabelle are the combinators. At first they seem to greatly obstruct the comprehension of the code, but after getting familiar with them, they actually ease the understanding and also the programming.

The simplest combinator is *I*, which is just the identity function defined as

```
fun I x = x
```

Another simple combinator is K, defined as

```
fun K x = fn _ => x
```

K "wraps" a function around the argument x. However, this function ignores its argument. As a result, K defines a constant function always returning x.

The next combinator is reverse application, />, defined as:

```
fun x > f = f x
```

While just syntactic sugar for the usual function application, the purpose of this combinator is to implement functions in a "waterfall fashion". Consider for example the function

```
fun inc_by_five x =
    x |> (fn x => x + 1)
    |> (fn x => (x, x))
    |> fst
    |> (fn x => x + 4)
```

which increments its argument x by 5. It proceeds by first incrementing the argument by 1 (Line 2); then storing the result in a pair (Line 3); taking the first component of the pair (Line 4) and finally incrementing the first component by 4 (Line 5). This kind of cascading manipulations of values is quite common when dealing with theories (for example by adding a definition, followed by lemmas and so on). The reverse application allows you to read what happens in a top-down manner. This kind of coding should also be familiar, if you have been exposed to Haskell's do-notation. Writing the function  $inc_by_five$  using the reverse application is much clearer than writing

2.3. COMBINATORS 13

```
fun inc_by_five x = fst ((fn x \Rightarrow (x, x)) (x + 1)) + 4
```

or

```
fun inc_by_five x = ((fn \ x \Rightarrow x + 4) \ o \ fst \ o \ (fn \ x \Rightarrow (x, x)) \ o \ (fn \ x \Rightarrow x + 1)) \ x
```

and typographically more economical than

```
fun inc_by_five x =
let val y1 = x + 1
    val y2 = (y1, y1)
    val y3 = fst y2
    val y4 = y3 + 4
in y4 end
```

Another reason why the let-bindings in the code above are better to be avoided: it is more than easy to get the intermediate values wrong, not to mention the nightmares the maintenance of this code causes!

In Isabelle, a "real world" example for a function written in the waterfall fashion might be the following code:

This function takes a term and a context as argument. If the term is of function type, then  $apply\_fresh\_args$  returns the term with distinct variables applied to it. For example below variables are applied to the term  $P::nat \Rightarrow int \Rightarrow unit \Rightarrow bool$ :

```
let
  val t = @{term "P::nat \Rightarrow int \Rightarrow unit \Rightarrow bool"}
  val ctxt = @{context}
in
  apply_fresh_args t ctxt
  /> Syntax.string_of_term ctxt
  /> tracing
end
> P z za zb
```

You can read off this behaviour from how apply\_fresh\_args is coded: in Line 2, the function fastype\_of calculates the type of the term; binder\_types in the next line produces the list of argument types (in the case above the list [nat, int, unit]);

Line 4 pairs up each type with the string z; the function  $variant\_frees$  generates for each z a unique name avoiding the given f; the list of name-type pairs is turned into a list of variable terms in Line 6, which in the last line is applied by the function  $list\_comb$  to the term. In this last step we have to use the function curry, because  $list\_comb$  expects the function and the variables list as a pair. This kind of functions is often needed when constructing terms with fresh variables. The infrastructure helps tremendously to avoid any name clashes. Consider for example:

```
let
  val t = @{term "za::'a \Rightarrow 'b \Rightarrow 'c"}
  val ctxt = @{context}

in
  apply_fresh_args t ctxt
  /> Syntax.string_of_term ctxt
  /> tracing
end
> za z zb
```

where the za is correctly avoided.

The combinator #> is the reverse function composition. It can be used to define the following function

```
val inc_by_six =
          (fn x => x + 1)
#> (fn x => x + 2)
#> (fn x => x + 3)
```

which is the function composed of first the increment-by-one function and then increment-by-two, followed by increment-by-three. Again, the reverse function composition allows you to read the code top-down.

The remaining combinators described in this section add convenience for the "waterfall method" of writing functions. The combinator tap allows you to get hold of an intermediate result (to do some side-calculations for instance). The function

increments the argument first by 1 and then by 2. In the middle (Line 3), however, it uses tap for printing the "plus-one" intermediate result inside the tracing buffer. The function tap can only be used for side-calculations, because any value that is computed cannot be merged back into the "main waterfall". To do this, you can use the next combinator.

The combinator ' (a backtick) is similar to tap, but applies a function to the value and returns the result together with the value (as a pair). For example the function

2.3. COMBINATORS 15

```
fun inc_as_pair x =
x |> '(fn x => x + 1)
|> (fn (x, y) => (x, y + 1))
```

takes x as argument, and then increments x, but also keeps x. The intermediate result is therefore the pair (x + 1, x). After that, the function increments the right-hand component of the pair. So finally the result will be (x + 1, x + 1).

The combinators />> and //> are defined for functions manipulating pairs. The first applies the function to the first component of the pair, defined as

```
fun (x, y) > f = (f x, y)
```

and the second combinator to the second component, defined as

```
fun (x, y) | | > f = (x, f y)
```

With the combinator /-> you can re-combine the elements from a pair. This combinator is defined as

```
fun (x, y) \mid \rightarrow f = f x y
```

and can be used to write the following roundabout version of the double function:

```
fun double x = x > (fn x => (x, x))
|-> (fn x => fn y => x + y)
```

The combinator //>> plays a central rôle whenever your task is to update a theory and the update also produces a side-result (for example a theorem). Functions for such tasks return a pair whose second component is the theory and the fist component is the side-result. Using //>>, you can do conveniently the update and also accumulate the side-results. Consider the following simple function.

The purpose of Line 2 is to just pair up the argument with a dummy value (since //> operates on pairs). Each of the next three lines just increment the value by one, but also nest the intermediate results to the left. For example

```
acc_incs 1 > (((("", 1), 2), 3), 4)
```

You can continue this chain with:

```
acc_incs 1 ||>> (fn x => (x, x + 2))
> ((((("", 1), 2), 3), 4), 6)
```

(FIXME: maybe give a "real world" example for this combinator)

Recall that /> is the reverse function application. Recall also that the related reverse function composition is #>. In fact all the combinators /->, />> , //> and //>> described above have related combinators for function composition, namely #->, #>>, ##> and ##>>. Using #->, for example, the function double can also be written as:

```
val double =

(fn \ x \Rightarrow (x, \ x))
#-> (fn \ x \Rightarrow fn \ y \Rightarrow x + y)
```

(FIXME: find a good exercise for combinators)

#### Read More

The most frequently used combinators are defined in the files Pure/library.ML and Pure/General/basics.ML. Also [Impl. Man., Sec. B.1] contains further information about combinators.

(FIXME: say something about calling conventions)

## 2.4 Antiquotations

The main advantage of embedding all code in a theory is that the code can contain references to entities defined on the logical level of Isabelle. By this we mean definitions, theorems, terms and so on. This kind of reference is realised with antiquotations, sometimes also referred to as ML-antiquotations. For example, one can print out the name of the current theory by typing

```
Context.theory_name @{theory}
> "FirstSteps"
```

where <code>@{theory}</code> is an antiquotation that is substituted with the current theory (remember that we assumed we are inside the theory <code>FirstSteps</code>). The name of this theory can be extracted using the function <code>theory\_name</code>.

Note, however, that antiquotations are statically linked, that is their value is determined at "compile-time", not "run-time". For example the function

```
fun not_current_thyname () = Context.theory_name @{theory}
```

does *not* return the name of the current theory, if it is run in a different theory. Instead, the code above defines the constant function that always returns the string "FirstSteps", no matter where the function is called. Operationally speaking, the antiquotation <code>@{theory}</code> is not replaced with code that will look up the current theory in some data structure and return it. Instead, it is literally replaced with the value representing the theory name.

In a similar way you can use antiquotations to refer to proved theorems:  $Q\{thm ...\}$  for a single theorem

and O{thms ...} for more than one

```
@{thms conj_ac}
> (?P ∧ ?Q) = (?Q ∧ ?P)
> (?P ∧ ?Q ∧ ?R) = (?Q ∧ ?P ∧ ?R)
> ((?P ∧ ?Q) ∧ ?R) = (?P ∧ ?Q ∧ ?R)
```

The point of these antiquotations is that referring to theorems in this way makes your code independent from what theorems the user might have stored under this name (this becomes especially important when you deal with theorem lists; see Section 2.10).

You can also refer to the current simpset via an antiquotation. To illustrate this we implement the function that extracts the theorem names stored in a simpset.

```
fun get_thm_names_from_ss simpset =
let
  val {simps,...} = MetaSimplifier.dest_ss simpset
in
  map #1 simps
end
```

The function <code>dest\_ss</code> returns a record containing all information stored in the simpset, but we are only interested in the names of the simp-rules. Now you can feed in the current simpset into this function. The current simpset can be referred to using the antiquotation <code>@{simpset}</code>.

```
get_thm_names_from_ss @{simpset}
> ["Nat.of_nat_eq_id", "Int.of_int_eq_id", "Nat.One_nat_def", ...]
```

Again, this way of referencing simpsets makes you independent from additions of lemmas to the simpset by the user that can potentially cause loops in your code.

On the ML-level of Isabelle, you often have to work with qualified names. These are strings with some additional information, such as positional information and qualifiers. Such qualified names can be generated with the antiquotation <code>@{binding...}</code>. For example

```
@{binding "name"}
> name
```

An example where a qualified name is needed is the function define. This function is used below to define the constant TrueConj as the conjunction  $True \land True$ .

```
local_setup {*
    snd o LocalTheory.define Thm.internalK
        ((@{binding "TrueConj"}, NoSyn),
        (Attrib.empty_binding, @{term "True \( \) True"})) *}
```

Now querying the definition you obtain:

```
thm TrueConj_def
> TrueConj ≡ True ∧ True
```

(FIXME give a better example why bindings are important; maybe give a pointer to **local\_setup**; if not, then explain why *snd* is needed)

It is also possible to define your own antiquotations. But you should exercise care when introducing new ones, as they can also make your code also difficult to read. In the next section we will see how the (build in) antiquotation <code>@{term ...}</code> can be used to construct terms. A restriction of this antiquotation is that it does not allow you to use schematic variables. If you want an antiquotation that does not have this restriction, you can implement your own using the function <code>ML\_Antiquote.inline</code>. The code is as follows.

We call the new antiquotation <code>term\_pat</code> (Line 1); the parser in Line 2 provides us with a context and a string; this string is transformed into a term using the function <code>read\_term\_pattern</code> (Line 4); the next two lines print the term so that the ML-system can understand them. An example of this antiquotation is as follows.

```
@\{term\_pat "Suc (?x::nat)"\} > Const ("Suc", "nat <math>\Rightarrow nat") $ Var (("x", 0), "nat")
```

which is the internal representation of the term Suc?x.

#### Read More

The file Pure/ML/ml\_antiquote.ML contains the the definitions for most antiquotations.

Note one source of possible confusion about antiquotations. There are two kinds of them in Isabelle, which have very different purpose and infrastructures. The first kind, described in this section, are *ML-antiquotations*. They are used to refer to entities (like terms, types etc) from Isabelle's logic layer inside ML-code. They are "linked" statically at compile-time, which limits sometimes their usefulness in cases where, for example, terms needs to be built up dynamically.

The other kind of antiquotations are *document* antiquotations. They are used only in the text parts of Isabelle and their purpose is to print logical entities inside Lagardaceuments. They are part of the user level and therefore we are not interested in them in this Tutorial, except in Appendix A.1 where we show how to implement your own document antiquotations.

### 2.5 Terms and Types

One way to construct Isabelle terms, is by using the antiquotation <code>@{term ...}</code>. For example

```
@{term "(a::nat) + b = c"}
> Const ("op =", ...) $
> (Const ("HOL.plus_class.plus", ...) $ ... $ ...) $ ...
```

will show the term a + b = c, but printed using the internal representation corresponding to the datatype term defined as follows:

```
datatype term =
Const of string * typ

Free of string * typ

Var of indexname * typ

Bound of int

Abs of string * typ * term

for indexname * typ * term

for indexname * typ * term

for indexname * typ * term
```

This datatype implements lambda-terms typed in Church-style. As can be seen in Line 5, terms use the usual de Bruijn index mechanism for representing bound variables. For example in

```
0{term "\lambdax y. x y"} > Abs ("x", "'a \Rightarrow 'b", Abs ("y", "'a", Bound 1 $ Bound 0))
```

the indices refer to the number of Abstractions (*Abs*) that we need to skip until we hit the *Abs* that binds the corresponding variable. Constructing a term with dangling de Bruijn indices is possible, but will be flagged as ill-formed when you try to typecheck

or certify it (see Section 2.7). Note that the names of bound variables are kept at abstractions for printing purposes, and so should be treated only as "comments". Application in Isabelle is realised with the term-constructor \$.

Isabelle makes a distinction between *free* variables (term-constructor Free) and variables (term-constructor Var). The latter correspond to the schematic variables that when printed show up with a question mark in front of them. Consider the following two examples

```
let
    val v1 = Var (("x", 3), @{typ bool})
    val v2 = Var (("x1", 3), @{typ bool})
    val v3 = Free ("x", @{typ bool})
in
    map (Syntax.string_of_term @{context}) [v1, v2, v3]
    |> cat_lines
    |> tracing
end
> ?x3
> ?x1.3
> x
```

When constructing terms, you are usually concerned with free variables (as mentioned earlier, you cannot construct schematic variables using the antiquotation <code>@{term ...}</code>). If you deal with theorems, you have to, however, observe the distinction. A similar distinction holds for types (see below).

#### Read More

Terms and types are described in detail in [Impl. Man., Sec. 2.2]. Their definition and many useful operations are implemented in Pure/term.ML. For constructing terms involving HOL constants, many helper functions are defined in HOL/Tools/hologic.ML.

Constructing terms via antiquotations has the advantage that only typable terms can be constructed. For example

```
@{term "x x"}
> Type unification failed: Occurs check!
```

raises a typing error, while it perfectly ok to construct the term

```
let
  val omega = Free ("x", @{typ nat}) $ Free ("x", @{typ nat})
in
  tracing (Syntax.string_of_term @{context} omega)
end
> x x
```

with the raw ML-constructors.

Sometimes the internal representation of terms can be surprisingly different from what you see at the user-level, because the layers of parsing/type-checking/pretty printing can be quite elaborate.

**Exercise 2.5.1.** Look at the internal term representation of the following terms, and find out why they are represented like this:

- case x of  $0 \Rightarrow 0$  | Suc  $y \Rightarrow y$
- $\lambda(x, y)$ . P y x
- $\{[x] \mid x < -2\}$

Hint: The third term is already quite big, and the pretty printer may omit parts of it by default. If you want to see all of it, you can use the following ML-function to set the printing depth to a higher value:

```
print_depth 50
```

The antiquotation <code>@{prop ...}</code> constructs terms of propositional type, inserting the invisible <code>Trueprop</code>-coercions whenever necessary. Consider for example the pairs

```
(@{term "P x"}, @{prop "P x"})
> (Free ("P", ...) $ Free ("x", ...),
> Const ("Trueprop", ...) $ (Free ("P", ...) $ Free ("x", ...)))
```

where a coercion is inserted in the second component and

```
(@{term "P x \Longrightarrow Q x"}, @{prop "P x \Longrightarrow Q x"}) > (Const ("==>", ...) $ ... $ ..., Const ("==>", ...) $ ... $ ...)
```

where it is not (since it is already constructed by a meta-implication).

As already seen above, types can be constructed using the antiquotation  $@\{typ ...\}$ . For example:

```
@{typ "bool \Rightarrow nat"}
> bool \Rightarrow nat
```

The corresponding datatype is

```
datatype typ =
  Type of string * typ list
| TFree of string * sort
| TVar of indexname * sort
```

Like with terms, there is the distinction between free type variables (term-constructor *TFree* and schematic type variables (term-constructor *TVar*). A type constant, like *int* or *bool*, are types with an empty list of argument types. However, it is a bit difficult to show an example, because Isabelle always pretty-prints types (unlike terms). Here is a contrived example:

```
if Type ("bool", []) = @{typ "bool"} then true else false
> true
```

#### Read More

Types are described in detail in [Impl. Man., Sec. 2.1]. Their definition and many useful operations are implemented in Pure/type.ML.

### 2.6 Constructing Terms and Types Manually

While antiquotations are very convenient for constructing terms, they can only construct fixed terms (remember they are "linked" at compile-time). However, you often need to construct terms dynamically. For example, a function that returns the implication  $\bigwedge (x::nat)$ .  $P x \implies Q x$  taking P and Q as arguments can only be written as:

```
fun make_imp P Q =
let
  val x = Free ("x", @{typ nat})
in
  Logic.all x (Logic.mk_implies (P $ x, Q $ x))
end
```

The reason is that you cannot pass the arguments *P* and *Q* into an antiquotation.<sup>2</sup> For example the following does *not* work.

```
fun make_wrong_imp P Q = \mathcal{Q}\{prop \ " \land (x::nat). \ P x \implies Q x"\}
```

To see this, apply  $\mathcal{Q}\{term\ S\}$  and  $\mathcal{Q}\{term\ T\}$  to both functions. With  $make\_imp$  you obtain the intended term involving the given arguments

```
make_imp @{term S} @{term T}
> Const ... $
> Abs ("x", Type ("nat",[]),
> Const ... $ (Free ("S",...) $ ...) $ (Free ("T",...) $ ...))
```

whereas with  $make\_wrong\_imp$  you obtain a term involving the P and Q from the antiquotation.

```
make_wrong_imp @{term S} @{term T}
> Const ... $
> Abs ("x", ...,
> Const ... $ (Free ("P",...) $ ...)) $
> (Const ... $ (Free ("Q",...) $ ...)))
```

<sup>&</sup>lt;sup>2</sup>At least not at the moment.

There are a number of handy functions that are frequently used for constructing terms. One is the function <code>list\_comb</code>, which takes a term and a list of terms as arguments, and produces as output the term list applied to the term. For example

```
let
  val trm = @{term "P::nat"}
  val args = [@{term "True"}, @{term "False"}]
in
  list_comb (trm, args)
end
> Free ("P", "nat") $ Const ("True", "bool") $ Const ("False", "bool")
```

Another handy function is lambda, which abstracts a variable in a term. For example

```
let
  val x_nat = @{term "x::nat"}
  val trm = @{term "(P::nat \Rightarrow bool) x"}
in
  lambda x_nat trm
end
> Abs ("x", "nat", Free ("P", "bool \Rightarrow bool") $ Bound 0)
```

In this example, lambda produces a de Bruijn index (i.e. Bound 0), and an abstraction. It also records the type of the abstracted variable and for printing purposes also its name. Note that because of the typing annotation on P, the variable x in P x is of the same type as the abstracted variable. If it is of different type, as in

```
let
    val x_int = @{term "x::int"}
    val trm = @{term "(P::nat \Rightarrow bool) x"}
in
    lambda x_int trm
end
> Abs ("x", "int", Free ("P", "nat \Rightarrow bool") $ Free ("x", "nat"))
```

then the variable *Free* ("x", "int") is not abstracted. This is a fundamental principle of Church-style typing, where variables with the same name still differ, if they have different type.

There is also the function  $subst\_free$  with which terms can be replaced by other terms. For example below, we will replace in f 0 x the subterm f 0 by y, and x by True.

```
let  val \ s1 = ( @\{term \ "(f::nat \Rightarrow nat \Rightarrow nat) \ 0"\}, \ @\{term \ "y::nat \Rightarrow nat"\})   val \ s2 = ( @\{term \ "x::nat"\}, \ @\{term \ "True"\})   val \ trm =  @\{term \ "((f::nat \Rightarrow nat \Rightarrow nat) \ 0) \ x"\}  in
```

```
subst_free [s1, s2] trm end > Free ("y", "nat \Rightarrow nat") $ Const ("True", "bool")
```

As can be seen, <code>subst\_free</code> does not take typability into account. However it takes alpha-equivalence into account:

```
let
  val s = (@{term "(\lambda y::nat. y)"}, @{term "x::nat"})
  val trm = @{term "(\lambda x::nat. x)"}
in
  subst_free [s] trm
end
> Free ("x", "nat")
```

There are many convenient functions that construct specific HOL-terms. For example  $mk\_eq$  constructs an equality out of two terms. The types needed in this equality are calculated from the type of the arguments. For example

```
let
  val eq = HOLogic.mk_eq (@{term "True"}, @{term "False"})
in
  tracing (Syntax.string_of_term @{context} eq)
end
> True = False
```

#### Read More

```
Most of the HOL-specific operations on terms and types are defined in HOL/Tools/hologic.ML.
```

When constructing terms manually, there are a few subtle issues with constants. They usually crop up when pattern matching terms or types, or when constructing them. While it is perfectly ok to write the function <code>is\_true</code> as follows

this does not work for picking out ∀-quantified terms. Because the function

will not correctly match the formula " $\forall x::nat. P x$ ":

```
is_all @{term "∀x::nat. P x"} > false
```

The problem is that the @term-antiquotation in the pattern fixes the type of the constant All to be ('a  $\Rightarrow$  bool)  $\Rightarrow$  bool for an arbitrary, but fixed type 'a. A properly working alternative for this function is

because now

```
is_all @{term "∀x::nat. P x"}
> true
```

matches correctly (the first wildcard in the pattern matches any type and the second any term).

However there is still a problem: consider the similar function that attempts to pick out *Nil*-terms:

Unfortunately, also this function does not work as expected, since

```
is_nil @{term "Nil"}
> false
```

The problem is that on the ML-level the name of a constant is more subtle than you might expect. The function <code>is\_all</code> worked correctly, because <code>All</code> is such a fundamental constant, which can be referenced by <code>Const</code> ("All", <code>some\_type</code>). However, if you look at

```
@{term "Nil"}
> Const ("List.list.Nil", ...)
```

the name of the constant Nil depends on the theory in which the term constructor is defined (List) and also in which datatype (list). Even worse, some constants have a name involving type-classes. Consider for example the constants for zero and (op \*):

```
(@{term "0::nat"}, @{term "(op *)"})
> (Const ("HOL.zero_class.zero", ...),
> Const ("HOL.times_class.times", ...))
```

While you could use the complete name, for example <code>Const ("List.list.Nil", some\_type)</code>, for referring to or matching against <code>Nil</code>, this would make the code rather brittle. The reason is that the theory and the name of the datatype can easily change. To make the code more robust, it is better to use the antiquotation <code>@{const\_name ...}</code>. With this antiquotation you can harness the variable parts of the constant's name. Therefore a function for matching against constants that have a polymorphic type should be written as follows.

#### Read More

There are many functions in Pure/term.ML, Pure/logic.ML and HOL/Tools/hologic.ML that make such manual constructions of terms and types easier.

Have a look at these files and try to solve the following two exercises:

**Exercise 2.6.1.** Write a function rev\_sum: term  $\rightarrow$  term that takes a term of the form  $t_1 + t_2 + \ldots + t_n$  (whereby n might be one) and returns the reversed sum  $t_n + \ldots + t_2 + t_1$ . Assume the  $t_i$  can be arbitrary expressions and also note that + associates to the left. Try your function on some examples.

**Exercise 2.6.2.** Write a function which takes two terms representing natural numbers in unary notation (like Suc (Suc (Suc 0))), and produces the number representing their sum.

The antiquotation for properly referencing type constants is <code>@{type\_name ...}</code>. For example

```
@{type_name "list"}
> "List.list"
```

(FIXME: Explain the following better.)

Occasionally you have to calculate what the "base" name of a given constant is. For this you can use the function Sign.extern\_const or Long\_Name.base\_name. For example:

```
Sign.extern_const @{theory} "List.list.Nil" > "Nil"
```

The difference between both functions is that <code>extern\_const</code> returns the smallest name that is still unique, whereas <code>base\_name</code> always strips off all qualifiers.

#### Read More

Functions about naming are implemented in Pure/General/name\_space.ML; functions about signatures in Pure/sign.ML.

Although types of terms can often be inferred, there are many situations where you need to construct types manually, especially when defining constants. For example the function returning a function type is as follows:

```
fun make_fun_type ty1 ty2 = Type ("fun", [ty1, ty2])
```

This can be equally written with the combinator --> as:

```
fun make_fun_type ty1 ty2 = ty1 --> ty2
```

If you want to construct a function type with more than one argument type, then you can use --->.

```
fun make_fun_types tys ty = tys ---> ty
```

A handy function for manipulating terms is <code>map\_types</code>: it takes a function and applies it to every type in a term. You can, for example, change every <code>nat</code> in a term into an <code>int</code> using the function:

```
fun nat_to_int ty =
  (case ty of
     @{typ nat} => @{typ int}
     | Type (s, tys) => Type (s, map nat_to_int tys)
     | _ => ty)
```

Here is an example:

If you want to obtain the list of free type-variables of a term, you can use the function add\_tfrees (similarly add\_tvars for the schematic type-variables). One would expect that such functions take a term as input and return a list of types. But their type is actually

```
Term.term -> (string * Term.sort) list -> (string * Term.sort) list
```

that is they take, besides a term, also a list of type-variables as input. So in order to obtain the list of type-variables of a term you have to call them as follows

```
Term.add_tfrees @{term "(a, b)"} []
> [("'b", ["HOL.type"]), ("'a", ["HOL.type"])]
```

The reason for this definition is that add\_tfrees can be easily folded over a list of terms. Similarly for all functions named add\_\* in Pure/term.ML.

### 2.7 Type-Checking

You can freely construct and manipulate terms and types, since they are just arbitrary unchecked trees. However, you eventually want to see if a term is well-formed, or type-checks, relative to a theory. Type-checking is done via the function cterm\_of, which converts a term into a cterm, a certified term. Unlike terms, which are just trees, cterms are abstract objects that are guaranteed to be type-correct, and they can only be constructed via "official interfaces".

Type-checking is always relative to a theory context. For now we use the <code>@{theory}</code> antiquotation to get hold of the current theory. For example you can write:

```
cterm_of @{theory} @{term "(a::nat) + b = c"}
> a + b = c
```

This can also be written with an antiquotation:

```
@{cterm "(a::nat) + b = c"}
> a + b = c
```

Attempting to obtain the certified term for

```
@{cterm "1 + True"}
> Type unification failed ...
```

yields an error (since the term is not typable). A slightly more elaborate example that type-checks is:

```
let
  val natT = @{typ "nat"}
  val zero = @{term "0::nat"}
in
  cterm_of @{theory}
     (Const (@{const_name plus}, natT --> natT --> natT) $ zero $ zero)
end
> 0 + 0
```

In Isabelle not just terms need to be certified, but also types. For example, you obtain the certified type for the Isabelle type  $nat \Rightarrow bool$  on the ML-level as follows:

```
ctyp_of @{theory} (@{typ nat} --> @{typ bool})
> nat ⇒ bool
```

or with the antiquotation:

```
@{ctyp "nat \Rightarrow bool"}
> nat \Rightarrow bool
```

#### Read More

For functions related to cterms and ctyps see the file Pure/thm.ML.

**Exercise 2.7.1.** Check that the function defined in Exercise 2.6.1 returns a result that type-checks. See what happens to the solutions of this exercise given in B when they receive an ill-typed term as input.

Remember Isabelle follows the Church-style typing for terms, i.e., a term contains enough typing information (constants, free variables and abstractions all have typing information) so that it is always clear what the type of a term is. Given a well-typed term, the function *type\_of* returns the type of a term. Consider for example:

To calculate the type, this function traverses the whole term and will detect any typing inconsistency. For example changing the type of the variable x from nat to int will result in the error message:

```
type_of (0{term "f::nat \Rightarrow bool"} $ 0{term "x::int"}) > *** Exception- TYPE ("type_of: type mismatch in application" ...
```

Since the complete traversal might sometimes be too costly and not necessary, there is the function <code>fastype\_of</code>, which also returns the type of a term.

However, efficiency is gained on the expense of skipping some tests. You can see this in the following example

where no error is detected.

Sometimes it is a bit inconvenient to construct a term with complete typing annotations, especially in cases where the typing information is redundant. A short-cut is to use the "place-holder" type <code>dummyT</code> and then let type-inference figure out the complete type. An example is as follows:

```
let
  val c = Const (O{const_name "plus"}, dummyT)
  val o = O{term "1::nat"}
  val v = Free ("x", dummyT)
in
    Syntax.check_term O{context} (c $ o $ v)
end
> Const ("HOL.plus_class.plus", "nat ⇒ nat ⇒ nat") $
> Const ("HOL.one_class.one", "nat") $ Free ("x", "nat")
```

Instead of giving explicitly the type for the constant plus and the free variable x, type-inference fills in the missing information.

#### Read More

See Pure/Syntax/syntax.ML where more functions about reading, checking and pretty-printing of terms are defined. Functions related to type-inference are implemented in Pure/type.ML and Pure/type\_infer.ML.

(FIXME: say something about sorts)

#### 2.8 Theorems

Just like *cterms*, theorems are abstract objects of type *thm* that can only be built by going through interfaces. As a consequence, every proof in Isabelle is correct by construction. This follows the tradition of the LCF approach [2].

To see theorems in "action", let us give a proof on the ML-level for the following statement:

```
lemma assumes assm_1: "\bigwedge(x::nat). P x \implies Q x" and assm_2: "P t" shows "Q t"
```

The corresponding ML-code is as follows:

```
let  val \ assm1 = Q\{cprop \ "\ (x::nat). \ P \ x \implies Q \ x"\}   val \ assm2 = Q\{cprop \ "(P::nat \Rightarrow bool) \ t"\}   val \ Pt_implies_Qt =   assume \ assm1   |> forall_elim \ Q\{cterm \ "t::nat"\};   val \ Qt = implies_elim \ Pt_implies_Qt \ (assume \ assm2);  in  Qt   |> implies_intr \ assm2   |> implies_intr \ assm1  end  > [\ \land x. \ P \ x \implies Q \ x; \ P \ t] \implies Q \ t
```

This code-snippet constructs the following proof:

```
\frac{\bigwedge x. \ P \ x \implies Q \ x \vdash \bigwedge x. \ P \ x \implies Q \ x}{\bigwedge x. \ P \ x \implies Q \ x \vdash P \ t \implies Q \ t} \xrightarrow{(\bigwedge -elim)} \frac{}{P \ t \vdash P \ t} \xrightarrow{(assume)} \xrightarrow{(\Longrightarrow -elim)} \frac{\bigwedge x. \ P \ x \implies Q \ x, P \ t \vdash Q \ t}{\bigwedge x. \ P \ x \implies Q \ x \vdash P \ t \implies Q \ t} \xrightarrow{(\Longrightarrow -intro)} \xrightarrow{\vdash [\![\bigwedge x. \ P \ x \implies Q \ x; \ P \ t]\!] \implies Q \ t} \xrightarrow{(\Longrightarrow -intro)}
```

However, while we obtained a theorem as result, this theorem is not yet stored in Isabelle's theorem database. So it cannot be referenced later on. How to store theorems will be explained in Section 2.12.

#### Read More

For the functions assume, forall\_elim etc see [Impl. Man., Sec. 2.3]. The basic functions for theorems are defined in Pure/thm.ML.

(FIXME: handy functions working on theorems, like rulify and so on)

(FIXME: how to add case-names to goal states - maybe in the next section)

(FIXME: example for how to add theorem styles)

```
setup {*
   TermStyle.add_style "no_all_prem1" (style_parm_premise 1) #>
   TermStyle.add_style "no_all_prem2" (style_parm_premise 2)
*}
```

## 2.9 Setups (TBD)

In the previous section we used **setup** in order to make a theorem attribute known to Isabelle. What happens behind the scenes is that **setup** expects a function of type

theory -> theory: the input theory is the current theory and the output the theory where the theory attribute has been stored.

This is a fundamental principle in Isabelle. A similar situation occurs for example with declaring constants. The function that declares a constant on the ML-level is add\_consts\_i. If you write<sup>3</sup>

```
Sign.add_consts_i [(@{binding "BAR"}, @{typ "nat"}, NoSyn)] @{theory}
```

for declaring the constant BAR with type nat and run the code, then you indeed obtain a theory as result. But if you query the constant on the Isabelle level using the command **term** 

```
term "BAR" > "BAR" :: "'a"
```

you do not obtain a constant of type <code>nat</code>, but a free variable (printed in blue) of polymorphic type. The problem is that the ML-expression above did not register the declaration with the current theory. This is what the command <code>setup</code> is for. The constant is properly declared with

```
setup {* Sign.add_consts_i [(@{binding "BAR"}, @{typ "nat"}, NoSyn)] *}
```

Now

```
term "BAR"
> "BAR" :: "nat"
```

returns a (black) constant with the type nat.

A similar command is **local\_setup**, which expects a function of type <code>local\_theory</code> -> <code>local\_theory</code>. Later on we will also use the commands **method\_setup** for installing methods in the current theory and **simproc\_setup** for adding new simprocs to the current simpset.

#### 2.10 Theorem Attributes

Theorem attributes are [symmetric], [THEN ...], [simp] and so on. Such attributes are neither tags nor flags annotated to theorems, but functions that do further processing once a theorem is proved. In particular, it is not possible to find out what are all theorems that have a given attribute in common, unless of course the function behind the attribute stores the theorems in a retrievable data structure.

If you want to print out all currently known attributes a theorem can have, you can use the Isabelle command

<sup>&</sup>lt;sup>3</sup>Recall that ML-code needs to be enclosed in ML {\* ... \*}.

#### print\_attributes

```
> COMP: direct composition with rules (no lifting)
> HOL.dest: declaration of Classical destruction rule
> HOL.elim: declaration of Classical elimination rule
> ...
```

The theorem attributes fall roughly into two categories: the first category manipulates the proved theorem (for example [symmetric] and [THEN ...]), and the second stores the proved theorem somewhere as data (for example [simp], which adds the theorem to the current simpset).

To explain how to write your own attribute, let us start with an extremely simple version of the attribute [symmetric]. The purpose of this attribute is to produce the "symmetric" version of an equation. The main function behind this attribute is

```
val my_symmetric = Thm.rule_attribute (fn _ => fn thm => thm RS @{thm sym})
```

where the function  $rule_attribute$  expects a function taking a context (which we ignore in the code above) and a theorem (thm), and returns another theorem (namely thm resolved with the theorem sym:  $s = t \implies t = s$ ).<sup>4</sup> The function  $rule_attribute$  then returns an attribute.

Before we can use the attribute, we need to set it up. This can be done using the Isabelle command **attribute\_setup** as follows:

```
attribute_setup my_sym = {* Scan.succeed my_symmetric *}
"applying the sym rule"
```

Inside the {\* ... \*}, we have to specify a parser for the theorem attribute. Since the attribute does not expect any further arguments (unlike [THEN ...], for example), we use the parser <code>Scan.succeed</code>. Later on we will also consider attributes taking further arguments. An example for the attribute [my\_sym] is the proof

```
lemma test[my_sym]: "2 = Suc (Suc 0)" by simp
```

which stores the theorem  $Suc\ (Suc\ 0) = 2$  under the name test. You can see this, if you query the lemma:

```
thm test > Suc (Suc 0) = 2
```

We can also use the attribute when referring to this theorem:

```
thm test[my_sym]
> 2 = Suc (Suc 0)
```

An alternative for setting up an attribute is the function <code>setup</code>. So instead of using <code>attribute\_setup</code>, you can also set up the attribute as follows:

<sup>&</sup>lt;sup>4</sup>The function RS is explained in Section 4.2.

```
Attrib.setup @{binding "my_sym"} (Scan.succeed my_symmetric)
"applying the sym rule"
```

This gives a function from Context.theory -> Context.theory, which can be used for example with **setup**.

As an example of a slightly more complicated theorem attribute, we implement our own version of <code>[THEN ...]</code>. This attribute will take a list of theorems as argument and resolve the proved theorem with this list (one theorem after another). The code for this attribute is

```
fun MY_THEN thms =
  Thm.rule_attribute (fn _ => fn thm => foldl ((op RS) o swap) thm thms)
```

where swap swaps the components of a pair. The setup of this theorem attribute uses the parser thms, which parses a list of theorems.

```
attribute_setup MY_THEN = {* Attrib.thms >> MY_THEN *}
"resolving the list of theorems with the proved theorem"
```

You can, for example, use this theorem attribute to turn an equation into a metaequation:

```
thm test[MY_THEN eq_reflection] > Suc (Suc 0) \equiv 2
```

If you need the symmetric version as a meta-equation, you can write

```
thm test[MY_THEN sym eq_reflection] > 2 \equiv Suc (Suc 0)
```

It is also possible to combine different theorem attributes, as in:

```
thm test[my_sym, MY_THEN eq_reflection] > 2 \equiv Suc (Suc \ 0)
```

However, here also a weakness of the concept of theorem attributes shows through: since theorem attributes can be arbitrary functions, they do not in general commute. If you try

```
thm test[MY_THEN eq_reflection, my_sym]
> exception THM 1 raised: RSN: no unifiers
```

you get an exception indicating that the theorem sym does not resolve with meta-equations.

The purpose of *rule\_attribute* is to directly manipulate theorems. Another usage of theorem attributes is to add and delete theorems from stored data. For example

the theorem attribute [simp] adds or deletes a theorem from the current simpset. For these applications, you can use declaration\_attribute. To illustrate this function, let us introduce a reference containing a list of theorems.

```
val my_thms = ref ([] : thm list)
```

The purpose of this reference is to store a list of theorems. We are going to modify it by adding and deleting theorems. However, a word of warning: such references must not be used in any code that is meant to be more than just for testing purposes! Here it is only used to illustrate matters. We will show later how to store data properly without using references.

We need to provide two functions that add and delete theorems from this list. For this we use the two functions:

```
fun my_thm_add thm ctxt =
   (my_thms := Thm.add_thm thm (!my_thms); ctxt)

fun my_thm_del thm ctxt =
   (my_thms := Thm.del_thm thm (!my_thms); ctxt)
```

These functions take a theorem and a context and, for what we are explaining here it is sufficient that they just return the context unchanged. They change however the reference  $my\_thms$ , whereby the function  $add\_thm$  adds a theorem if it is not already included in the list, and  $del\_thm$  deletes one (both functions use the predicate  $eq\_thm\_prop$ , which compares theorems according to their proved propositions modulo alpha-equivalence).

You can turn functions my\_thm\_add and my\_thm\_del into attributes with the code

```
val my_add = Thm.declaration_attribute my_thm_add
val my_del = Thm.declaration_attribute my_thm_del
```

and set up the attributes as follows

```
attribute_setup my_thms = {* Attrib.add_del my_add my_del *}
"maintaining a list of my_thms - rough test only!"
```

The parser add\_del is a predefined parser for adding and deleting lemmas. Now if you prove the next lemma and attach to it the attribute [my\_thms]

```
lemma trueI_2[my_thms]: "True" by simp
```

then you can see it is added to the initially empty list.

```
!my_thms
> ["True"]
```

You can also add theorems using the command **declare**.

```
declare test[my_thms] trueI_2[my_thms add]
```

With this attribute, the add operation is the default and does not need to be explicitly given. These three declarations will cause the theorem list to be updated as:

```
!my_thms > ["True", "Suc (Suc 0) = 2"]
```

The theorem *trueI\_2* only appears once, since the function *add\_thm* tests for duplicates, before extending the list. Deletion from the list works as follows:

```
declare test[my_thms del]
```

After this, the theorem list is again:

```
!my_thms
> ["True"]
```

We used in this example two functions declared as *declaration\_attribute*, but there can be any number of them. We just have to change the parser for reading the arguments accordingly.

However, as said at the beginning of this example, using references for storing theorems is *not* the received way of doing such things. The received way is to start a "data slot", below called *MyThmsData*, generated by the functor *GenericDataFun*:

```
structure MyThmsData = GenericDataFun
(type T = thm list
val empty = []
val extend = I
fun merge _ = Thm.merge_thms)
```

The type T of this data slot is thm list.<sup>5</sup> To use this data slot, you only have to change my\_thm\_add and my\_thm\_del to:

```
val my_thm_add = MyThmsData.map o Thm.add_thm
val my_thm_del = MyThmsData.map o Thm.del_thm
```

where MyThmsData.map updates the data appropriately. The corresponding theorem attributes are

```
val my_add = Thm.declaration_attribute my_thm_add
val my_del = Thm.declaration_attribute my_thm_del
```

and the setup is as follows

<sup>&</sup>lt;sup>5</sup>FIXME: give a pointer to where data slots are explained properly.

```
attribute_setup my_thms2 = {* Attrib.add_del my_add my_del *}
"properly maintaining a list of my_thms"
```

Initially, the data slot is empty

```
MyThmsData.get (Context.Proof @{context})
> []
```

but if you prove

```
lemma three[my_thms2]: "3 = Suc (Suc (Suc 0))" by simp
```

then the lemma is recorded.

```
MyThmsData.get (Context.Proof @{context})
> ["3 = Suc (Suc (Suc 0))"]
```

With theorem attribute my\_thms2 you can also nicely see why it is important to store data in a "data slot" and not in a reference. Backtrack to the point just before the lemma three was proved and check the the content of MyThmsData: it should be empty. The addition has been properly retracted. Now consider the proof:

```
lemma four[my_thms]: "4 = Suc (Suc (Suc (Suc 0)))" by simp
```

Checking the content of my\_thms gives

```
!my_thms
> ["4 = Suc (Suc (Suc (Suc 0)))", "True"]
```

as expected, but if you backtrack before the lemma *four*, the content of *my\_thms* is unchanged. The backtracking mechanism of Isabelle is completely oblivious about what to do with references, but properly treats "data slots"!

Since storing theorems in a list is such a common task, there is the special functor Named\_Thms, which does most of the work for you. To obtain a named theorem list, you just declare

```
structure FooRules = Named_Thms
  (val name = "foo"
  val description = "Rules for foo")
```

and set up the FooRules with the command

```
setup {* FooRules.setup *}
```

This code declares a data slot where the theorems are stored, an attribute *foo* (with the *add* and *del* options for adding and deleting theorems) and an internal ML interface to retrieve and modify the theorems.

Furthermore, the facts are made available on the user-level under the dynamic fact name *foo*. For example you can declare three lemmas to be of the kind *foo* by:

```
lemma rule1[foo]: "A" sorry
lemma rule2[foo]: "B" sorry
lemma rule3[foo]: "C" sorry

and undeclare the first one by:
declare rule1[foo del]

and query the remaining ones with:
thm foo
> ?C
> ?B
```

On the ML-level the rules marked with foo can be retrieved using the function FooRules.get:

```
FooRules.get @{context}
> ["?C", "?B"]
```

#### Read More

For more information see Pure/Tools/named\_thms.ML.

(FIXME What are: theory\_attributes, proof\_attributes?)

#### Read More

FIXME: Pure/more\_thm.ML; parsers for attributes is in Pure/Isar/attrib.ML...also explained in the chapter about parsing.

## 2.11 Theories, Contexts and Local Theories (TBD)

There are theories, proof contexts and local theories (in this order, if you want to order them).

In contrast to an ordinary theory, which simply consists of a type signature, as well as tables for constants, axioms and theorems, a local theory contains additional context information, such as locally fixed variables and local assumptions that may be used by the package. The type <code>local\_theory</code> is identical to the type of <code>proof contexts Proof.context</code>, although not every proof context constitutes a valid local theory.

## 2.12 Storing Theorems (TBD)

```
add_thms_dynamic
local_setup {*
  LocalTheory.note Thm.theoremK
```

```
((0{binding "allI_alt"}, []), [0{thm allI}]) #> snd *}
```

## 2.13 Pretty-Printing

So far we printed out only plain strings without any formatting except for occasional explicit line breaks using " $\n$ ". This is sufficient for "quick-and-dirty" printouts. For something more sophisticated, Isabelle includes an infrastructure for properly formatting text. This infrastructure is loosely based on a paper by Oppen [4]. Most of its functions do not operate on strings, but on instances of the type:

```
Pretty.T
```

The function str transforms a (plain) string into such a pretty type. For example

```
Pretty.str "test" > String ("test", 4)
```

where the result indicates that we transformed a string with length 4. Once you have a pretty type, you can, for example, control where linebreaks may occur in case the text wraps over a line, or with how much indentation a text should be printed. However, if you want to actually output the formatted text, you have to transform the pretty type back into a *string*. This can be done with the function *string\_of*. In what follows we will use the following wrapper function for printing a pretty type:

```
fun pprint prt = tracing (Pretty.string_of prt)
```

The point of the pretty-printing infrastructure is to give hints about how to layout text and let Isabelle do the actual layout. Let us first explain how you can insert places where a line break can occur. For this assume the following function that replicates a string n times:

```
fun rep n str = implode (replicate n str)
```

and suppose we want to print out the string:

```
val test_str = rep 8 "foooooooooooooooobaaaaaaaaaaa"
```

We deliberately chose a large string so that it spans over more than one line. If we print out the string using the usual "quick-and-dirty" method, then we obtain the ugly output:

We obtain the same if we just use

However by using pretty types you have the ability to indicate a possible line break for example at each space. You can achieve this with the function *breaks*, which expects a list of pretty types and inserts a possible line break in between every two elements in this list. To print this list of pretty types as a single string, we concatenate them with the function *blk* as follows:

Here the layout of  $test\_str$  is much more pleasing to the eye. The 0 in blk stands for no indentation of the printed string. You can increase the indentation and obtain

where starting from the second line the indent is 3. If you want that every line starts with the same indent, you can use the function *indent* as follows:

If you want to print out a list of items separated by commas and have the linebreaks handled properly, you can use the function *commas*. For example

```
let
   val ptrs = map (Pretty.str o string_of_int) (99998 upto 100020)
in
   pprint (Pretty.blk (0, Pretty.commas ptrs))
end
> 99998, 99999, 100000, 100001, 100002, 100003, 100004, 100005, 100006,
> 100007, 100008, 100009, 100010, 100011, 100012, 100013, 100014, 100015,
> 100016, 100017, 100018, 100019, 100020
```

where upto generates a list of integers. You can print out this list as a "set", that means enclosed inside "{" and "}", and separated by commas using the function enum. For example

```
let
   val ptrs = map (Pretty.str o string_of_int) (99998 upto 100020)
in
   pprint (Pretty.enum "," "{" "}" ptrs)
end
> {99998, 99999, 100000, 100001, 100002, 100003, 100004, 100005, 100006,
> 100007, 100008, 100009, 100010, 100011, 100012, 100013, 100014, 100015,
> 100016, 100017, 100018, 100019, 100020}
```

As can be seen, this function prints out the "set" so that starting from the second, each new line as an indentation of 2.

If you print out something that goes beyond the capabilities of the standard functions, you can do relatively easily the formatting yourself. Assume you want to print out a list of items where like in "English" the last two items are separated by "and". For this you can write the function

```
fun and_list [] = []

/ and_list [x] = [x]

and_list xs =

let

val (front, last) = split_last xs
```

```
in
(Pretty.commas front) @
[Pretty.brk 1, Pretty.str "and", Pretty.brk 1, last]
end
```

where Line 7 prints the beginning of the list and Line 8 the last item. We have to use *Pretty.brk 1* in order to insert a space (of length 1) before the "and". This space is also a place where a line break can occur. We do the same after the "and". This gives you for example

```
let
  val ptrs = map (Pretty.str o string_of_int) (1 upto 22)
in
  pprint (Pretty.blk (0, and_list ptrs))
end
> 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12, 13, 14, 15, 16, 17, 18, 19, 20, 21
> and 22
```

Next we like to pretty-print a term and its type. For this we use the function tell\_type.

```
fun tell_type ctxt t =
2
  let
     fun pstr s = Pretty.breaks (map Pretty.str (space_explode " " s))
3
     val ptrm = Pretty.quote (Syntax.pretty_term ctxt t)
4
     val pty = Pretty.quote (Syntax.pretty_typ ctxt (fastype_of t))
5
6
     pprint (Pretty.blk (0,
7
       (pstr "The term " @ [ptrm] @ pstr " has type "
8
         @ [pty, Pretty.str "."])))
9
  end
10
```

In Line 3 we define a function that inserts possible linebreaks in places where a space is. In Lines 4 and 5 we pretty-print the term and its type using the functions <code>pretty\_term</code> and <code>pretty\_typ</code>. We also use the function <code>quote</code> in order to enclose the term and type inside quotation marks. In Line 9 we add a period right after the type without the possibility of a line break, because we do not want that a line break occurs there.

Now you can write

```
tell_type @\{context\} @\{term "min (Suc 0)"\} > The term "min (Suc 0)" has type "nat \Rightarrow nat".
```

To see the proper line breaking, you can try out the function on a bigger term and type. For example:

```
tell_type 0{\text{context}} 0{\text{term "op = (op = (op = (op = op =))))"}} > The term "op = (op = (op = (op = op =))))" has type > "((((('a \Rightarrow 'a \Rightarrow bool) \Rightarrow bool) \Rightarrow bool) \Rightarrow bool) \Rightarrow bool) \Rightarrow bool).
```

FIXME: TBD below

```
pprint (Pretty.big_list "header" (map (Pretty.str o string_of_int) (4
upto 10)))
```

chunks inserts forced breaks (unlike blk where you have to do this yourself)

the infrastructure of Syntax-pretty-term makes sure it is printed nicely

```
Pretty.mark Markup.hilite (Pretty.str "foo") |> Pretty.string_of |>
tracing
  (Pretty.str "bar") |> Pretty.string_of |> tracing

Pretty.mark Markup.subgoal (Pretty.str "foo") |> Pretty.string_of |> tracing
  (Pretty.str "bar") |> Pretty.string_of |> tracing
```

Still to be done:

What happens with big formulae?

#### Read More

The general infrastructure for pretty-printing is implemented in the file Pure/General/pretty.ML. The file Pure/Syntax.ML contains pretty-printing functions for terms, types, theorems and so on.

```
Pure/General/markup.ML
```

printing into the goal buffer as part of the proof state

```
Pretty.mark Markup.hilite (Pretty.str "foo") |> Pretty.string_of |>
tracing
  (Pretty.str "bar") |> Pretty.string_of |> tracing
```

writing into the goal buffer

```
fun my_hook interactive state =
          (interactive ? Proof.goal_message (fn () => Pretty.str
"foo")) state
```

```
setup {* Context.theory_map (Specification.add_theorem_hook my_hook) *}
lemma "False"
oops
```

## 2.14 Misc (TBD)

```
Datatype.get_info @{theory} "List.list"
```

## 2.15 Name Space Issues (TBD)

## Chapter 3

# **Parsing**

Isabelle distinguishes between *outer* and *inner* syntax. Commands, such as **definition**, **inductive** and so on, belong to the outer syntax, whereas terms, types and so on belong to the inner syntax. For parsing inner syntax, Isabelle uses a rather general and sophisticated algorithm, which is driven by priority grammars. Parsers for outer syntax are built up by functional parsing combinators. These combinators are a well-established technique for parsing, which has, for example, been described in Paulson's classic ML-book [5]. Isabelle developers are usually concerned with writing these outer syntax parsers, either for new definitional packages or for calling methods with specific arguments.

#### Read More

The library for writing parser combinators is split up, roughly, into two parts. The first part consists of a collection of generic parser combinators defined in the structure Scan in the file Pure/General/scan.ML. The second part of the library consists of combinators for dealing with specific token types, which are defined in the structure OuterParse in the file Pure/Isar/outer\_parse.ML. Specific parsers for packages are defined in Pure/Isar/spec\_parse.ML. Parsers for method arguments are defined in Pure/Isar/args.ML.

## 3.1 Building Generic Parsers

Let us first have a look at parsing strings using generic parsing combinators. The function \$\$ takes a string as argument and will "consume" this string from a given input list of strings. "Consume" in this context means that it will return a pair consisting of this string and the rest of the input list. For example:

```
($$ "h") (Symbol.explode "hello")
> ("h", ["e", "l", "o"])
```

```
($$ "w") (Symbol.explode "world") > ("w", ["o", "r", "d"])
```

The function \$\$ will either succeed (as in the two examples above) or raise the exception FAIL if no string can be consumed. For example trying to parse

```
($$ "x") (Symbol.explode "world")
> Exception FAIL raised
```

will raise the exception *FAIL*. There are three exceptions used in the parsing combinators:

- FAIL is used to indicate that alternative routes of parsing might be explored.
- MORE indicates that there is not enough input for the parser. For example in (\$\$ "h") [].
- ABORT is the exception that is raised when a dead end is reached. It is used for example in the function !! (see below).

However, note that these exceptions are private to the parser and cannot be accessed by the programmer (for example to handle them).

In the examples above we use the function <code>Symbol.explode</code>, instead of the more standard library function <code>explode</code>, for obtaining an input list for the parser. The reason is that <code>Symbol.explode</code> is aware of character sequences, for example <code>\<foo></code>, that have a special meaning in Isabelle. To see the difference consider

```
let
  val input = "\<foo> bar"
in
  (explode input, Symbol.explode input)
end
> (["\", "<", "f", "o", "o", ">", " ", "b", "a", "r"],
> ["\<foo>", " ", "b", "a", "r"])
```

Slightly more general than the parser \$\$ is the function one, in that it takes a predicate as argument and then parses exactly one item from the input list satisfying this predicate. For example the following parser either consumes an "h" or a "w":

```
let
    val hw = Scan.one (fn x => x = "h" orelse x = "w")
    val input1 = Symbol.explode "hello"
    val input2 = Symbol.explode "world"
in
        (hw input1, hw input2)
end
> (("h", ["e", "l", "l", "o"]),("w", ["o", "r", "l", "d"]))
```

Two parsers can be connected in sequence by using the function --. For example parsing h, e and 1 (in this order) you can achieve by:

```
($$ "h" -- $$ "e" -- $$ "l") (Symbol.explode "hello") > ((("h", "e"), "l"), ["l", "o"])
```

Note how the result of consumed strings builds up on the left as nested pairs. If, as in the previous example, you want to parse a particular string, then you should use the function this\_string:

```
Scan.this_string "hell" (Symbol.explode "hello") > ("hell", ["o"])
```

Parsers that explore alternatives can be constructed using the function //. The parser  $(p \ // \ q)$  returns the result of p, in case it succeeds, otherwise it returns the result of q. For example:

```
let
  val hw = $$ "h" || $$ "w"
  val input1 = Symbol.explode "hello"
  val input2 = Symbol.explode "world"
in
    (hw input1, hw input2)
end
> (("h", ["e", "l", "l", "o"]), ("w", ["o", "r", "l", "d"]))
```

The functions /-- and --/ work like the sequencing function for parsers, except that they discard the item being parsed by the first (respectively second) parser. For example:

```
let
  val just_e = $$ "h" |-- $$ "e"
  val just_h = $$ "h" --| $$ "e"
  val input = Symbol.explode "hello"
in
  (just_e input, just_h input)
end
> (("e", ["l", "l", "o"]), ("h", ["l", "l", "o"]))
```

The parser  $Scan.optional\ p\ x$  returns the result of the parser p, if it succeeds; otherwise it returns the default value x. For example:

```
let
  val p = Scan.optional ($$ "h") "x"
  val input1 = Symbol.explode "hello"
  val input2 = Symbol.explode "world"
in
    (p input1, p input2)
end
> (("h", ["e", "l", "l", "o"]), ("x", ["w", "o", "r", "l", "d"]))
```

The function *option* works similarly, except no default value can be given. Instead, the result is wrapped as an *option*-type. For example:

```
let
    val p = Scan.option ($$ "h")
    val input1 = Symbol.explode "hello"
    val input2 = Symbol.explode "world"
in
        (p input1, p input2)
end
> ((SOME "h", ["e", "l", "l", "o"]), (NONE, ["w", "o", "r", "l", "d"]))
```

The function !! helps to produce appropriate error messages for parsing. For example if you want to parse p immediately followed by q, or start a completely different parser r, you might write:

```
(p -- q) || r
```

However, this parser is problematic for producing an appropriate error message, if the parsing of (p-q) fails. Because in that case you lose the information that p should be followed by q. To see this assume that p is present in the input, but it is not followed by q. That means (p-q) will fail and hence the alternative parser r will be tried. However, in many circumstances this will be the wrong parser for the input "p-followed-by-something" and therefore will also fail. The error message is then caused by the failure of r, not by the absence of q in the input. This kind of situation can be avoided when using the function ! !. This function aborts the whole process of parsing in case of a failure and prints an error message. For example if you invoke the parser

```
!! (fn _ => "foo") ($$ "h")
```

on "hello", the parsing succeeds

```
(!! (fn _ => "foo") ($$ "h")) (Symbol.explode "hello") > ("h", ["e", "l", "o"])
```

but if you invoke it on "world"

```
(!! (fn _ => "foo") ($$ "h")) (Symbol.explode "world")
> Exception ABORT raised
```

then the parsing aborts and the error message *foo* is printed. In order to see the error message properly, you need to prefix the parser with the function *error*. For example:

```
Scan.error (!! (fn _ => "foo") ($$ "h"))
> Exception Error "foo" raised
```

This "prefixing" is usually done by wrappers such as <code>local\_theory</code> (see Section 3.7 which explains this function in more detail).

Let us now return to our example of parsing (p -- q) // r. If you want to generate the correct error message for p-followed-by-q, then you have to write:

```
fun p_followed_by_q p q r =
let
  val err_msg = fn _ => p ^ " is not followed by " ^ q
in
   ($$ p -- (!! err_msg ($$ q))) || ($$ r -- $$ r)
end
```

Running this parser with the arguments "h", "e" and "w", and the input "holle"

```
Scan.error (p_followed_by_q "h" "e" "w") (Symbol.explode "holle") > Exception ERROR "h is not followed by e" raised
```

produces the correct error message. Running it with

```
Scan.error (p_followed_by_q "h" "e" "w") (Symbol.explode "wworld") > (("w", "w"), ["o", "r", "l", "d"])
```

yields the expected parsing.

The function *Scan.repeat p* will apply a parser *p* as often as it succeeds. For example:

```
Scan.repeat ($$ "h") (Symbol.explode "hhhhello") > (["h", "h", "h", "h"], ["e", "l", "o"])
```

Note that repeat stores the parsed items in a list. The function repeat1 is similar, but requires that the parser p succeeds at least once.

Also note that the parser would have aborted with the exception MORE, if you had run it only on just "hhhh". This can be avoided by using the wrapper finite and the "stopper-token" stopper. With them you can write:

```
Scan.finite Symbol.stopper (Scan.repeat ($$ "h")) (Symbol.explode "hhhh") > (["h", "h", "h", "h"], [])
```

Symbol.stopper is the "end-of-input" indicator for parsing strings; other stoppers need to be used when parsing, for example, tokens. However, this kind of manually wrapping is often already done by the surrounding infrastructure.

The function repeat can be used with one to read any string as in

```
let
   val p = Scan.repeat (Scan.one Symbol.not_eof)
   val input = Symbol.explode "foo bar foo"
in
   Scan.finite Symbol.stopper p input
end
> (["f", "o", "o", " ", "b", "a", "r", " ", "f", "o", "o"], [])
```

where the function not\_eof ensures that we do not read beyond the end of the input string (i.e. stopper symbol).

The function  $Scan.unless\ p\ q$  takes two parsers: if the first one can parse the input, then the whole parser fails; if not, then the second is tried. Therefore

```
Scan.unless ($$ "h") ($$ "w") (Symbol.explode "hello") > Exception FAIL raised
```

fails, while

```
Scan.unless ($$ "h") ($$ "w") (Symbol.explode "world") > ("w",["o", "r", "l", "d"])
```

succeeds.

The functions repeat and unless can be combined to read any input until a certain marker symbol is reached. In the example below the marker symbol is a "\*".

```
let
    val p = Scan.repeat (Scan.unless ($$ "*") (Scan.one Symbol.not_eof))
    val input1 = Symbol.explode "foo*ooo"
    val input2 = Symbol.explode "foo*ooo"
in
        (Scan.finite Symbol.stopper p input1,
        Scan.finite Symbol.stopper p input2)
end
> ((["f", "o", "o", "o", "o", "o"], []),
> (["f", "o", "o"], ["*", "o", "o"]))
```

After parsing is done, you almost always want to apply a function to the parsed items. One way to do this is the function >> where (p >> f) runs first the parser p and upon successful completion applies the function f to the result. For example

```
let
  fun double (x, y) = (x ^ x, y ^ y)
in
  (($$ "h") -- ($$ "e") >> double) (Symbol.explode "hello")
end
> (("hh", "ee"), ["l", "l", "o"])
```

doubles the two parsed input strings; or

```
let
   val p = Scan.repeat (Scan.one Symbol.not_eof)
   val input = Symbol.explode "foo bar foo"
in
     Scan.finite Symbol.stopper (p >> implode) input
end
> ("foo bar foo",[])
```

where the single-character strings in the parsed output are transformed back into one string.

(FIXME: move to an earlier place)

The function ahead parses some input, but leaves the original input unchanged. For example:

```
Scan.ahead (Scan.this_string "foo") (Symbol.explode "foo") > ("foo", ["f", "o", "o"])
```

The function *lift* takes a parser and a pair as arguments. This function applies the given parser to the second component of the pair and leaves the first component untouched. For example

```
Scan.lift ($$ "h" -- $$ "e") (1, Symbol.explode "hello") > (("h", "e"), (1, ["l", "l", "o"]))
```

(FIXME: In which situations is this useful? Give examples.)

## 3.2 Parsing Theory Syntax

Most of the time, however, Isabelle developers have to deal with parsing tokens, not strings. These token parsers have the type:

```
type 'a parser = OuterLex.token list -> 'a * OuterLex.token list
```

The reason for using token parsers is that theory syntax, as well as the parsers for the arguments of proof methods, use the type <code>OuterLex.token</code>.

#### Read More

The parser functions for the theory syntax are contained in the structure OuterParse defined in the file Pure/Isar/outer\_parse.ML. The definition for tokens is in the file Pure/Isar/outer\_lex.ML.

The structure OuterLex defines several kinds of tokens (for example Ident for identifiers, Keyword for keywords and Command for commands). Some token parsers take into account the kind of tokens. The first example shows how to generate a token list out of a string using the function scan. It is given the argument Position.none since, at the moment, we are not interested in generating precise error messages. The following code<sup>1</sup>

```
OuterSyntax.scan Position.none "hello world"
> [Token (...,(Ident, "hello"),...),
> Token (...,(Space, " "),...),
> Token (...,(Ident, "world"),...)]
```

produces three tokens where the first and the last are identifiers, since "hello" and "world" do not match any other syntactic category. The second indicates a space.

We can easily change what is recognised as a keyword with *keyword*. For example calling this function

```
val _ = OuterKeyword.keyword "hello"
```

then lexing "hello world" will produce

```
OuterSyntax.scan Position.none "hello world"
> [Token (..., (Keyword, "hello"),...),
> Token (..., (Space, " "),...),
> Token (..., (Ident, "world"),...)]
```

Many parsing functions later on will require white space, comments and the like to have already been filtered out. So from now on we are going to use the functions filter and is\_proper to do this. For example:

```
let
    val input = OuterSyntax.scan Position.none "hello world"
in
    filter OuterLex.is_proper input
end
> [Token (...,(Ident, "hello"), ...), Token (...,(Ident, "world"), ...)]
```

For convenience we define the function:

<sup>&</sup>lt;sup>1</sup>Note that because of a possible bug in the PolyML runtime system, the result is printed as "?", instead of the tokens.

```
fun filtered_input str =
  filter OuterLex.is_proper (OuterSyntax.scan Position.none str)
```

If you now parse

```
filtered_input "inductive | for"
> [Token (..., (Command, "inductive"),...),
> Token (..., (Keyword, "|"),...),
> Token (..., (Keyword, "for"),...)]
```

you obtain a list consisting of only one command and two keyword tokens. If you want to see which keywords and commands are currently known to Isabelle, type:

```
let
  val (keywords, commands) = OuterKeyword.get_lexicons ()
in
  (Scan.dest_lexicon commands, Scan.dest_lexicon keywords)
end
> (["}", "{", ...], ["\Rightarrow", "\Rightarrow"])
```

You might have to adjust the *print\_depth* in order to see the complete list. The parser *\$\$\$* parses a single keyword. For example:

```
let
  val input1 = filtered_input "where for"
  val input2 = filtered_input "| in"
in
    (OuterParse.$$$ "where" input1, OuterParse.$$$ "|" input2)
end
> (("where",...), ("|",...))
```

Any non-keyword string can be parsed with the function reserved. For example:

```
let
  val p = OuterParse.reserved "bar"
  val input = filtered_input "bar"
in
  p input
end
> ("bar",[])
```

Like before, you can sequentially connect parsers with --. For example:

```
let
  val input = filtered_input "| in"
in
    (OuterParse.$$$ "|" -- OuterParse.$$$ "in") input
end
> (("|", "in"), [])
```

The parser  $OuterParse.enum\ s\ p$  parses a possibly empty list of items recognised by the parser p, where the items being parsed are separated by the string s. For example:

```
let
  val input = filtered_input "in | in | in foo"
in
    (OuterParse.enum "|" (OuterParse.$$$ "in")) input
end
> (["in", "in", "in"], [...])
```

enum1 works similarly, except that the parsed list must be non-empty. Note that we had to add a string "foo" at the end of the parsed string, otherwise the parser would have consumed all tokens and then failed with the exception MORE. Like in the previous section, we can avoid this exception using the wrapper Scan.finite. This time, however, we have to use the "stopper-token" OuterLex.stopper. We can write:

The following function will help to run examples.

```
fun parse p input = Scan.finite OuterLex.stopper (Scan.error p) input
```

The function !!! can be used to force termination of the parser in case of a dead end, just like <code>Scan.!!</code> (see previous section). Except that the error message of <code>OuterParse.!!!</code> is fixed to be "Outer syntax error" together with a relatively precise description of the failure. For example:

```
let
  val input = filtered_input "in |"
  val parse_bar_then_in = OuterParse.$$$ "|" -- OuterParse.$$$ "in"
in
  parse (OuterParse.!!! parse_bar_then_in) input
end
> Exception ERROR "Outer syntax error: keyword "|" expected,
> but keyword in was found" raised
```

**Exercise 3.2.1.** (FIXME) A type-identifier, for example 'a, is a token of kind Keyword. It can be parsed using the function type\_ident.

(FIXME: or give parser for numbers)

Whenever there is a possibility that the processing of user input can fail, it is a good idea to give all available information about where the error occurred. For this Isabelle can attach positional information to tokens and then thread this information up the processing chain. To see this, modify the function <code>filtered\_input</code> described earlier to

```
fun filtered_input' str =
    filter OuterLex.is_proper (OuterSyntax.scan (Position.line 7) str)
```

where we pretend the parsed string starts on line 7. An example is

```
filtered_input' "foo \n bar"
> [Token (("foo", ({line=7, end_line=7}, {line=7})), (Ident, "foo"), ...),
> Token (("bar", ({line=8, end_line=8}, {line=8})), (Ident, "bar"), ...)]
```

in which the " $\n$ " causes the second token to be in line 8.

By using the parser <code>OuterParse.position</code> you can access the token position and return it as part of the parser result. For example

```
let
  val input = filtered_input' "where"
in
  parse (OuterParse.position (OuterParse.$$$ "where")) input
end
> (("where", {line=7, end_line=7}), [])
```

#### Read More

The functions related to positions are implemented in the file Pure/General/position.ML.

(FIXME: there are also handy parsers for ML-expressions and ML-files)

## 3.3 Context Parser (TBD)

Used for example in **attribute\_setup** and **method\_setup**.

## 3.4 Argument and Attribute Parsers (TBD)

## 3.5 Parsing Inner Syntax

There is usually no need to write your own parser for parsing inner syntax, that is for terms and types: you can just call the predefined parsers. Terms can be parsed using the function term. For example:

```
let
  val input = OuterSyntax.scan Position.none "foo"
in
  OuterParse.term input
end
> ("\^E\^Ftoken\^Efoo\^E\^F\^E", [])
```

The function *prop* is similar, except that it gives a different error message, when parsing fails. As you can see, the parser not just returns the parsed string, but also some encoded information. You can decode the information with the function *parse*. For example

```
YXML.parse "\^E\^Ftoken\^Efoo\^E\^F\^E"
> XML.Elem ("token", [], [XML.Text "foo"])
```

The result of the decoding is an XML-tree. You can see better what is going on if you replace Position.none by Position.line 42, say:

```
let
  val input = OuterSyntax.scan (Position.line 42) "foo"
in
  YXML.parse (fst (OuterParse.term input))
end
> XML.Elem ("token", [("line", "42"), ("end_line", "42")], [XML.Text "foo"])
```

The positional information is stored as part of an XML-tree so that code called later on will be able to give more precise error messages.

#### Read More

The functions to do with input and output of XML and YXML are defined in Pure/General/xml.ML and Pure/General/yxml.ML.

## 3.6 Parsing Specifications

There are a number of special purpose parsers that help with parsing specifications of function definitions, inductive predicates and so on. In Chapter 5, for example, we will need to parse specifications for inductive predicates of the form:

# simple\_inductive even and odd where even0: "even 0" | evenS: "odd n ⇒ even (Suc n)" | oddS: "even n ⇒ odd (Suc n)"

For this we are going to use the parser:

```
val spec_parser =

OuterParse.fixes --

Scan.optional

(OuterParse.$$ "where" |--

OuterParse.!!!

(OuterParse.enum1 "|"

(SpecParse.opt_thm_name ":" -- OuterParse.prop))) []
```

Note that the parser must not parse the keyword **simple\_inductive**, even if it is meant to process definitions as shown above. The parser of the keyword will be given by the infrastructure that will eventually call <code>spec\_parser</code>.

To see what the parser returns, let us parse the string corresponding to the definition of even and odd:

As you see, the result is a pair consisting of a list of variables with optional typeannotation and syntax-annotation, and a list of rules where every rule has optionally a name and an attribute.

The function *fixes* in Line 2 of the parser reads an **and**-separated list of variables that can include optional type annotations and syntax translations. For example:<sup>2</sup>

```
let
  val input = filtered_input
        "foo::\"int \Rightarrow bool\" and bar::nat (\"BAR\" 100) and blonk"
in
  parse OuterParse.fixes input
end
> ([(foo, SOME "\^E\^Ftoken\^Eint \Rightarrow bool\^E\^F\^E", NoSyn),
> (bar, SOME "\^E\^Ftoken\^Enat\^E\^F\^E", Mixfix ("BAR", [], 100)),
> (blonk, NONE, NoSyn)],[])
```

Whenever types are given, they are stored in the *SOMEs*. The types are not yet used to type the variables: this must be done by type-inference later on. Since types are part

<sup>&</sup>lt;sup>2</sup>Note that in the code we need to write \"int  $\Rightarrow$  bool\" in order to properly escape the double quotes in the compound type.

of the inner syntax they are strings with some encoded information (see previous section). If a mixfix-syntax is present for a variable, then it is stored in the <code>Mixfix</code> data structure; no syntax translation is indicated by <code>NoSyn</code>.

#### Read More

The data structure for mixfix annotations is defined in Pure/Syntax/mixfix.ML.

Lines 3 to 7 in the function <code>spec\_parser</code> implement the parser for a list of introduction rules, that is propositions with theorem annotations such as rule names and attributes. The introduction rules are propositions parsed by <code>prop</code>. However, they can include an optional theorem name plus some attributes. For example

```
let
  val input = filtered_input "foo_lemma[intro,dest!]:"
  val ((name, attrib), _) = parse (SpecParse.thm_name ":") input
in
      (name, map Args.dest_src attrib)
end
> (foo_lemma, [(("intro", []), ...), (("dest", [...]), ...)])
```

The function  $opt_thm_name$  is the "optional" variant of  $thm_name$ . Theorem names can contain attributes. The name has to end with ":"—see the argument of the function  $SpecParse.opt_thm_name$  in Line 7.

#### Read More

Attributes and arguments are implemented in the files Pure/Isar/attrib.ML and Pure/Isar/args.ML.

Exercise 3.6.1. Have a look at how the parser SpecParse.where\_alt\_specs is implemented in file Pure/Isar/spec\_parse.ML. This parser corresponds to the "where-part" of the introduction rules given above. Below we paraphrase the code of where\_alt\_specs adapted to our purposes.

```
val spec_parser' =
1
        OuterParse.fixes --
2
3
        Scan.optional
        (OuterParse.$$$ "where" |--
4
           OuterParse.!!!
5
             (OuterParse.enum1 "|"
                 ((SpecParse.opt_thm_name ":" -- OuterParse.prop) --|
                      Scan.option (Scan.ahead (OuterParse.name | |
8
                      OuterParse. $$$ "[") --
9
                      OuterParse.!!! (OuterParse.$$$ "|"))))) []
10
```

Both parsers accept the same input, but if you look closely, you can notice an additional "tail" (Lines 8 to 10) in spec\_parser'. What is the purpose of this additional "tail"?

(FIXME: OuterParse.type\_args, OuterParse.typ, OuterParse.opt\_mixfix)

## 3.7 New Commands and Keyword Files

Often new commands, for example for providing new definitional principles, need to be implemented. While this is not difficult on the ML-level, new commands, in order to be useful, need to be recognised by ProofGeneral. This results in some subtle configuration issues, which we will explain in this section.

To keep things simple, let us start with a "silly" command that does nothing at all. We shall name this command **foobar**. On the ML-level it can be defined as:

```
let
  val do_nothing = Scan.succeed (LocalTheory.theory I)
  val kind = OuterKeyword.thy_decl
in
  OuterSyntax.local_theory "foobar" "description of foobar" kind do_nothing
end
```

The crucial function <code>local\_theory</code> expects a name for the command, a short description, a kind indicator (which we will explain later more thoroughly) and a parser producing a local theory transition (its purpose will also explained later).

While this is everything you have to do on the ML-level, you need a keyword file that can be loaded by ProofGeneral. This is to enable ProofGeneral to recognise **foobar** as a command. Such a keyword file can be generated with the command-line:

```
$ isabelle keywords -k foobar some_log_files
```

The option -k foobar indicates which postfix the name of the keyword file will be assigned. In the case above the file will be named <code>isar-keywords-foobar.el</code>. This command requires log files to be present (in order to extract the keywords from them). To generate these log files, you first need to package the code above into a separate theory file named <code>Command.thy</code>, say—see Figure 3.1 for the complete code.

For our purposes it is sufficient to use the log files of the theories *Pure*, *HOL* and *Pure-ProofGeneral*, as well as the log file for the theory *Command.thy*, which contains the new **foobar**-command. If you target other logics besides HOL, such as Nominal or ZF, then you need to adapt the log files appropriately.

*Pure* and *HOL* are usually compiled during the installation of Isabelle. So log files for them should be already available. If not, then they can be conveniently compiled with the help of the build-script from the Isabelle distribution.

```
$ ./build -m "Pure"
$ ./build -m "HOL"
```

The Pure-ProofGeneral theory needs to be compiled with:

```
$ ./build -m "Pure-ProofGeneral" "Pure"
```

For the theory Command. thy, you first need to create a "managed" subdirectory with:

```
$ isabelle mkdir FoobarCommand
```

```
theory Command
imports Main
begin
ML {*
let
    val do_nothing = Scan.succeed (LocalTheory.theory I)
    val kind = OuterKeyword.thy_decl
in
    OuterSyntax.local_theory "foobar" "description of foobar" kind do_nothing
end
*}
end
```

Figure 3.1: This file can be used to generate a log file. This log file in turn can be used to generate a keyword file containing the command **foobar**.

This generates a directory containing the files:

```
./IsaMakefile
./FoobarCommand/ROOT.ML
./FoobarCommand/document
./FoobarCommand/document/root.tex
```

You need to copy the file Command.thy into the directory FoobarCommand and add the line

```
no_document use_thy "Command";
```

to the file ./FoobarCommand/ROOT.ML. You can now compile the theory by just typing:

```
$ isabelle make
```

If the compilation succeeds, you have finally created all the necessary log files. They are stored in the directory

```
~/.isabelle/heaps/Isabelle2009/polyml-5.2.1_x86-linux/log
```

or something similar depending on your Isabelle distribution and architecture. One quick way to assign a shell variable to this directory is by typing

```
$ ISABELLE_LOGS="$(isabelle getenv -b ISABELLE_OUTPUT)"/log
```

on the Unix prompt. If you now type 1s \$ISABELLE\_LOGS, then the directory should include the files:

```
Pure.gz
HOL.gz
Pure-ProofGeneral.gz
HOL-FoobarCommand.gz
```

From them you can create the keyword files. Assuming the name of the directory is in *\$ISABELLE\_LOGS*, then the Unix command for creating the keyword file is:

```
$ isabelle keywords -k foobar
$ISABELLE_LOGS/{Pure.gz,HOL.gz,Pure-ProofGeneral.gz,HOL-FoobarCommand.gz}
```

The result is the file <code>isar-keywords-foobar.el</code>. It should contain the string <code>foobar</code> twice.<sup>3</sup> This keyword file needs to be copied into the directory <code>~/.isabelle/etc</code>. To make ProofGeneral aware of this keyword file, you have to start Isabelle with the option <code>-k foobar</code>, that is:

```
$ isabelle emacs -k foobar a_theory_file
```

If you now build a theory on top of *Command.thy*, then the command **foobar** can be used. Similarly with any other new command, and also any new keyword that is introduced with

```
val _ = OuterKeyword.keyword "blink"
```

At the moment the command **foobar** is not very useful. Let us refine it a bit next by letting it take a proposition as argument and printing this proposition inside the tracing buffer.

The crucial part of a command is the function that determines the behaviour of the command. In the code above we used a "do-nothing"-function, which because of <code>succeed</code> does not parse any argument, but immediately returns the simple function <code>LocalTheory.theory I.</code> We can replace this code by a function that first parses a proposition (using the parser <code>OuterParse.prop</code>), then prints out the tracing information (using a new function <code>trace\_prop</code>) and finally does nothing. For this you can write:

```
let
  fun trace_prop str =
    LocalTheory.theory (fn lthy => (tracing str; lthy))

val trace_prop_parser = OuterParse.prop >> trace_prop
  val kind = OuterKeyword.thy_decl
in
  OuterSyntax.local_theory "foobar" "traces a proposition"
    kind trace_prop_parser
end
```

#### Now you can type

<sup>&</sup>lt;sup>3</sup>To see whether things are fine, check that grep foobar on this file returns something non-empty.

```
foobar "True ∧ False"
> "True ∧ False"
```

and see the proposition in the tracing buffer.

Note that so far we used  $thy\_decl$  as the kind indicator for the command. This means that the command finishes as soon as the arguments are processed. Examples of this kind of commands are **definition** and **declare**. In other cases, commands are expected to parse some arguments, for example a proposition, and then "open up" a proof in order to prove the proposition (for example **lemma**) or prove some other properties (for example **function**). To achieve this kind of behaviour, you have to use the kind indicator  $thy\_goal$  and the function  $local\_theory\_to\_proof$  to set up the command. Note, however, once you change the "kind" of a command from  $thy\_decl$  to  $thy\_goal$  then the keyword file needs to be re-created!

Below we change **foobar** so that it takes a proposition as argument and then starts a proof in order to prove it. Therefore in Line 13, we set the kind indicator to thy\_goal.

```
let
1
2
     fun prove_prop str lthy =
3
         val prop = Syntax.read_prop lthy str
4
         Proof.theorem_i NONE (K I) [[(prop,[])]] lthy
6
7
8
     val prove_prop_parser = OuterParse.prop >> prove_prop
9
     val kind = OuterKeyword.thy_goal
10
11
     OuterSyntax.local_theory_to_proof "foobar" "proving a proposition"
12
       kind prove_prop_parser
13
   end
14
```

The function <code>prove\_prop</code> in Lines 2 to 7 takes a string (the proposition to be proved) and a context as argument. The context is necessary in order to be able to use <code>read\_prop</code>, which converts a string into a proper proposition. In Line 6 the function <code>theorem\_i</code> starts the proof for the proposition. Its argument <code>NONE</code> stands for a locale (which we chose to omit); the argument (<code>K I</code>) stands for a function that determines what should be done with the theorem once it is proved (we chose to just forget about it). Line 9 contains the parser for the proposition.

If you now type **foobar** "True \( \) True", you obtain the following proof state

```
foobar "True ∧ True"
goal (1 subgoal):
1. True ∧ True
```

and you can build the following proof

```
foobar "True \( \) True"
apply(rule conjI)
```

63

```
apply(rule TrueI)+
done
```

(FIXME: read a name and show how to store theorems; see note)

### 3.8 Methods (TBD)

(FIXME: maybe move to after the tactic section)

Methods are central to Isabelle. They are the ones you use for example in **apply**. To print out all currently known methods you can use the Isabelle command:

#### print\_methods

```
> methods:
> -: do nothing (insert current facts only)
> HOL.default: apply some intro/elim rule (potentially classical)
> ...
```

An example of a very simple method is:

```
method_setup foo =
  {* Scan.succeed
      (K (SIMPLE_METHOD ((etac @{thm conjE} THEN' rtac @{thm conjI}) 1))) *}
      "foo method for conjE and conjI"
```

It defines the method foo, which takes no arguments (therefore the parser Scan. succeed) and only applies a single tactic, namely the tactic which applies conjE and then conjI. The function SIMPLE\_METHOD turns such a tactic into a method. The method foo can be used as follows

```
lemma shows "A \wedge B \Longrightarrow C \wedge D" apply(foo)
```

where it results in the goal state

```
goal (2 subgoals):

1. \llbracket A; B \rrbracket \implies C 2. \llbracket A; B \rrbracket \implies D
```

(FIXME: explain a version of rule-tac)

# Chapter 4

# **Tactical Reasoning**

One of the main reason for descending to the ML-level of Isabelle is to be able to implement automatic proof procedures. Such proof procedures usually lessen considerably the burden of manual reasoning, for example, when introducing new definitions. These proof procedures are centred around refining a goal state using tactics. This is similar to the **apply**-style reasoning at the user-level, where goals are modified in a sequence of proof steps until all of them are solved. However, there are also more structured operations available on the ML-level that help with the handling of variables and assumptions.

## 4.1 Basics of Reasoning with Tactics

To see how tactics work, let us first transcribe a simple **apply**-style proof into ML. Suppose the following proof.

```
lemma disj\_swap: "P \lor Q \implies Q \lor P" apply(erule disjE) apply(rule disjI2) apply(assumption) apply(rule disjI1) apply(assumption) done
```

This proof translates to the following ML-code.

```
let  val \ ctxt =  @\{context\}   val \ goal =  @\{prop \ "P \lor Q \implies Q \lor P"\}  in  Goal.prove \ ctxt \ ["P", \ "Q"] \ [] \ goal   (fn \_ =>   etac \ @\{thm \ disjE\} \ 1   THEN \ rtac \ @\{thm \ disjI2\} \ 1   THEN \ atac \ 1   THEN \ atac \ 1)  end
```

```
> ?P ∨ ?Q ⇒ ?Q ∨ ?P
```

To start the proof, the function  $Goal.prove\ ctxt\ xs\ As\ C\ tac\ sets$  up a goal state for proving the goal C (that is  $P\ \lor\ Q\ \Longrightarrow\ Q\ \lor\ P$  in the proof at hand) under the assumptions As (happens to be empty) with the variables xs that will be generalised once the goal is proved (in our case P and Q). The tac is the tactic that proves the goal; it can make use of the local assumptions (there are none in this example). The tactics etac, rtac and atac in the code above correspond roughly to erule, rule and assumption, respectively. The operator THEN strings the tactics together.

#### Read More

To learn more about the function prove see [Impl. Man., Sec. 4.3] and the file Pure/goal.ML. See Pure/tactic.ML and Pure/tactical.ML for the code of basic tactics and tactic combinators; see also Chapters 3 and 4 in the old Isabelle Reference Manual, and Chapter 3 in the Isabelle/Isar Implementation Manual.

Note that in the code above we use antiquotations for referencing the theorems. Many theorems also have ML-bindings with the same name. Therefore, we could also just have written  $etac\ disjE\ 1$ , or in case where there is no ML-binding obtain the theorem dynamically using the function thm; for example  $etac\ (thm\ "disjE")\ 1$ . Both ways however are considered bad style! The reason is that the binding for disjE can be re-assigned by the user and thus one does not have complete control over which theorem is actually applied. This problem is nicely prevented by using antiquotations, because then the theorems are fixed statically at compile-time.

During the development of automatic proof procedures, you will often find it necessary to test a tactic on examples. This can be conveniently done with the command **apply**(tactic {\* ... \*}). Consider the following sequence of tactics

```
val foo_tac =
    (etac @{thm disjE} 1
    THEN rtac @{thm disjI2} 1
    THEN atac 1
    THEN rtac @{thm disjI1} 1
    THEN atac 1)
```

and the Isabelle proof:

```
lemma "P ∨ Q ⇒ Q ∨ P"
apply(tactic {* foo_tac *})
done
```

By using tactic {\* ... \*} you can call from the user-level of Isabelle the tactic foo\_tac or any other function that returns a tactic.

The tactic <code>foo\_tac</code> is just a sequence of simple tactics stringed together by <code>THEN</code>. As can be seen, each simple tactic in <code>foo\_tac</code> has a hard-coded number that stands for the subgoal analysed by the tactic (1 stands for the first, or top-most, subgoal). This hard-coding of goals is sometimes wanted, but usually it is not. To avoid the explicit numbering, you can write

```
val foo_tac' =
    (etac @{thm disjE}
    THEN' rtac @{thm disjI2}
    THEN' atac
    THEN' rtac @{thm disjI1}
    THEN' atac)
```

where THEN' is used instead of THEN. With foo\_tac' you can give the number for the subgoal explicitly when the tactic is called. So in the next proof you can first discharge the second subgoal, and subsequently the first.

```
lemma "P1 \lor Q1 \Longrightarrow Q1 \lor P1"
and "P2 \lor Q2 \Longrightarrow Q2 \lor P2"
apply(tactic {* foo_tac' 2 *})
apply(tactic {* foo_tac' 1 *})
done
```

This kind of addressing is more difficult to achieve when the goal is hard-coded inside the tactic. For most operators that combine tactics (*THEN* is only one such operator) a "primed" version exists.

The tactics  $foo_tac$  and  $foo_tac$ ' are very specific for analysing goals being only of the form  $P \lor Q \Longrightarrow Q \lor P$ . If the goal is not of this form, then these tactics return the error message:

```
*** empty result sequence -- proof command failed 
*** At command "apply".
```

This means they failed. The reason for this error message is that tactics are functions mapping a goal state to a (lazy) sequence of successor states. Hence the type of a tactic is:

```
type tactic = thm -> thm Seq.seq
```

By convention, if a tactic fails, then it should return the empty sequence. Therefore, if you write your own tactics, they should not raise exceptions willy-nilly; only in very grave failure situations should a tactic raise the exception *THM*.

The simplest tactics are no\_tac and all\_tac. The first returns the empty sequence and is defined as

```
fun no_tac thm = Seq.empty
```

which means no\_tac always fails. The second returns the given theorem wrapped in a single member sequence; it is defined as

<sup>&</sup>lt;sup>1</sup>To be precise, tactics do not produce this error message, it originates from the **apply** wrapper.

```
fun all_tac thm = Seq.single thm
```

which means all\_tac always succeeds, but also does not make any progress with the proof.

The lazy list of possible successor goal states shows through at the user-level of Isabelle when using the command **back**. For instance in the following proof there are two possibilities for how to apply <code>foo\_tac'</code>: either using the first assumption or the second.

```
lemma "[P \lor Q; P \lor Q] \Longrightarrow Q \lor P"
apply(tactic {* foo_tac' 1 *})
back
done
```

By using **back**, we construct the proof that uses the second assumption. While in the proof above, it does not really matter which assumption is used, in more interesting cases provability might depend on exploring different possibilities.

#### Read More

See Pure/General/seq.ML for the implementation of lazy sequences. In day-to-day Isabelle programming, however, one rarely constructs sequences explicitly, but uses the predefined tactics and tactic combinators instead.

It might be surprising that tactics, which transform one goal state to the next, are functions from theorems to theorem (sequences). The surprise resolves by knowing that every goal state is indeed a theorem. To shed more light on this, let us modify the code of all\_tac to obtain the following tactic

```
fun my_print_tac ctxt thm =
let
  val _ = tracing (string_of_thm_no_vars ctxt thm)
in
  Seq.single thm
end
```

which prints out the given theorem (using the string-function defined in Section 2.2) and then behaves like all\_tac. With this tactic we are in the position to inspect every goal state in a proof. Consider now the proof in Figure 4.1: as can be seen, internally every goal state is an implication of the form

```
A_1 \implies \ldots \implies A_n \implies \#C
```

where C is the goal to be proved and the  $A_i$  are the subgoals. So after setting up the lemma, the goal state is always of the form  $C \Longrightarrow \#C$ ; when the proof is finished we are left with #C. Since the goal C can potentially be an implication, there is a "protector" wrapped around it (the wrapper is the outermost constant

<sup>&</sup>lt;sup>2</sup>This only applies to single statements. If the lemma contains more than one statement, then one has more such implications.

```
notation (output) "prop" ("#_" [1000] 1000)
lemma shows "\llbracket A; B \rrbracket \implies A \land B"
apply(tactic {* my_print_tac @{context} *})
goal (1 subgoal):
 1. [A; B] \implies A \land B
 internal goal state:
 (\llbracket A; B \rrbracket \implies A \land B) \implies \#(\llbracket A; B \rrbracket \implies A \land B)
apply(rule conjI)
apply(tactic {* my_print_tac @{context} *})
goal (2 subgoals):
 1. \llbracket A; B \rrbracket \implies A
 2. \llbracket A; B \rrbracket \implies B
 internal goal state:
 \llbracket \llbracket A; \ B \rrbracket \implies A; \ \llbracket A; \ B \rrbracket \implies B \rrbracket \implies \#(\llbracket A; \ B \rrbracket \implies A \ \land \ B)
apply(assumption)
apply(tactic {* my_print_tac @{context} *})
goal (1 subgoal):
 1. \llbracket A; B \rrbracket \implies B
 internal goal state:
 (\llbracket A; B \rrbracket \implies B) \implies \#(\llbracket A; B \rrbracket \implies A \land B)
apply(assumption)
apply(tactic {* my_print_tac @{context} *})
No subgoals!
 internal goal state:
 \#(\llbracket A; B \rrbracket \implies A \land B)
```

Figure 4.1: The figure shows a proof where each intermediate goal state is printed by the Isabelle system and by  $my\_print\_tac$ . The latter shows the goal state as represented internally (highlighted boxes). This tactic shows that every goal state in Isabelle is represented by a theorem: when you start the proof of  $\llbracket A; B \rrbracket \implies A \land B$  the theorem is  $(\llbracket A; B \rrbracket \implies A \land B) \implies \#(\llbracket A; B \rrbracket \implies A \land B)$ ; when you finish the proof the theorem is  $\#(\llbracket A; B \rrbracket \implies A \land B)$ .

Const ("prop", bool  $\Rightarrow$  bool); in the figure we make it visible as an #). This wrapper prevents that premises of C are misinterpreted as open subgoals. While tactics can operate on the subgoals (the  $A_i$  above), they are expected to leave the conclusion C intact, with the exception of possibly instantiating schematic variables. If you use the predefined tactics, which we describe in the next section, this will always be the case.

#### Read More

For more information about the internals of goals see [Impl. Man., Sec. 3.1].

## 4.2 Simple Tactics

Let us start with explaining the simple tactic <code>print\_tac</code>, which is quite useful for low-level debugging of tactics. It just prints out a message and the current goal state. Unlike <code>my\_print\_tac</code> shown earlier, it prints the goal state as the user would see it. For example, processing the proof

```
lemma shows "False => True"
apply(tactic {* print_tac "foo message" *})
gives:
foo message
False => True
1. False => True
```

A simple tactic for easy discharge of any proof obligations is <code>cheat\_tac</code>. This tactic corresponds to the Isabelle command **sorry** and is sometimes useful during the development of tactics.

```
lemma shows "False" and "Goldbach_conjecture"
apply(tactic {* SkipProof.cheat_tac @{theory} *})
No subgoals!
```

This tactic works however only if the quick-and-dirty mode of Isabelle is switched on.

Another simple tactic is the function atac, which, as shown in the previous section, corresponds to the assumption command.

```
lemma shows "P \implies P"
apply(tactic {* atac 1 *})
No subgoals!
```

Similarly, rtac, dtac, etac and ftac correspond (roughly) to rule, drule, erule and frule, respectively. Each of them takes a theorem as argument and attempts to apply it to a goal. Below are three self-explanatory examples.

```
lemma shows "P \wedge Q" apply(tactic {* rtac @{thm conjI} 1 *})
```

```
goal (2 subgoals):
    1. P
    2. Q

lemma shows "P \land Q \Longrightarrow False"
apply(tactic {* etac @{thm conjE} 1 *})

goal (1 subgoal):
    1. [P; Q] \Longrightarrow False

lemma shows "False \land True \Longrightarrow False"
apply(tactic {* dtac @{thm conjunct2} 1 *})

goal (1 subgoal):
    1. True \Longrightarrow False
```

The function <code>resolve\_tac</code> is similar to <code>rtac</code>, except that it expects a list of theorems as arguments. From this list it will apply the first applicable theorem (later theorems that are also applicable can be explored via the lazy sequences mechanism). Given the code

```
val resolve_xmp_tac = resolve_tac [@{thm impI}, @{thm conjI}]
```

an example for resolve\_tac is the following proof where first an outermost implication is analysed and then an outermost conjunction.

```
lemma shows "C \longrightarrow (A \land B)" and "(A \longrightarrow B) \land C" apply(tactic {* resolve_xmp_tac 1 *}) apply(tactic {* resolve_xmp_tac 2 *})

goal (3 subgoals):

1. C \Longrightarrow A \land B

2. A \longrightarrow B

3. C
```

Similar versions taking a list of theorems exist for the tactics dtac (dresolve\_tac), etac (eresolve\_tac) and so on.

Another simple tactic is *cut\_facts\_tac*. It inserts a list of theorems into the assumptions of the current goal state. For example

```
lemma shows "True \neq False" apply(tactic {* cut_facts_tac [0{thm True_def}, 0{thm False_def}] 1 *}) produces the goal state goal (1 subgoal):

1. [True \equiv (\lambda x. \ x) = (\lambda x. \ x); False \equiv \forall P. P] \implies True \neq False
```

Since rules are applied using higher-order unification, an automatic proof procedure might become too fragile, if it just applies inference rules as shown above. The reason is that a number of rules introduce meta-variables into the goal state. Consider for example the proof

```
lemma shows "\forall x \in A. P x \Longrightarrow Q x" apply(tactic {* dtac Q{thm bspec} 1 *})
```

```
goal (2 subgoals):

1. ?x \in A

2. P ?x \Longrightarrow Q x
```

where the application of rule *bspec* generates two subgoals involving the metavariable ?x. Now, if you are not careful, tactics applied to the first subgoal might instantiate this meta-variable in such a way that the second subgoal becomes unprovable. If it is clear what the ?x should be, then this situation can be avoided by introducing a more constrained version of the *bspec*-rule. Such constraints can be given by pre-instantiating theorems with other theorems. One function to do this is RS

which in the example instantiates the first premise of the conjI-rule with the rule disjI1. If the instantiation is impossible, as in the case of

```
0{thm conjI} RS 0{thm mp} > *** Exception- THM ("RSN: no unifiers", 1,  
> ["[?P; ?Q]] \Longrightarrow ?P \land ?Q", "[?P \longrightarrow ?Q; ?P] \Longrightarrow ?Q"]) raised
```

then the function raises an exception. The function RSN is similar to RS, but takes an additional number as argument that makes explicit which premise should be instantiated.

To improve readability of the theorems we shall produce below, we will use the function  $no\_vars$  from Section 2.2, which transforms schematic variables into free ones. Using this function for the first RS-expression above produces the more readable result:

```
no_vars Q{context} (Q{thm disjI1} RS Q{thm conjI}) > [P; Q] \implies (P \lor Qa) \land Q
```

If you want to instantiate more than one premise of a theorem, you can use the function MRS:

```
no_vars @{context} ([@{thm disjI1}, @{thm disjI2}] MRS @{thm conjI}) > [P; Q] \implies (P \lor Qa) \land (Pa \lor Q)
```

If you need to instantiate lists of theorems, you can use the functions *RL* and *MRL*. For example in the code below, every theorem in the second list is instantiated with every theorem in the first.

```
map (no_vars Q\{context\})

([Q\{thm\ impI\}, Q\{thm\ disjI2\}] RL [Q\{thm\ conjI\}, Q\{thm\ disjI1\}])

> [[P\implies Q;\ Qa] \implies (P\longrightarrow Q)\land Qa,

> [Q;\ Qa] \implies (P\lor Q)\land Qa,

> (P\implies Q) \implies (P\longrightarrow Q)\lor Qa,

> Q\implies (P\lor Q)\lor Qa]
```

#### Read More

The combinators for instantiating theorems are defined in Pure/drule.ML.

Often proofs on the ML-level involve elaborate operations on assumptions and  $\land$  quantified variables. To do such operations using the basic tactics shown so far is very unwieldy and brittle. Some convenience and safety is provided by the functions *FOCUS* and *SUBPROOF*. These tactics fix the parameters and bind the various components of a goal state to a record. To see what happens, use the function defined in Figure 4.2, which takes a record and just prints out the contents of this record. Consider now the proof:

```
lemma shows "\bigwedge x y. A x y \Longrightarrow B y x \longrightarrow C (?z y) x" apply(tactic {* Subgoal.FOCUS foc_tac @{context}} 1 *})
```

The tactic produces the following printout:

```
params: x, y
schematics: ?z
assumptions: A \times y
conclusion: B \times x \longrightarrow C \times x
premises: A \times y
```

(FIXME: Find out how nowadays the schematics are handled)

Notice in the actual output the brown colour of the variables x, and y. Although they are parameters in the original goal, they are fixed inside the tactic. By convention these fixed variables are printed in brown colour. Similarly the schematic variable ?z. The assumption, or premise,  $A \times y$  is bound as cterm to the record-variable asms, but also as thm to prems.

If we continue the proof script by applying the *impI*-rule

```
apply(rule impI)
apply(tactic {* Subgoal.FOCUS foc_tac @{context} 1 *})
```

then the tactic prints out:

```
params: x, y
schematics: ?z
assumptions: A \times y, B \times y
conclusion: C \times (z \times y) \times x
premises: A \times y, B \times y \times x
```

Now also B y x is an assumption bound to asms and prems.

Figure 4.2: A function that prints out the various parameters provided by *FOCUS* and *SUBPROOF*. It uses the functions defined in Section 2.2 for extracting strings from *cterms* and *thms*.

The difference between the tactics *SUBPROOF* and *FOCUS* is that the former expects that the goal is solved completely, which the latter does not. *SUBPROOF* can also not instantiate an schematic variables.

One convenience of both FOCUS and SUBPROOF is that we can apply the assumptions using the usual tactics, because the parameter prems contains them as theorems. With this you can easily implement a tactic that behaves almost like atac:

```
val atac' = Subgoal.FOCUS (fn {prems, ...} => resolve_tac prems 1)

If you apply atac' to the next lemma
lemma shows "[B x y; A x y; C x y] => A x y"
apply(tactic {* atac' @{context} 1 *})

it will produce
No subgoals!
```

Notice that <code>atac</code>' inside <code>FOCUS</code> calls <code>resolve\_tac</code> with the subgoal number 1 and also the outer call to <code>FOCUS</code> in the <code>apply</code>-step uses 1. This is another advantage of <code>FOCUS</code> and <code>SUBPROOF</code>: the addressing inside it is completely local to the tactic inside the subproof. It is therefore possible to also apply <code>atac</code>' to the second goal by just writing:

```
lemma shows "True" and "[B \times y; A \times y; C \times y] \implies A \times y" apply(tactic {* atac' @\{context\} \ 2 \ *\}) apply(rule TrueI) done
```

#### Read More

The functions FOCUS and SUBPROOF are defined in Pure/subgoal.ML and also described in [Impl. Man., Sec. 4.3].

Similar but less powerful functions than FOCUS, respectively SUBPROOF, are SUBGOAL and CSUBGOAL. They allow you to inspect a given subgoal (the former presents the subgoal as a term, while the latter as a cterm). With this you can implement a tactic that applies a rule according to the topmost logic connective in the subgoal (to illustrate this we only analyse a few connectives). The code of the tactic is as follows.

```
fun select_tac (t, i) =
    case t of
        @{term "Trueprop"} $ t' => select_tac (t', i)
        | @{term "op \infty"} $ _ $ t' => select_tac (t', i)
        | @{term "op \cap "} $ _ $ _ => rtac @{thm conjI} i
        | @{term "op \cap "} $ _ $ _ => rtac @{thm impI} i
        | @{term "Not"} $ _ => rtac @{thm notI} i
        | Const (@{const_name "All"}, _) $ _ => rtac @{thm allI} i
        | _ => all_tac
```

The input of the function is a term representing the subgoal and a number specifying the subgoal of interest. In Line 3 you need to descend under the outermost Trueprop in order to get to the connective you like to analyse. Otherwise goals like  $A \wedge B$  are not properly analysed. Similarly with meta-implications in the next line. While for the first five patterns we can use the @term-antiquotation to construct the patterns, the pattern in Line 8 cannot be constructed in this way. The reason is that an antiquotation would fix the type of the quantified variable. So you really have to construct the pattern using the basic term-constructors. This is not necessary in other cases, because their type is always fixed to function types involving only the type bool. (See Section 2.6 about constructing terms manually.) For the catch-all pattern, we chose to just return  $all_tac$ . Consequently,  $select_tac$  never fails.

Let us now see how to apply this tactic. Consider the four goals:

```
lemma shows "A \land B" and "A \longrightarrow B \longrightarrow C" and "\forall x. D x" and "E \Longrightarrow F" apply(tactic {* SUBGOAL select_tac 4 *}) apply(tactic {* SUBGOAL select_tac 3 *}) apply(tactic {* SUBGOAL select_tac 2 *}) apply(tactic {* SUBGOAL select_tac 1 *}) goal (5 subgoals):

1. A

2. B

3. A \Longrightarrow B \longrightarrow C

4. \bigwedgex. D x

5. E \Longrightarrow F
```

where in all but the last the tactic applied an introduction rule. Note that we applied the tactic to the goals in "reverse" order. This is a trick in order to be independent from the subgoals that are produced by the rule. If we had applied it in the other order

```
lemma shows "A \wedge B" and "A \longrightarrow B \longrightarrow C" and "\forall x. D x" and "E \Longrightarrow F" apply(tactic {* SUBGOAL select_tac 1 *}) apply(tactic {* SUBGOAL select_tac 3 *}) apply(tactic {* SUBGOAL select_tac 4 *}) apply(tactic {* SUBGOAL select_tac 5 *})
```

then we have to be careful to not apply the tactic to the two subgoals produced by the first goal. To do this can result in quite messy code. In contrast, the "reverse application" is easy to implement.

Of course, this example is contrived: there are much simpler methods available in Isabelle for implementing a tactic analysing a goal according to its topmost connective. These simpler methods use tactic combinators, which we will explain in the next section.

## 4.3 Tactic Combinators

The purpose of tactic combinators is to build compound tactics out of smaller tactics. In the previous section we already used *THEN*, which just strings together two tactics in a sequence. For example:

```
lemma shows "(Foo \( \) Bar) \( \) False"
apply(tactic {* rtac @{thm conjI} 1 THEN rtac @{thm conjI} 1 *})
goal (3 subgoals):
    1. Foo
    2. Bar
    3. False
```

If you want to avoid the hard-coded subgoal addressing, then, as seen earlier, you can use the "primed" version of *THEN*. For example:

```
lemma shows "(Foo \( \) Bar) \( \) False"
apply(tactic {* (rtac \( \) {thm conjI} \) THEN' rtac \( \) {thm conjI} ) 1 *})
goal (3 subgoals):
1. Foo
2. Bar
3. False
```

Here you have to specify the subgoal of interest only once and it is consistently applied to the component tactics. For most tactic combinators such a "primed" version exists and in what follows we will usually prefer it over the "unprimed" one.

If there is a list of tactics that should all be tried out in sequence, you can use the combinator *EVERY'*. For example the function *foo\_tac'* from page 67 can also be written as:

There is even another way of implementing this tactic: in automatic proof procedures (in contrast to tactics that might be called by the user) there are often long lists of

tactics that are applied to the first subgoal. Instead of writing the code above and then calling foo\_tac', 1, you can also just write

and call foo\_tac1.

With the combinators *THEN'*, *EVERY'* and *EVERY1* it must be guaranteed that all component tactics successfully apply; otherwise the whole tactic will fail. If you rather want to try out a number of tactics, then you can use the combinator *ORELSE'* for two tactics, and *FIRST'* (or *FIRST1*) for a list of tactics. For example, the tactic

```
val orelse_xmp_tac = rtac @{thm disjI1} ORELSE' rtac @{thm conjI}
```

will first try out whether rule disjI applies and in case of failure will try conjI. To see this consider the proof

```
lemma shows "True \( \) False" and "Foo \( \) Bar"
apply(tactic \{* orelse_xmp_tac 2 *\})
apply(tactic \{* orelse_xmp_tac 1 *\})
which results in the goal state
goal (3 subgoals):
1. True
2. False
3. Foo
```

Using FIRST' we can simplify our select\_tac from Page 75 as follows:

Since we like to mimic the behaviour of <code>select\_tac</code> as closely as possible, we must include <code>all\_tac</code> at the end of the list, otherwise the tactic will fail if no rule applies (we also have to wrap <code>all\_tac</code> using the <code>K-combinator</code>, because it does not take a subgoal number as argument). You can test the tactic on the same goals:

```
lemma shows "A \wedge B" and "A \longrightarrow B \longrightarrow C" and "\forall x. D x" and "E \Longrightarrow F" apply(tactic {* select_tac' 4 *}) apply(tactic {* select_tac' 2 *}) apply(tactic {* select_tac' 1 *}) goal (5 subgoals):

1. A
2. B
3. A \Longrightarrow B \longrightarrow C
4. A \searrow D x
5. E \Longrightarrow F
```

Since such repeated applications of a tactic to the reverse order of *all* subgoals is quite common, there is the tactic combinator *ALLGOALS* that simplifies this. Using this combinator you can simply write:

```
lemma shows "A \land B" and "A \longrightarrow B \longrightarrow C" and "\forall x. D x" and "E \Longrightarrow F" apply(tactic {* ALLGOALS select_tac' *})

goal (5 subgoals):

1. A

2. B

3. A \Longrightarrow B \longrightarrow C

4. \bigwedgex. D x

5. E \Longrightarrow F
```

Remember that we chose to implement <code>select\_tac</code>' so that it always succeeds by adding <code>all\_tac</code> at the end of the tactic list. The same can be achieved with the tactic combinator <code>TRY</code>. For example:

This tactic behaves in the same way as <code>select\_tac'</code>: it tries out one of the given tactics and if none applies leaves the goal state unchanged. This, however, can be potentially very confusing when visible to the user, for example, in cases where the goal is the form

```
lemma shows "E \Longrightarrow F"
apply(tactic {* select_tac' 1 *})
goal (1 subgoal):
1. E \Longrightarrow F
```

In this case no rule applies, but because of TRY or the inclusion of all\_tac the tactics do not fail. The problem with this is that for the user there is little chance to see whether or not progress in the proof has been made. By convention therefore, tactics visible to the user should either change something or fail.

To comply with this convention, we could simply delete the <code>K</code> <code>all\_tac</code> from the end of the theorem list. As a result <code>select\_tac</code>, would only succeed on goals where it can make progress. But for the sake of argument, let us suppose that this deletion is <code>not</code> an option. In such cases, you can use the combinator <code>CHANGED</code> to make sure the subgoal has been changed by the tactic. Because now

```
lemma shows "E => F"
apply(tactic {* CHANGED (select_tac' 1) *})
gives the error message:

*** empty result sequence -- proof command failed
*** At command "apply".
```

We can further extend <code>select\_tac'</code> so that it not just applies to the topmost connective, but also to the ones immediately "underneath", i.e. analyse the goal completely. For this you can use the tactic combinator <code>REPEAT</code>. As an example suppose the following tactic

```
val repeat_xmp_tac = REPEAT (CHANGED (select_tac' 1))
```

which applied to the proof

3.  $C \longrightarrow D$ 

```
lemma shows "((\neg A) \land (\forall x. B x)) \land (C \longrightarrow D)"
apply(tactic {* repeat_xmp_tac *})
produces
goal (3 subgoals):
 1. A \Longrightarrow False
 2. \forall x. B x
```

Here it is crucial that select\_tac' is prefixed with CHANGED, because otherwise REPEAT runs into an infinite loop (it applies the tactic as long as it succeeds). The function REPEAT1 is similar, but runs the tactic at least once (failing if this is not possible).

If you are after the "primed" version of repeat\_xmp\_tac, then you can implement it as

```
val repeat_xmp_tac' = REPEAT o CHANGED o select_tac'
```

since there are no "primed" versions of REPEAT and CHANGED.

If you look closely at the goal state above, then you see the tactics repeat\_xmp\_tac and repeat\_xmp\_tac' are not yet quite what we are after: the problem is that goals 2 and 3 are not analysed. This is because the tactic is applied repeatedly only to the first subgoal. To analyse also all resulting subgoals, you can use the tactic combinator REPEAT\_ALL\_NEW. Suppose the tactic

```
val repeat_all_new_xmp_tac = REPEAT_ALL_NEW (CHANGED o select_tac')
```

you see that the following goal

```
lemma shows "((\neg A) \land (\forall x. B x)) \land (C \longrightarrow D)"
apply(tactic {* repeat_all_new_xmp_tac 1 *})
goal (3 subgoals):
 1. A \Longrightarrow False
 2. \bigwedge x. B x
 3. C \implies D
```

is completely analysed according to the theorems we chose to include in select\_tac'.

Recall that tactics produce a lazy sequence of successor goal states. These states can be explored using the command back. For example

```
lemma "\llbracket P1 \lor Q1; P2 \lor Q2 \rrbracket \Longrightarrow R"
apply(tactic {* etac @{thm disjE} 1 *})
```

applies the rule to the first assumption yielding the goal state:

```
goal (2 subgoals):
 1. \llbracket P2 \lor Q2; P1 \rrbracket \implies R
 2. \llbracket P2 \lor Q2; Q1 \rrbracket \implies R
```

After typing

#### back

the rule now applies to the second assumption.

```
goal (2 subgoals):

1. \llbracket P1 \lor Q1; P2 \rrbracket \implies R

2. \llbracket P1 \lor Q1; Q2 \rrbracket \implies R
```

Sometimes this leads to confusing behaviour of tactics and also has the potential to explode the search space for tactics. These problems can be avoided by prefixing the tactic with the tactic combinator *DETERM*.

```
lemma "\llbracket P1 \lor Q1; P2 \lor Q2 \rrbracket \Longrightarrow R" apply(tactic {* DETERM (etac \mathbb{Q}\{\text{thm disjE}\}\ 1)\ *\})
goal (2 subgoals):

1. \llbracket P2 \lor Q2; P1 \rrbracket \Longrightarrow R
2. \llbracket P2 \lor Q2; Q1 \rrbracket \Longrightarrow R
```

This combinator will prune the search space to just the first successful application. Attempting to apply **back** in this goal states gives the error message:

```
*** back: no alternatives
*** At command "back".
```

Recall that we implemented <code>select\_tac</code>, on Page 77 specifically so that it always succeeds. We achieved this by adding at the end the tactic <code>all\_tac</code>. We can achieve this also by using the combinator <code>TRY</code>. Suppose, for example the tactic

which will fail if none of the rules applies. However, if you prefix it as follows

```
val select_tac'' = TRY o select_tac''
```

then the tactic <code>select\_tac'</code> will be tried out and any failure is harnessed. We again have to use the construction with <code>TRY o ...</code> since there is no primed version of <code>TRY</code>. The tactic combinator <code>TRYALL</code> will try out a tactic on all subgoals. For example the tactic

```
val triv_tac = TRYALL (rtac @{thm TrueI} ORELSE' etac @{thm FalseE})
```

will solve all trivial subgoals involving True or False.

```
(FIXME: say something about COND and COND') (FIXME: PARALLEL-CHOICE PARALLEL-GOALS)
```

#### Read More

Most tactic combinators described in this section are defined in Pure/tactical.ML. Some combinators for the purpose of proof search are implemented in Pure/search.ML.

## 4.4 Simplifier Tactics

A lot of convenience in the reasoning with Isabelle derives from its powerful simplifier. The power of the simplifier is a strength and a weakness at the same time, because you can easily make the simplifier run into a loop and in general its behaviour can be difficult to predict. There is also a multitude of options that you can configure to control the behaviour of the simplifier. We describe some of them in this and the next section.

There are the following five main tactics behind the simplifier (in parentheses is their user-level counterpart):

```
simp_tac(simp (no_asm))asm_simp_tac(simp (no_asm_simp))full_simp_tac(simp (no_asm_use))asm_lr_simp_tac(simp (asm_lr))asm_full_simp_tac(simp)
```

All of the tactics take a simpset and an integer as argument (the latter as usual to specify the goal to be analysed). So the proof

```
lemma "Suc (1 + 2) < 3 + 2" apply(simp) done
```

corresponds on the ML-level to the tactic

```
lemma "Suc (1 + 2) < 3 + 2"
apply(tactic {* asm_full_simp_tac @{simpset} 1 *})
done</pre>
```

If the simplifier cannot make any progress, then it leaves the goal unchanged, i.e., does not raise any error message. That means if you use it to unfold a definition for a constant and this constant is not present in the goal state, you can still safely apply the simplifier.

When using the simplifier, the crucial information you have to provide is the simpset. If this information is not handled with care, then the simplifier can easily run into a loop. Therefore a good rule of thumb is to use simpsets that are as minimal as possible. It might be surprising that a simpset is more complex than just a simple collection of theorems used as simplification rules. One reason for the complexity is that the simplifier must be able to rewrite inside terms and should also be able to rewrite according to rules that have preconditions.

The rewriting inside terms requires congruence rules, which are meta-equalities typical of the form

$$\frac{\texttt{t}_1 \equiv \texttt{s}_1 \ \dots \ \texttt{t}_n \equiv \texttt{s}_n}{\texttt{constr} \ \texttt{t}_1 \dots \texttt{t}_n \equiv \texttt{constr} \ \texttt{s}_1 \dots \texttt{s}_n}$$

with *constr* being a constant, like *If* or *Let*. Every simpset contains only one congruence rule for each term-constructor, which however can be overwritten. The user can declare lemmas to be congruence rules using the attribute <code>[cong]</code>. In HOL, the user usually states these lemmas as equations, which are then internally transformed into meta-equations.

The rewriting with rules involving preconditions requires what is in Isabelle called a subgoaler, a solver and a looper. These can be arbitrary tactics that can be installed in a simpset and which are called at various stages during simplification. However, simpsets also include simprocs, which can produce rewrite rules on demand (see next section). Another component are split-rules, which can simplify for example the "then" and "else" branches of if-statements under the corresponding preconditions.

#### Read More

For more information about the simplifier see Pure/meta\_simplifier.ML and Pure/simplifier.ML. The simplifier for HOL is set up in HOL/Tools/simpdata.ML. The generic splitter is implemented in Provers/splitter.ML.

#### Read More

FIXME: Find the right place: Discrimination nets are implemented in Pure/net.ML.

The most common combinators to modify simpsets are:

```
addsimps delsimps
addcongs delcongs
addsimprocs delsimprocs
```

(FIXME: What about splitters? addsplits, delsplits)

To see how they work, consider the function in Figure 4.3, which prints out some parts of a simpset. If you use it to print out the components of the empty simpset, i.e., <code>empty\_ss</code>

```
print_ss @{context} empty_ss
> Simplification rules:
> Congruences rules:
> Simproc patterns:
```

you can see it contains nothing. This simpset is usually not useful, except as a building block to build bigger simpsets. For example you can add to <code>empty\_ss</code> the simplification rule <code>Diff\_Int</code> as follows:

```
val ss1 = empty_ss addsimps [0{thm Diff_Int} RS 0{thm eq_reflection}]
```

Printing then out the components of the simpset gives:

```
fun print_ss ctxt ss =
let
  val {simps, congs, procs, ...} = Simplifier.dest_ss ss

fun name_thm (nm, thm) =
        " " ^ nm ^ ": " ^ (string_of_thm_no_vars ctxt thm)
fun name_ctrm (nm, ctrm) =
        " " ^ nm ^ ": " ^ (string_of_cterms ctxt ctrm)

val s = ["Simplification rules:"] @ map name_thm simps @
        ["Congruences rules:"] @ map name_thm congs @
        ["Simproc patterns:"] @ map name_ctrm procs
in
    s |> cat_lines
    |> tracing
end
```

Figure 4.3: The function *dest\_ss* returns a record containing all printable information stored in a simpset. We are here only interested in the simplification rules, congruence rules and simprocs.

(FIXME: Why does it print out ??.unknown) Adding also the congruence rule *UN\_cong* 

```
val ss2 = ss1 addcongs [@{thm UN_cong} RS @{thm eq_reflection}]
```

gives

```
print_ss @{context} ss2 > Simplification rules: > ??.unknown: A - B \cap C \equiv A - B \cup (A - C) > Congruences rules: > UNION: [A = B; \land x. \ x \in B \implies C \ x = D \ x] \implies \bigcup x \in A. \ C \ x \equiv \bigcup x \in B. \ D \ x > Simproc patterns:
```

Notice that we had to add these lemmas as meta-equations. The <code>empty\_ss</code> expects this form of the simplification and congruence rules. However, even when adding these lemmas to <code>empty\_ss</code> we do not end up with anything useful yet.

In the context of HOL, the first really useful simpset is <code>HOL\_basic\_ss</code>. While printing out the components of this simpset

```
print_ss @{context} HOL_basic_ss
> Simplification rules:
> Congruences rules:
> Simproc patterns:
```

also produces "nothing", the printout is misleading. In fact the  $HOL\_basic\_ss$  is setup so that it can solve goals of the form

```
True, t = t, t \equiv t and False \Longrightarrow P;
```

and also resolve with assumptions. For example:

#### lemma

```
"True" and "t = t" and "t \equiv t" and "False \Longrightarrow Foo" and "[A; B; C] \Longrightarrow A" apply(tactic {* ALLGOALS (simp_tac HOL_basic_ss) *}) done
```

This behaviour is not because of simplification rules, but how the subgoaler, solver and looper are set up in <code>HOL\_basic\_ss</code>.

The simpset *HOL\_ss* is an extension of *HOL\_basic\_ss* containing already many useful simplification and congruence rules for the logical connectives in HOL.

The simplifier is often used to unfold definitions in a proof. For this the simplifier implements the tactic <code>rewrite\_goals\_tac.3</code> Suppose for example the definition

```
definition "MyTrue ≡ True"
```

then in the following proof we can unfold this constant

```
lemma shows "MyTrue ⇒ True ∧ True"
apply(rule conjI)
apply(tactic {* rewrite_goals_tac @{thms MyTrue_def} *})
producing the goal state
goal (2 subgoals):
1. True ⇒ True
2. True ⇒ True
```

As you can see, the tactic unfolds the definitions in all subgoals.

The simplifier is often used in order to bring terms into a normal form. Unfortunately, often the situation arises that the corresponding simplification rules will cause the simplifier to run into an infinite loop. Consider for example the simple theory about

<sup>&</sup>lt;sup>3</sup>FIXME: see LocalDefs infrastructure.

```
types prm = "(nat × nat) list"
consts perm :: "prm \Rightarrow 'a \Rightarrow 'a" ("_ \cdot _" [80,80] 80)
overloading
  \texttt{perm\_nat} \; \equiv \; \texttt{"perm} \; :: \; \texttt{prm} \; \Rightarrow \; \texttt{nat} \; \Rightarrow \; \texttt{nat"}
  perm\_prod \equiv "perm :: prm \Rightarrow ('a \times 'b) \Rightarrow ('a \times 'b)"
  \texttt{perm\_list} \ \equiv \ \texttt{"perm} \ :: \ \texttt{prm} \ \Rightarrow \ \texttt{'a list} \ \Rightarrow \ \texttt{'a list"}
begin
\mathbf{fun} \ \mathtt{swap} \colon : \texttt{"nat} \ \Rightarrow \ \mathtt{nat} \ \Rightarrow \ \mathtt{nat} \ \Rightarrow \ \mathtt{nat} "
where
 "swap a b c = (if c=a then b else (if c=b then a else c))"
primrec perm_nat
where
   "perm_nat [] c = c"
| "perm_nat (ab#pi) c = swap (fst ab) (snd ab) (perm_nat pi c)"
fun perm_prod
where
 "perm_prod pi (x, y) = (pi \cdot x, pi \cdot y)"
primrec perm_list
where
   "perm_list pi [] = []"
| "perm_list pi (x#xs) = (pi \cdot x)#(perm_list pi xs)"
end
lemma perm_append[simp]:
fixes c:: "nat" and pi1 pi2:: "prm"
shows "((pi_1@pi_2) \cdot c) = (pi_1 \cdot (pi_2 \cdot c))"
by (induct pi<sub>1</sub>) (auto)
lemma perm_bij[simp]:
fixes c d::"nat" and pi::"prm"
shows "(pi \cdot c = pi \cdot d) = (c = d)"
by (induct pi) (auto)
lemma perm_rev[simp]:
fixes c:: "nat" and pi:: "prm"
shows "pi·((rev pi)·c) = c"
by (induct pi arbitrary: c) (auto)
lemma perm_compose:
fixes c:: "nat" and pi1 pi2:: "prm"
shows "pi_1 \cdot (pi_2 \cdot c) = (pi_1 \cdot pi_2) \cdot (pi_1 \cdot c)"
by (induct pi2) (auto)
```

Figure 4.4: A simple theory about permutations over *nats*. The point is that the lemma *perm\_compose* cannot be directly added to the simplifier, as it would cause the simplifier to loop. It can still be used as a simplification rule if the permutation in the right-hand side is sufficiently protected.

permutations over natural numbers shown in Figure 4.4. The purpose of the lemmas is to push permutations as far inside as possible, where they might disappear by Lemma <code>perm\_rev</code>. However, to fully normalise all instances, it would be desirable to add also the lemma <code>perm\_compose</code> to the simplifier for pushing permutations over other permutations. Unfortunately, the right-hand side of this lemma is again an instance of the left-hand side and so causes an infinite loop. There seems to be no easy way to reformulate this rule and so one ends up with clunky proofs like:

#### lemma

```
fixes c d::"nat" and pi1 pi2::"prm"
shows "pi1 (c, pi2 ((rev pi1) d)) = (pi1 c, (pi1 pi2) d)"
apply(simp)
apply(rule trans)
apply(rule perm_compose)
apply(simp)
done
```

It is however possible to create a single simplifier tactic that solves such proofs. The trick is to introduce an auxiliary constant for permutations and split the simplification into two phases (below actually three). Let assume the auxiliary constant is

#### definition

```
perm_aux :: "prm \Rightarrow 'a \Rightarrow 'a" ("_ \dark aux _" [80,80] 80) where

"pi \dark aux c \equiv pi \dark c"

Now the two lemmas

lemma perm_aux_expand:
fixes c::"nat" and pi_1 pi_2::"prm"
shows "pi_1\dark (pi_2\dark c) = pi_1 \dark aux \dark (pi_2\dark c)"
unfolding perm_aux_def by (rule refl)

lemma perm_compose_aux:
fixes c::"nat" and pi_1 pi_2::"prm"
shows "pi_1\dark (pi_2\dark aux c) = (pi_1\dark pi_2) \dark aux \dark (pi_1\dark c)"
unfolding perm_aux_def by (rule perm_compose)
```

are simple consequence of the definition and <code>perm\_compose</code>. More importantly, the lemma <code>perm\_compose\_aux</code> can be safely added to the simplifier, because now the right-hand side is not anymore an instance of the left-hand side. In a sense it freezes all redexes of permutation compositions after one step. In this way, we can split simplification of permutations into three phases without the user noticing anything about the auxiliary constant. We first freeze any instance of permutation compositions in the term using lemma "<code>perm\_aux\_expand"</code> (Line 9); then simplify all other permutations including pushing permutations over other permutations by rule <code>perm\_compose\_aux</code> (Line 10); and finally "unfreeze" all instances of permutation compositions by unfolding the definition of the auxiliary constant.

```
val perm_simp_tac =
let
val thms1 = [@{thm perm_aux_expand}]
val thms2 = [@{thm perm_append}, @{thm perm_bij}, @{thm perm_rev},
```

4.5. SIMPROCS 87

For all three phases we have to build simpsets adding specific lemmas. As is sufficient for our purposes here, we can add these lemmas to <code>HOL\_basic\_ss</code> in order to obtain the desired results. Now we can solve the following lemma

#### lemma

```
fixes c d::"nat" and pi_1 pi_2::"prm" shows "pi_1 \cdot (c, pi_2 \cdot ((rev pi_1) \cdot d)) = (pi_1 \cdot c, (pi_1 \cdot pi_2) \cdot d)" apply(tactic {* perm_simp_tac 1 *}) done
```

in one step. This tactic can deal with most instances of normalising permutations. In order to solve all cases we have to deal with corner-cases such as the lemma being an exact instance of the permutation composition lemma. This can often be done easier by implementing a simproc or a conversion. Both will be explained in the next two chapters.

(FIXME: Is it interesting to say something about op =simp=>?)

(FIXME: What are the second components of the congruence rules—something to do with weak congruence constants?)

(FIXME: Anything interesting to say about Simplifier.clear\_ss?)

(FIXME: ObjectLogic.full\_atomize\_tac, ObjectLogic.rulify\_tac)

(FIXME: what are mksimps\_pairs; used in Nominal.thy)

(FIXME: explain simplify and Simplifier.rewrite\_rule etc.)

# 4.5 Simprocs

In Isabelle you can also implement custom simplification procedures, called *simprocs*. Simprocs can be triggered by the simplifier on a specified term-pattern and rewrite a term according to a theorem. They are useful in cases where a rewriting rule must be produced on "demand" or when rewriting by simplification is too unpredictable and potentially loops.

To see how simprocs work, let us first write a simproc that just prints out the pattern which triggers it and otherwise does nothing. For this you can use the function:

```
fun fail_simproc simpset redex =
let
val ctxt = Simplifier.the_context simpset
val _ = tracing ("The redex: " ^ (string_of_cterm ctxt redex))
in
```

```
6 NONE
7 end
```

This function takes a simpset and a redex (a cterm) as arguments. In Lines 3 and 4, we first extract the context from the given simpset and then print out a message containing the redex. The function returns NONE (standing for an optional thm) since at the moment we are not interested in actually rewriting anything. We want that the simproc is triggered by the pattern Suc n. This can be done by adding the simproc to the current simpset as follows

```
simproc_setup fail ("Suc n") = {* K fail_simproc *}
```

where the second argument specifies the pattern and the right-hand side contains the code of the simproc (we have to use *K* since we are ignoring an argument about morphisms. After this, the simplifier is aware of the simproc and you can test whether it fires on the lemma:

```
lemma shows "Suc 0 = 1"
apply(simp)
> The redex: Suc 0
> The redex: Suc 0
```

This will print out the message twice: once for the left-hand side and once for the right-hand side. The reason is that during simplification the simplifier will at some point reduce the term 1 to Suc 0, and then the simproc "fires" also on that term.

We can add or delete the simproc from the current simpset by the usual **declare**statement. For example the simproc will be deleted with the declaration

```
declare [[simproc del: fail]]
```

If you want to see what happens with just *this* simproc, without any interference from other rewrite rules, you can call *fail* as follows:

```
lemma shows "Suc 0 = 1"
apply(tactic {* simp_tac (HOL_basic_ss addsimprocs [@{simproc fail}]) 1*})
```

Now the message shows up only once since the term 1 is left unchanged.

Setting up a simproc using the command **simproc\_setup** will always add automatically the simproc to the current simpset. If you do not want this, then you have to use a slightly different method for setting up the simproc. First the function <code>fail\_simproc</code> needs to be modified to

```
fun fail_simproc' simpset redex =
let
  val ctxt = Simplifier.the_context simpset
  val _ = tracing ("The redex: " ^ (Syntax.string_of_term ctxt redex))
in
  NONE
end
```

4.5. SIMPROCS 89

Here the redex is given as a term, instead of a cterm (therefore we printing it out using the function string\_of\_term). We can turn this function into a proper simprocusing the function Simplifier.simproc\_i:

```
val fail' =
let
  val thy = @{theory}
  val pat = [@{term "Suc n"}]
in
  Simplifier.simproc_i thy "fail_simproc'" pat (K fail_simproc')
end
```

Here the pattern is given as term (instead of cterm). The function also takes a list of patterns that can trigger the simproc. Now the simproc is set up and can be explicitly added using addsimprocs to a simpset whenever needed.

Simprocs are applied from inside to outside and from left to right. You can see this in the proof

```
lemma shows "Suc (Suc 0) = (Suc 1)"
apply(tactic {* simp_tac (HOL_basic_ss addsimprocs [fail']) 1*})
```

The simproc fail, prints out the sequence

```
> Suc 0
> Suc (Suc 0)
> Suc 1
```

To see how a simproc applies a theorem, let us implement a simproc that rewrites terms according to the equation:

```
lemma plus_one:

shows "Suc n \equiv n + 1" by simp
```

Simprocs expect that the given equation is a meta-equation, however the equation can contain preconditions (the simproc then will only fire if the preconditions can be solved). To see that one has relatively precise control over the rewriting with simprocs, let us further assume we want that the simproc only rewrites terms "greater" than *Suc 0*. For this we can write

and set up the simproc as follows.

```
val plus_one =
let
  val thy = @{theory}
  val pat = [@{term "Suc n"}]
```

```
in
    Simplifier.simproc_i thy "sproc +1" pat (K plus_one_simproc)
end
```

Now the simproc is set up so that it is triggered by terms of the form *Suc n*, but inside the simproc we only produce a theorem if the term is not *Suc 0*. The result you can see in the following proof

```
lemma shows "P (Suc (Suc (Suc 0))) (Suc 0)"
apply(tactic {* simp_tac (HOL_basic_ss addsimprocs [plus_one]) 1*})
where the simproc produces the goal state
goal (1 subgoal):
    1. P (Suc 0 + 1 + 1) (Suc 0)
```

As usual with rewriting you have to worry about looping: you already have a loop with <code>plus\_one</code>, if you apply it with the default simpset (because the default simpset contains a rule which just does the opposite of <code>plus\_one</code>, namely rewriting "+ 1" to a successor). So you have to be careful in choosing the right simpset to which you add a simproc.

Next let us implement a simproc that replaces terms of the form  $Suc\ n$  with the number n increased by one. First we implement a function that takes a term and produces the corresponding integer value.

It uses the library function <code>dest\_number</code> that transforms (Isabelle) terms, like 0, 1, 2 and so on, into integer values. This function raises the exception <code>TERM</code>, if the term is not a number. The next function expects a pair consisting of a term <code>t</code> (containing <code>Sucs</code>) and the corresponding integer value <code>n</code>.

```
fun get_thm ctxt (t, n) =
let
val num = HOLogic.mk_number @{typ "nat"} n
val goal = Logic.mk_equals (t, num)
in
Goal.prove ctxt [] [] goal (K (Arith_Data.arith_tac ctxt 1))
end
```

From the integer value it generates the corresponding number term, called num (Line 3), and then generates the meta-equation  $t \equiv num$  (Line 4), which it proves by the arithmetic tactic in Line 6.

For our purpose at the moment, proving the meta-equation using <code>arith\_tac</code> is fine, but there is also an alternative employing the simplifier with a special simpset. For the kind of lemmas we want to prove here, the simpset <code>num\_ss</code> should suffice.

4.5. SIMPROCS 91

The advantage of <code>get\_thm\_alt</code> is that it leaves very little room for something to go wrong; in contrast it is much more difficult to predict what happens with <code>arith\_tac</code>, especially in more complicated circumstances. The disadvantage of <code>get\_thm\_alt</code> is to find a simpset that is sufficiently powerful to solve every instance of the lemmas we like to prove. This requires careful tuning, but is often necessary in "production code".<sup>4</sup>

Anyway, either version can be used in the function that produces the actual theorem for the simproc.

```
fun nat_number_simproc ss t =
let
  val ctxt = Simplifier.the_context ss
in
  SOME (get_thm ctxt (t, dest_suc_trm t))
  handle TERM _ => NONE
end
```

This function uses the fact that <code>dest\_suc\_trm</code> might raise an exception <code>TERM</code>. In this case there is nothing that can be rewritten and therefore no theorem is produced (i.e. the function returns <code>NONE</code>). To try out the simproc on an example, you can set it up as follows:

```
val nat_number =
let
  val thy = 0{theory}
  val pat = [0{term "Suc n"}]
in
    Simplifier.simproc_i thy "nat_number" pat (K nat_number_simproc)
end
```

Now in the lemma

```
lemma "P (Suc (Suc 2)) (Suc 99) (0::nat) (Suc 4 + Suc 0) (Suc (0 + 0))"
apply(tactic {* simp_tac (HOL_ss addsimprocs [nat_number]) 1*})

you obtain the more legible goal state
goal (1 subgoal):
    1. P 4 100 0 (5 + 1) (Suc (0 + 0))
```

<sup>&</sup>lt;sup>4</sup>It would be of great help if there is another way than tracing the simplifier to obtain the lemmas that are successfully applied during simplification. Alas, there is none.

where the simproc rewrites all Sucs except in the last argument. There it cannot rewrite anything, because it does not know how to transform the term Suc (0 + 0) into a number. To solve this problem have a look at the next exercise.

**Exercise 4.5.1.** Write a simproc that replaces terms of the form  $t_1 + t_2$  by their result. You can assume the terms are "proper" numbers, that is of the form 0, 1, 2 and so on.

(FIXME: We did not do anything with morphisms. Anything interesting one can say about them?)

## 4.6 Conversions

Conversions are a thin layer on top of Isabelle's inference kernel, and can be viewed as a controllable, bare-bone version of Isabelle's simplifier. One difference between conversions and the simplifier is that the former act on cterms while the latter acts on thms. However, we will also show in this section how conversions can be applied to theorems via tactics. The type for conversions is

```
type conv = cterm -> thm
```

whereby the produced theorem is always a meta-equality. A simple conversion is the function all\_conv, which maps a cterm to an instance of the (meta)reflexivity theorem. For example:

```
Conv.all_conv @{cterm "Foo \times Bar"}
> Foo \times Bar \equiv Foo \times Bar
```

Another simple conversion is no\_conv which always raises the exception CTERM.

```
Conv.no_conv @{cterm True}
> *** Exception- CTERM ("no conversion", []) raised
```

A more interesting conversion is the function <code>beta\_conversion</code>: it produces a meta-equation between a term and its beta-normal form. For example

```
let val \ add =  @\{cterm \ "\lambda x \ y. \ x + (y::nat)"\}  val \ two =  @\{cterm \ "2::nat"\}  val \ ten =  @\{cterm \ "10::nat"\}  val \ ctrm =  Thm.capply \ (Thm.capply \ add \ two) \ ten  in Thm.beta\_conversion \ true \ ctrm  end  >  ((\lambda x \ y. \ x + y) \ 2) \ 10 \ \equiv \ 2 + 10
```

4.6. CONVERSIONS 93

If you run this example, you will notice that the actual response is the seemingly nonsensical  $2 + 10 \equiv 2 + 10$ . The reason is that the pretty-printer for *cterms* etanormalises terms and therefore produces this output. If we get hold of the "raw" representation of the produced theorem, we obtain the expected result.

```
let
  val add = @{cterm "\lambda x y. x + (y::nat)"}
  val two = @{cterm "2::nat"}
  val ten = @{cterm "10::nat"}
  val ctrm = Thm.capply (Thm.capply add two) ten
in
    Thm.prop_of (Thm.beta_conversion true ctrm)
end
> Const ("==",...) $
> (Abs ("x",...,Abs ("y",...,...)) $....$...)
> (Const ("HOL.plus_class.plus",...) $ ....$...)
```

The argument true in beta\_conversion indicates that the right-hand side should be fully beta-normalised. If instead false is given, then only a single beta-reduction is performed on the outer-most level.

The main point of conversions is that they can be used for rewriting cterms. One example is the function rewr\_conv, which expects a meta-equation as an argument. Suppose the following meta-equation.

```
lemma true_conj1: "True \land P \equiv P" by simp
```

It can be used for example to rewrite  $True \land (Foo \longrightarrow Bar)$  to  $Foo \longrightarrow Bar$ . The code is as follows.

```
let
  val ctrm = @{cterm "True ∧ (Foo → Bar)"}
in
  Conv.rewr_conv @{thm true_conj1} ctrm
end
> True ∧ (Foo → Bar) ≡ Foo → Bar
```

Note, however, that the function  $rewr\_conv$  only rewrites the outer-most level of the cterm. If the given cterm does not match exactly the left-hand side of the theorem, then  $rewr\_conv$  fails, raising the exception CTERM.

This very primitive way of rewriting can be made more powerful by combining several conversions into one. For this you can use conversion combinators. The simplest conversion combinator is <code>then\_conv</code>, which applies one conversion after another. For example

```
let
  val conv1 = Thm.beta_conversion false
  val conv2 = Conv.rewr_conv @{thm true_conj1}
  val ctrm = Thm.capply @{cterm "λx. x ∧ False"} @{cterm "True"}
```

```
in (conv1 then\_conv conv2) ctrm end > (\lambda x. x \land False) True \equiv False
```

where we first beta-reduce the term and then rewrite according to *true\_conj1*. (When running this example recall the problem with the pretty-printer normalising all terms.)

The conversion combinator <code>else\_conv</code> tries out the first one, and if it does not apply, tries the second. For example

```
let val\ conv = Conv.rewr\_conv\ @\{thm\ true\_conj1\}\ else\_conv\ Conv.all\_conv val\ ctrm1 = @\{cterm\ "True\ \land\ Q"\} val\ ctrm2 = @\{cterm\ "P\ \lor\ (True\ \land\ Q)"\} in (conv\ ctrm1,\ conv\ ctrm2) end > (True\ \land\ Q\ \equiv\ Q,\ P\ \lor\ True\ \land\ Q\ \equiv\ P\ \lor\ True\ \land\ Q)
```

Here the conversion of true\_conj1 only applies in the first case, but fails in the second. The whole conversion does not fail, however, because the combinator else\_conv will then try out all\_conv, which always succeeds.

The conversion combinator *try\_conv* constructs a conversion which is tried out on a term, but in case of failure just does nothing. For example

```
let
  val conv = Conv.try_conv (Conv.rewr_conv @{thm true_conj1})
  val ctrm = @{cterm "True \vee P"}
in
  conv ctrm
end
> True \vee P \equiv True \vee P
```

Apart from the function beta\_conversion, which is able to fully beta-normalise a term, the conversions so far are restricted in that they only apply to the outermost level of a cterm. In what follows we will lift this restriction. The combinators fun\_conv and arg\_conv will apply a conversion to the first, respectively second, argument of an application. For example

```
let
  val conv = Conv.arg_conv (Conv.rewr_conv @{thm true_conj1})
  val ctrm = @{cterm "P \leftarrow (True \land Q)"}
in
  conv ctrm
end
> P \leftarrow (True \land Q) \equiv P \leftarrow Q
```

4.6. CONVERSIONS 95

The reason for this behaviour is that  $(op \lor)$  expects two arguments. Therefore the term must be of the form  $(Const \ldots \$ t1) \$ t2$ . The conversion is then applied to t2, which in the example above stands for  $True \land Q$ . The function  $fun\_conv$  would apply the conversion to the term  $(Const \ldots \$ t1)$ .

The function abs\_conv applies a conversion under an abstraction. For example:

```
let val\ conv = Conv.rewr\_conv\ @\{thm\ true\_conj1\} \\ val\ ctrm = @\{cterm\ "\lambda P.\ True\ \land\ (P\ \land\ Foo)"\} \\ in \\ Conv.abs\_conv\ (K\ conv)\ @\{context\}\ ctrm \\ end \\ > \lambda P.\ True\ \land\ (P\ \land\ Foo)\ \equiv\ \lambda P.\ P\ \land\ Foo \\ \end{cases}
```

Note that this conversion needs a context as an argument. We also give the conversion as (K conv), which is a function that ignores its argument (the argument being a sufficiently freshened version of the variable that is abstracted and a context). The conversion that goes under an application is combination\_conv. It expects two conversions as arguments, each of which is applied to the corresponding "branch" of the application. An abbreviation for this conversion is the function comb\_conv, which applies the same conversion to both branches.

We can now apply all these functions in a conversion that recursively descends a term and applies a "true\_conj1"-conversion in all possible positions.

This function "fires" if the terms is of the form ( $True \land ...$ ). It descends under applications (Line 6 and 7) and abstractions (Line 8); otherwise it leaves the term unchanged (Line 9). In Line 2 we need to transform the cterm into a term in order to be able to pattern-match the term. To see this conversion in action, consider the following example:

```
let  val\ conv = all\_true1\_conv\ @\{context\} \\ val\ ctrm = @\{cterm\ "distinct\ [1,\ x]\ \longrightarrow\ True\ \land\ 1 \neq x"\} \\ in \\ conv\ ctrm \\ end \\ > \ distinct\ [1,\ x]\ \longrightarrow\ True\ \land\ 1 \neq x \equiv \ distinct\ [1,\ x]\ \longrightarrow\ 1 \neq x \\
```

To see how much control you have about rewriting by using conversions, let us make the task a bit more complicated by rewriting according to the rule <code>true\_conj1</code>, but only in the first arguments of <code>Ifs</code>. Then the conversion should be as follows.

```
fun if_true1_conv ctxt ctrm =
   case Thm.term_of ctrm of
   Const (@{const_name If}, _) $ _ =>
        Conv.arg_conv (all_true1_conv ctxt) ctrm
| _ $ _ => Conv.comb_conv (if_true1_conv ctxt) ctrm
| Abs _ => Conv.abs_conv (fn (_, ctxt) => if_true1_conv ctxt) ctxt ctrm
| _ => Conv.all_conv ctrm
```

Here is an example for this conversion:

So far we only applied conversions to *cterms*. Conversions can, however, also work on theorems using the function *fconv\_rule*. As an example, consider the conversion *all\_true1\_conv* and the lemma:

```
lemma foo_test: "P \lor (True \land \neg P)" by simp
```

Using the conversion all\_true1\_conv you can transform this theorem into a new theorem as follows

```
let
  val conv = Conv.fconv_rule (all_true1_conv @{context})
  val thm = @{thm foo_test}
in
  conv thm
end
> ?P \/ \ ¬ ?P
```

Finally, conversions can also be turned into tactics and then applied to goal states. This can be done with the help of the function *CONVERSION*, and also some predefined conversion combinators that traverse a goal state. The combinators for the goal state are:

• params\_conv for converting under parameters (i.e. where goals are of the form  $\bigwedge x$ .  $P x \implies Q x$ )

4.6. CONVERSIONS 97

- prems\_conv for applying a conversion to all premises of a goal, and
- *concl\_conv* for applying a conversion to the conclusion of a goal.

Assume we want to apply all\_truel\_conv only in the conclusion of the goal, and if\_truel\_conv should only apply to the premises. Here is a tactic doing exactly that:

```
fun true1_tac ctxt =
   CONVERSION
    (Conv.params_conv ~1 (fn ctxt =>
          (Conv.prems_conv ~1 (if_true1_conv ctxt) then_conv
          Conv.concl_conv ~1 (all_true1_conv ctxt))) ctxt)
```

We call the conversions with the argument  $^{\sim}1$ . This is to analyse all parameters, premises and conclusions. If we call them with a non-negative number, say n, then these conversions will only be called on n premises (similar for parameters and conclusions). To test the tactic, consider the proof

#### lemma

```
"if True \land P then P else True \land False \Longrightarrow (if True \land Q then True \land Q else P) \longrightarrow True \land (True \land Q)" apply(tactic {* true1_tac @{context}} 1 *}) where the tactic yields the goal state goal (1 subgoal):

1. if P then P else True \land False \Longrightarrow (if Q then Q else P) \longrightarrow Q
```

As you can see, the premises are rewritten according to *if\_true1\_conv*, while the conclusion according to *all\_true1\_conv*.

To sum up this section, conversions are more general than the simplifier or simprocs, but you have to do more work yourself. Also conversions are often much less efficient than the simplifier. The advantage of conversions, however, that they provide much less room for non-termination.

**Exercise 4.6.1.** Write a tactic that does the same as the simproc in exercise 4.5.1, but is based on conversions. You can make the same assumptions as in 4.5.1.

**Exercise 4.6.2.** Compare your solutions of Exercises 4.5.1 and 4.6.1, and try to determine which way of rewriting such terms is faster. For this you might have to construct quite large terms. Also see Recipe A.3 for information about timing.

### Read More

See Pure/conv.ML for more information about conversion combinators. Some basic conversions are defined in Pure/thm.ML, Pure/drule.ML and Pure/meta\_simplifier.ML.

(FIXME: check whether Pattern.match\_rew and Pattern.rewrite\_term are of any use/efficient)

# 4.7 Declarations (TBD)

# 4.8 Structured Proofs (TBD)

```
TBD

lemma True
proof

{
    fix A B C
    assume r: "A & B ⇒ C"
    assume A B
    then have "A & B" ..
    then have C by (rule r)
}

{
    fix A B C
    assume r: "A & B ⇒ C"
    assume A B
    note conjI [OF this]
    note r [OF this]
}
```

```
val ctxt0 = 0{context};

val ctxt = ctxt0;

val (_, ctxt) = Variable.add_fixes ["A", "B", "C"] ctxt;

val ([r], ctxt) = Assumption.add_assumes [0{cprop "A \& B \implies C"}] ctxt

val (this, ctxt) = Assumption.add_assumes [0{cprop "A"}, 0{cprop "B"}]

ctxt;

val this = [0{thm conjI} OF this];

val this = r OF this;

val this = Assumption.export false ctxt ctxt0 this

val this = Variable.export ctxt ctxt0 [this]
```

# Chapter 5

# How to Write a Definitional Package

HOL is based on just a few primitive constants, like equality and implication, whose properties are described by axioms. All other concepts, such as inductive predicates, datatypes or recursive functions, are defined in terms of those primitives, and the desired properties, for example induction theorems or recursion equations, are derived from the definitions by a formal proof. Since it would be very tedious for a user to define inductive predicates or datatypes "by hand" just using the primitive operators of higher order logic, *definitional packages* have been implemented to automate such work. Thanks to those packages, the user can give a high-level specification, for example a list of introduction rules or constructors, and the package then does all the low-level definitions and proofs behind the scenes. In this chapter we explain how such a package can be implemented.

As the running example we have chosen a rather simple package for defining inductive predicates. To keep things really simple, we will not use the general Knaster-Tarski fixpoint theorem on complete lattices, which forms the basis of Isabelle/HOL's standard inductive definition package. Instead, we will describe a simpler *impredicative* (i.e. involving quantification on predicate variables) encoding of inductive predicates. Due to its simplicity, this package will necessarily have a reduced functionality. It does neither support introduction rules involving arbitrary monotonic operators, nor does it prove case analysis rules (also called inversion rules). Moreover, it only proves a weaker form of the induction principle for inductive predicates. But it illustrates the implementation pf a typical package in Isabelle.

## 5.1 Preliminaries

The user will just give a specification of inductive predicate(s) and expects from the package to produce a convenient reasoning infrastructure. This infrastructure needs to be derived from the definition that correspond to the specified predicate(s). Before we start with explaining all parts of the package, let us first give some examples showing how to define inductive predicates and then also how to generate a reason-

ing infrastructure for them. From the examples we will figure out a general method for defining inductive predicates. This is usually the first step in writing a package for Isabelle. The aim in this section is *not* to write proofs that are as beautiful as possible, but as close as possible to the ML-implementation we will develop in later sections.

We first consider the transitive closure of a relation *R*. The "pencil-and-paper" specification for the transitive closure is:

$$\frac{R \times y \qquad trcl R y z}{trcl R \times x}$$

As mentioned before, the package has to make an appropriate definition for trcl. Since an inductively defined predicate is the least predicate closed under a collection of introduction rules, the predicate  $trcl\ R\ x\ y$  can be defined so that it holds if and only if  $P\ x\ y$  holds for every predicate P closed under the rules above. This gives rise to the definition

Note we have to use the object implication  $\longrightarrow$  and object quantification  $\forall$  for stating this definition (there is no other way for definitions in HOL). However, the introduction rules and induction principles associated with the transitive closure should use the meta-connectives, since they simplify the reasoning for the user.

With this definition, the proof of the induction principle for trcl is almost immediate. It suffices to convert all the meta-level connectives in the lemma to object-level connectives using the proof method atomize (Line 4 below), expand the definition of trcl (Line 5 and 6), eliminate the universal quantifier contained in it (Line 7), and then solve the goal by assumption (Line 8).

```
lemma trcl_induct:
assumes "trcl R x y"
shows "(\( \lambda x \) P x x) \improx (\( \lambda x y z \). R x y \improx P y z \improx P x z) \improx P x y"
apply(atomize (full))
apply(cut_tac prems)
apply(unfold trcl_def)
apply(drule spec[where x=P])
apply(assumption)
done
```

The proofs for the introduction rules are slightly more complicated. For the first one, we need to prove the following lemma:

```
lemma trcl_base:
shows "trcl R x x"
apply(unfold trcl_def)
apply(rule allI impI)+
apply(drule spec)
apply(assumption)
done
```

We again unfold first the definition and apply introduction rules for  $\forall$  and  $\longrightarrow$  as often as possible (Lines 3 and 4). We then end up in the goal state:

101

```
goal (1 subgoal):

1. \bigwedge P. \llbracket \forall x. P x x; \forall x y z. R x y \longrightarrow P y z \longrightarrow P x z \rrbracket \Longrightarrow P x x
```

The two assumptions come from the definition of *trc1* and correspond to the introduction rules. Thus, all we have to do is to eliminate the universal quantifier in front of the first assumption (Line 5), and then solve the goal by *assumption* (Line 6).

Next we have to show that the second introduction rule also follows from the definition. Since this rule has premises, the proof is a bit more involved. After unfolding the definitions and applying the introduction rules for  $\forall$  and  $\longrightarrow$ 

```
lemma trcl_step:

shows "R x y \Longrightarrow trcl R y z \Longrightarrow trcl R x z"

apply (unfold trcl_def)

apply (rule allI impI)+
```

we obtain the goal state

```
goal (1 subgoal):

1. \bigwedge P. \llbracket R \times y;
\forall P. (\forall x. P \times x) \longrightarrow (\forall x y z. R \times y \longrightarrow P y z \longrightarrow P \times z) \longrightarrow P y z;
\forall x. P \times x; \ \forall x y z. R \times y \longrightarrow P y z \longrightarrow P \times z \rrbracket
\Longrightarrow P \times z
```

To see better where we are, let us explicitly name the assumptions by starting a subproof.

The assumptions p1 and p2 correspond to the premises of the second introduction rule (unfolded); the assumptions r1 and r2 come from the definition of trc1. We apply r2 to the goal  $P \times z$ . In order for this assumption to be applicable as a rule, we have to eliminate the universal quantifier and turn the object-level implications into meta-level ones. This can be accomplished using the  $rule\_format$  attribute. So we continue the proof with:

```
apply (rule r2[rule_format])
```

This gives us two new subgoals

```
goal (2 subgoals):
  1. R x ?y
  2. P ?y z
```

which can be solved using assumptions p1 and p2. The latter involves a quantifier and implications that have to be eliminated before it can be applied. To avoid potential problems with higher-order unification, we explicitly instantiate the quantifier to P and also match explicitly the implications with p1 and p2. This gives the proof:

```
apply(rule p1)
apply(rule p2[THEN spec[where x=P], THEN mp, THEN mp, OF r1, OF r2])
done
qed
```

Now we are done. It might be surprising that we are not using the automatic tactics available in Isabelle/HOL for proving this lemmas. After all blast would easily dispense of it.

```
lemma trcl_step_blast:
shows "R x y \implies trcl R y z \implies trcl R x z"
apply(unfold trcl_def)
apply(blast)
done
```

Experience has shown that it is generally a bad idea to rely heavily on blast, auto and the like in automated proofs. The reason is that you do not have precise control over them (the user can, for example, declare new intro- or simplification rules that can throw automatic tactics off course) and also it is very hard to debug proofs involving automatic tactics whenever something goes wrong. Therefore if possible, automatic tactics in packages should be avoided or be constrained sufficiently.

The method of defining inductive predicates by impredicative quantification also generalises to mutually inductive predicates. The next example defines the predicates *even* and *odd* given by

$$\frac{\text{odd n}}{\text{even 0}} \quad \frac{\text{odd n}}{\text{even (Suc n)}} \quad \frac{\text{even n}}{\text{odd (Suc n)}}$$

Since the predicates *even* and *odd* are mutually inductive, each corresponding definition must quantify over both predicates (we name them below *P* and *Q*).

For proving the induction principles, we use exactly the same technique as in the transitive closure example, namely:

```
lemma even_induct: assumes "even n" shows "P 0 \Longrightarrow (\bigwedgem. Q m \Longrightarrow P (Suc m)) \Longrightarrow (\bigwedgem. P m \Longrightarrow Q (Suc m)) \Longrightarrow P n" apply(atomize (full)) apply(cut_tac prems) apply(unfold even_def)
```

5.1. PRELIMINARIES 103

```
apply(drule spec[where x=P])
apply(drule spec[where x=Q])
apply(assumption)
done
```

The only difference with the proof  $trcl\_induct$  is that we have to instantiate here two universal quantifiers. We omit the other induction principle that has  $even\ n$  as premise and  $Q\ n$  as conclusion. The proofs of the introduction rules are also very similar to the ones in the trcl-example. We only show the proof of the second introduction rule.

```
lemma evenS:
   shows "odd m \Longrightarrow even (Suc m)"
3 apply (unfold odd_def even_def)
   apply (rule allI impI)+
   proof -
      case (goal1 P Q)
      have p1: "\forall P Q. P O \longrightarrow (\forall m. Q m \longrightarrow P (Suc m))
                                       \longrightarrow (\forall m. P m \longrightarrow Q (Suc m)) \longrightarrow Q m" by fact
      have r1: "P 0" by fact
      have r2: "\forall m. Q m \longrightarrow P (Suc m)" by fact
10
      have r3: "\forall m. P m \longrightarrow Q (Suc m)" by fact
11
      show "P (Suc m)"
        apply(rule r2[rule_format])
13
         apply(rule p1[THEN spec[where x=P], THEN spec[where x=Q],
14
                            THEN mp, THEN mp, THEN mp, OF r1, OF r2, OF r3])
15
         done
16
   qed
17
```

The interesting lines are 7 to 15. Again, the assumptions fall into two categories: p1 corresponds to the premise of the introduction rule; r1 to r3 come from the definition of even. In Line 13, we apply the assumption r2 (since we prove the second introduction rule). In Lines 14 and 15 we apply assumption p1 (if the second introduction rule had more premises we have to do that for all of them). In order for this assumption to be applicable, the quantifiers need to be instantiated and then also the implications need to be resolved with the other rules.

Next we define the accessible part of a relation R given by the single rule:

$$\frac{\bigwedge y. \ R \ y \ x \implies \text{accpart } R \ y}{\text{accpart } R \ x}$$

The interesting point of this definition is that it contains a quantification that ranges only over the premise and the premise has also a precondition. The definition of accpart is:

```
definition "accpart \equiv \lambda R x. \forall P. (\forall x. (\forall y. R y x \longrightarrow P y) \longrightarrow P x) \longrightarrow P x"
```

The proof of the induction principle is again straightforward and omitted. Proving the introduction rule is a little more complicated, because the quantifier and the implication in the premise. The proof is as follows.

```
lemma accpartI:
shows "( \land y . R y x \implies accpart R y ) \implies accpart R x"
```

```
apply (unfold accpart_def)
   apply (rule allI impI)+
   proof -
      case (goal1 P)
      have p1: "\bigwedge y. R y x \Longrightarrow
                             (\forall P. \ (\forall x. \ (\forall y. \ R \ y \ x \longrightarrow P \ y) \longrightarrow P \ x) \longrightarrow P \ y)" by fact
8
      have r1: "\forall x. (\forall y. R y x \longrightarrow P y) \longrightarrow P x" by fact
9
      show "P x"
10
         apply(rule r1[rule_format])
11
         proof -
12
            case (goal1 y)
13
            have r1_prem: "R y x" by fact
14
            show "P y"
15
               apply(rule p1[OF r1_prem, THEN spec[where x=P], THEN mp, OF r1])
16
            done
17
      qed
18
19
    qed
```

As you can see, there are now two subproofs. The assumptions fall as usual into two categories (Lines 7 to 9). In Line 11, applying the assumption r1 generates a goal state with the new local assumption R y x, named  $r1\_prem$  in the second subproof (Line 14). This local assumption is used to solve the goal P y with the help of assumption p1.

**Exercise 5.1.1.** *Give the definition for the freshness predicate for lambda-terms. The rules for this predicate are:* 

$$\frac{a \neq b}{\text{fresh a (Var b)}} \frac{\text{fresh a t}}{\text{fresh a (App t s)}}$$

$$\frac{a \neq b}{\text{fresh a (Lam a t)}} \frac{a \neq b}{\text{fresh a t}} \frac{\text{fresh a t}}{\text{fresh a (Lam b t)}}$$

From the definition derive the induction principle and the introduction rules.

The point of all these examples is to get a feeling what the automatic proofs should do in order to solve all inductive definitions we throw at them. This is usually the first step in writing a package. We next explain the parsing and typing part of the package.

## 5.2 Parsing and Typing the Specification

To be able to write down the specifications of inductive predicates, we have to introduce a new command (see Section 3.7). As the keyword for the new command we chose **simple\_inductive**. Examples of specifications from the previous section are shown in Figure 5.1. The syntax used in these examples more or less translates directly into the parser:

```
simple_inductive
  trcl :: "('a \Rightarrow 'a \Rightarrow bool) \Rightarrow 'a \Rightarrow 'a \Rightarrow bool"
where
  base: "trcl R x x"
 | \  \, \text{step: "trcl R x y} \implies \text{R y z} \implies \text{trcl R x z"} 
simple_inductive
  even and odd
where
  even0: "even 0"
| evenS: "odd n \Longrightarrow even (Suc n)"
/ oddS: "even n \Longrightarrow odd (Suc n)"
simple_inductive
  accpart :: "('a \Rightarrow 'a \Rightarrow bool) \Rightarrow 'a \Rightarrow bool"
where
  accpartI: "(\landy. R y x \Longrightarrow accpart R y) \Longrightarrow accpart R x"
simple_inductive
  fresh :: "string \Rightarrow trm \Rightarrow bool"
  fresh_var: "a\neqb \Longrightarrow fresh a (Var b)"
| fresh_app: "[fresh a t; fresh a s] \Longrightarrow fresh a (App t s)"
| fresh_lam1: "fresh a (Lam a t)"
| fresh_lam2: "\llbracket a \neq b; fresh a t\rrbracket \implies fresh a (Lam b t)"
```

Figure 5.1: Specification given by the user for the inductive predicates trcl, even and odd, accpart and fresh.

which we explained in Section 3.6. There is no code included for parsing the keyword and what is called a *target*. The latter can be given optionally after the keyword. The target is an "advanced" feature which we will inherit for "free" from the infrastructure on which we shall build the package. The target stands for a locale and allows us to specify

```
locale rel =
  fixes R :: "'a ⇒ 'a ⇒ bool"

and then define the transitive closure and the accessible part of this locale as follows:
simple_inductive (in rel)
  trcl'
```

where

```
base: "trcl' x x"

| step: "trcl' x y \Longrightarrow R y z \Longrightarrow trcl' x z"

simple_inductive (in rel)
accpart'
where
accpartI: "(\bigwedgey. R y x \Longrightarrow accpart' y) \Longrightarrow accpart' x"
```

Note that in these definitions the parameter *R*, standing for the relation, is left implicit. For the moment we will ignore this kind of implicit parameters and rely on the fact that the infrastructure will deal with them. Later, however, we will come back to them.

If we feed into the parser the string that corresponds to our definition of even and odd

```
let
  val input = filtered_input
      ("even and odd " ^
       "where " ^
       " even0[intro]: \"even 0\" " ^
      "| evenS[intro]: \"odd n \Longrightarrow even (Suc n)\" " \hat{}
       "| oddS[intro]: \"even n \Longrightarrow odd (Suc n)\" ")
in
  parse spec_parser input
end
> (([(even, NONE, NoSyn), (odd, NONE, NoSyn)],
>
        [((even0,...), "^E^Ftoken^Eeven 0^E^F^E"),
         ((evenS,...), "\^E\^Ftoken\^Eodd n \Longrightarrow even (Suc n)\^E\^F\^E"),
>
         ((oddS,...), "\^E\^Ftoken\^Eeven n \implies odd (Suc n)\^E\^F\^E")]), [])
```

then we get back the specifications of the predicates (with type and syntax annotations), and specifications of the introduction rules. This is all the information we need for calling the package and setting up the keyword. The latter is done in Lines 5 to 7 in the code below.

We call <code>local\_theory</code> with the kind-indicator <code>thy\_decl</code> since the package does not need to open up any proof (see Section 3.7). The auxiliary function <code>specification</code> in Lines 1 to 3 gathers the information from the parser to be processed further by the function <code>add\_inductive\_cmd</code>, which we describe below.

Note that the predicates when they come out of the parser are just some "naked" strings: they have no type yet (even if we annotate them with types) and they are

also not defined constants yet (which the predicates eventually will be). Also the introduction rules are just strings. What we have to do first is to transform the parser's output into some internal datastructures that can be processed further. For this we can use the function  $read\_spec$ . This function takes some strings (with possible typing annotations) and some rule specifications, and attempts to find a typing according to the given type constraints given by the user and the type constraints by the "ambient" theory. It returns the type for the predicates and also returns typed terms for the introduction rules. So at the heart of the function  $add\_inductive\_cmd$  is a call to  $read\_spec$ .

Once we have the input data as some internal datastructure, we call the function add\_inductive. This function does the heavy duty lifting in the package: it generates definitions for the predicates and derives from them corresponding induction principles and introduction rules. The description of this function will span over the next two sections.

## 5.3 The Code in a Nutshell

The inductive package will generate the reasoning infrastructure for mutually recursive predicates, say  $pred_1...pred_n$ . In what follows we will have the convention that various, possibly empty collections of "things" (lists, terms, nested implications and so on) are indicated either by adding an "s" or by adding a superscript "\*". The shorthand for the predicates will therefore be preds or  $pred^*$ . In the case of the predicates there must be, of course, at least a single one in order to obtain a meaningful definition.

The input for the inductive package will be some *preds* with possible typing and syntax annotations, and also some introduction rules. We call below the introduction rules short as *rules*. Borrowing some idealised Isabelle notation, one such *rule* is assumed to be of the form

$$\textit{rule} ::= \bigwedge \textit{xs}. \underbrace{\textit{As}}_{\text{non-recursive premises}} \Longrightarrow \underbrace{(\bigwedge \textit{ys}. \; \textit{Bs} \; \Longrightarrow \; \textit{pred} \; \textit{ss})^*}_{\text{recursive premises}} \Longrightarrow \textit{pred} \; \textit{ts}$$

For the purposes here, we will assume the rules have this format and omit any code that actually tests this. Therefore "things" can go horribly wrong, if the rules are not of this form. The As and Bs in a rule stand for formulae not involving the inductive predicates preds; the instances pred ss and pred ts can stand for different predicates, like pred<sub>1</sub> ss and pred<sub>2</sub> ts, in case mutual recursive predicates are defined; the terms ss and ts are the arguments of these predicates. Every formula

left of " $\Longrightarrow$  pred ts" is a premise of the rule. The outermost quantified variables xs are usually omitted in the user's input. The quantification for the variables ys is local with respect to one recursive premise and must be given. Some examples of rules are

```
a \neq b \implies fresh a (Var b)
```

which has only a single non-recursive premise, whereas

```
odd n \implies even (Suc n)
```

has a single recursive premise; the rule

```
(\bigwedge y. R y x \Longrightarrow accpart R y) \Longrightarrow accpart R x
```

has a single recursive premise that has a precondition. As is custom all rules are stated without the leading meta-quantification  $\bigwedge xs$ .

The output of the inductive package will be definitions for the predicates, induction principles and introduction rules. For the definitions we need to have the *rules* in a form where the meta-quantifiers and meta-implications are replaced by their object logic equivalents. Therefore an *orule* is of the form

```
orule ::= \forall xs. As \longrightarrow (\forall ys. Bs \longrightarrow pred ss)* \longrightarrow pred ts
```

A definition for the predicate *pred* has then the form

```
def ::= pred \equiv \lambda zs. \forall preds. orules \longrightarrow pred zs
```

The induction principles for every predicate pred are of the form

```
ind ::= pred ?zs \Longrightarrow rules[preds := ?Ps] \Longrightarrow ?P ?zs
```

where in the rules-part every pred is replaced by a fresh meta-variable ?P.

In order to derive an induction principle for the predicate *pred*, we first transform *ind* into the object logic and fix the meta-variables. Hence we have to prove a formula of the form

```
pred zs \longrightarrow orules[preds := Ps] \longrightarrow P zs
```

If we assume pred zs and unfold its definition, then we have an assumption

```
\forall \, {\tt preds.} \, \, {\tt orules} \, \longrightarrow \, {\tt pred} \, \, {\tt zs}
```

and must prove the goal

```
orules[preds := Ps] \longrightarrow P zs
```

This can be done by instantiating the  $\forall$  *preds*-quantification with the *Ps*. Then we are done since we are left with a simple identity.

Although the user declares the introduction rules rules, they must also be derived from the defs. These derivations are a bit involved. Assuming we want to prove the introduction rule

$$\bigwedge$$
xs. As  $\Longrightarrow$   $(\bigwedge$ ys. Bs  $\Longrightarrow$  pred ss)\*  $\Longrightarrow$  pred ts

then we have assumptions of the form

(i) As (ii) 
$$(\slash ys. Bs \implies pred ss)^*$$

and must show the goal

pred ts

If we now unfold the definitions for the preds, we have assumptions

```
(i) As (ii) (\bigwedge ys. Bs \Longrightarrow \forall preds. orules \longrightarrow pred ss)* (iii) orules
```

and need to show

pred ts

In the last step we removed some quantifiers and moved the precondition *orules* into the assumption. The *orules* stand for all introduction rules that are given by the user. We apply the *orule* that corresponds to introduction rule we are proving. After transforming the object connectives into meta-connectives, this introduction rule must necessarily be of the form

As 
$$\Longrightarrow$$
  $(\bigwedge ys. Bs \Longrightarrow pred ss)^* \Longrightarrow pred ts$ 

When we apply this rule we end up in the goal state where we have to prove goals of the form

```
(a) As (b) (\bigwedge ys. Bs \implies pred ss)^*
```

We can immediately discharge the goals As using the assumptions in (i). The goals in (b) can be discharged as follows: we assume the Bs and prove pred ss. For this we resolve the Bs with the assumptions in (ii). This gives us the assumptions

```
(\forall \, \mathsf{preds.} \, \, \mathsf{orules} \, \longrightarrow \, \mathsf{pred} \, \, \mathsf{ss})^*
```

```
(* even-odd example *)
val eo_defs = [@{thm even_def}, @{thm odd_def}]
val eo_rules =
   [@{prop "even 0"},
    \mathbb{Q}\{\text{prop } " \land n. \text{ odd } n \implies \text{even } (\text{Suc } n)"\},
    \mathbb{Q}\{\text{prop } " \mid n. \text{ even } n \implies \text{odd } (\text{Suc } n) "\}]
val eo_orules =
   [@{prop "even 0"},
    \mathbb{Q}\{\text{prop } \forall n. \text{ odd } n \longrightarrow \text{even } (\text{Suc } n) \},
    \mathbb{Q}\{\text{prop } \forall n. \text{ even } n \longrightarrow \text{odd } (\text{Suc } n) \}
val eo_preds = [@{term "even::nat ⇒ bool"}, @{term "odd::nat ⇒ bool"}]
val eo_prednames = [@{binding "even"}, @{binding "odd"}]
val eo_mxs = [NoSyn, NoSyn]
val eo_arg_tyss = [[@{typ "nat"}], [@{typ "nat"}]]
val e_pred = @{term "even::nat ⇒ bool"}
val e_arg_tys = [0{typ "nat"}]
(* freshness example *)
val fresh_rules =
   [\emptyset{prop "\landa b. a \neq b \Longrightarrow fresh a (\forallar b)"},
    \mathbb{Q}\{\text{prop } \text{"} \land \text{a s t. fresh a t} \implies \text{fresh a s} \implies \text{fresh a (App t s)"}\},
    O{prop " \land a t. fresh a (Lam a t)"},
    \mathbb{Q}\{\text{prop } \text{"} \land \text{a b t. a} \neq \text{b} \implies \text{fresh a t} \implies \text{fresh a (Lam b t)"}\}
val fresh_orules =
   [@\{prop \ "\forall a \ b. \ a \neq b \longrightarrow fresh \ a \ (Var \ b)"\},
     @\{ prop \ "\forall \ a \ s \ t. \ fresh \ a \ t \ \longrightarrow \ fresh \ a \ s \ \longrightarrow \ fresh \ a \ (App \ t \ s) "\}, 
    Q\{prop \ "\forall a \ t. \ fresh \ a \ (Lam \ a \ t)"\},
     @\{prop \ "\forall \ a \ b \ t. \ a \neq b \ \longrightarrow \ fresh \ a \ t \ \longrightarrow \ fresh \ a \ (Lam \ b \ t)"\}] 
val fresh_pred = 0{\text{term "fresh::string}} \Rightarrow \text{trm} \Rightarrow \text{bool"}}
val fresh_arg_tys = [@{typ "string"}, @{typ "trm"}]
(* accessible-part example *)
val acc_rules =
        [\emptyset\{prop \ "\ R \ x. \ (\ y. \ R \ y \ x \implies accpart \ R \ y) \implies accpart \ R \ x"\}]
val acc_pred = 0{\text{term "accpart}::('a \Rightarrow 'a \Rightarrow bool)} \Rightarrow 'a \Rightarrow bool"}
```

Figure 5.2: Shorthands for the inductive predicates <code>even/odd</code>, <code>fresh</code> and <code>accpart</code>. The names of these shorthands follow the convention <code>rules</code>, <code>orules</code>, <code>preds</code> and so on. The purpose of these shorthands is to simplify the construction of testcases in Section 5.4.

Instantiating the universal quantifiers and then resolving with the assumptions in (iii) gives us pred ss, which is the goal we are after. This completes the proof for introduction rules.

What remains is to implement in Isabelle the reasoning outlined in this section. We will describe the code in the next section. For building testcases, we use the shorthands for even/odd, fresh and accpart defined in Figure 5.2.

## 5.4 The Gory Details

As mentioned before the code falls roughly into three parts: the code that deals with the definitions, with the induction principles and with the introduction rules. In addition there are some administrative functions that string everything together.

## **Definitions**

We first have to produce for each predicate the user specifies an appropriate definition, whose general form is

```
pred \equiv \lambda zs. \forall preds. orules \longrightarrow pred zs
```

and then "register" the definition inside a local theory. To do the latter, we use the following wrapper for the function *define*. The wrapper takes a predicate name, a syntax annotation and a term representing the right-hand side of the definition.

```
fun make_defn ((predname, mx), trm) lthy =
let
val arg = ((predname, mx), (Attrib.empty_binding, trm))
val ((_, (_, thm)), lthy') = LocalTheory.define Thm.internalK arg lthy
in
(thm, lthy')
end
```

It returns the definition (as a theorem) and the local theory in which the definition has been made. In Line 4, <code>internalK</code> is a flag attached to the theorem (other possibile flags are <code>definitionK</code> and <code>axiomK</code>). These flags just classify theorems and have no significant meaning, except for tools that, for example, find theorems in the theorem database. We also use <code>empty\_binding</code> in Line 3, since the definitions for our inductive predicates are not meant to be seen by the user and therefore do not need to have any theorem attributes. A testcase for this function is

```
local_setup {* fn lthy =>
let
  val arg = ((@{binding "My_True"}, NoSyn), @{term True})
```

<sup>&</sup>lt;sup>1</sup>FIXME: put in the section about theorems.

```
val (def, lthy') = make_defn arg lthy
in
  tracing (string_of_thm_no_vars lthy' def); lthy'
end *}
```

which introduces the definition  $My\_True \equiv True$  and then prints it out. Since we are testing the function inside **local\_setup**, i.e., make actual changes to the ambient theory, we can query the definition with the usual command **thm**:

```
thm "My_True_def"
> My_True = True
```

The next two functions construct the right-hand sides of the definitions, which are terms whose general form is:

```
\lambda zs. \ \forall \ preds. \ orules \longrightarrow pred \ zs
```

When constructing these terms, the variables zs need to be chosen so that they do not occur in the orules and also be distinct from the preds.

The first function, named defn\_aux, constructs the term for one particular predicate (the argument pred in the code below). The number of arguments of this predicate is determined by the number of argument types given in arg\_tys. The other arguments of the function are the orules and all the preds.

```
fun defn_aux lthy orules preds (pred, arg_tys) =
1
2
     fun mk_all x P = HOLogic.all_const (fastype_of x) $ lambda x P
3
     val fresh_args =
5
           arg_tys
6
           |> map (pair "z")
           /> Variable.variant_frees lthy (preds @ orules)
8
           /> map Free
9
  in
10
     list_comb (pred, fresh_args)
11
     |> fold_rev (curry HOLogic.mk_imp) orules
12
     /> fold_rev mk_all preds
13
     /> fold_rev lambda fresh_args
14
15
```

The function  $mk_all$  in Line 3 is just a helper function for constructing universal quantifications. The code in Lines 5 to 9 produces the fresh zs. For this it pairs every argument type with the string z (Line 7); then generates variants for all these strings so that they are unique w.r.t. to the predicates and crules (Line 8); in Line 9 it generates the corresponding variable terms for the unique strings.

The unique variables are applied to the predicate in Line 11 using the function <code>list\_comb</code>; then the <code>orules</code> are prefixed (Line 12); in Line 13 we quantify over all predicates; and in line 14 we just abstract over all the <code>zs</code>, i.e., the fresh arguments of the predicate. A testcase for this function is

```
local_setup {* fn lthy =>
let
  val def = defn_aux lthy eo_orules eo_preds (e_pred, e_arg_tys)
in
  tracing (Syntax.string_of_term lthy def); lthy
end *}
```

where we use the shorthands defined in Figure 5.2. The testcase calls  $defn_{aux}$  for the predicate even and prints out the generated definition. So we obtain as printout

```
\lambda z. \forall even odd. (even 0) \longrightarrow (\forall n. odd n \longrightarrow even (Suc n)) \longrightarrow even z
```

If we try out the function with the rules for freshness

we obtain

```
\lambda z za.

\forall fresh.

(\forall a b. a \neq b \longrightarrow fresh a (Var b)) \longrightarrow

(\forall a s t. fresh a t \longrightarrow fresh a s \longrightarrow fresh a (App t s)) \longrightarrow

(\forall a t. fresh a (Lam a t)) \longrightarrow

(\forall a b t. a \neq b \longrightarrow fresh a t \longrightarrow fresh a (Lam b t)) \longrightarrow fresh z za
```

The second function, named defns, has to iterate the function defn\_aux over all predicates. The argument preds is again the list of predicates as terms; the argument prednames is the list of binding names of the predicates; mxs are the list of syntax, or mixfix, annotations for the predicates; arg\_tyss is the list of argument-type-lists.

```
fun defns rules preds prednames mxs arg_typss lthy =
let

val thy = ProofContext.theory_of lthy

val orules = map (ObjectLogic.atomize_term thy) rules
val defs = map (defn_aux lthy orules preds) (preds ~~ arg_typss)

in

fold_map make_defn (prednames ~~ mxs ~~ defs) lthy
end
```

The user will state the introduction rules using meta-implications and meta-quantifications. In Line 4, we transform these introduction rules into the object logic (since

definitions cannot be stated with meta-connectives). To do this transformation we have to obtain the theory behind the local theory using the function <code>theory\_of</code> (Line 3); with this theory we can use the function <code>atomize\_term</code> to make the transformation (Line 4). The call to <code>defn\_aux</code> in Line 5 produces all right-hand sides of the definitions. The actual definitions are then made in Line 7. The result of the function is a list of theorems and a local theory (the theorems are registered with the local theory). A testcase for this function is

```
local_setup {* fn lthy =>
let
  val (defs, lthy') =
    defns eo_rules eo_preds eo_prednames eo_mxs eo_arg_tyss lthy
in
  tracing (string_of_thms_no_vars lthy' defs); lthy
end *}
```

where we feed into the function all parameters corresponding to the even/odd example. The definitions we obtain are:

Note that in the testcase we return the local theory <code>lthy</code> (not the modified <code>lthy</code>). As a result the test case has no effect on the ambient theory. The reason is that if we introduce the definition again, we pollute the name space with two versions of <code>even</code> and <code>odd</code>. We want to avoid this here.

This completes the code for introducing the definitions. Next we deal with the induction principles.

## **Induction Principles**

Recall that the manual proof for the induction principle of even was:

```
lemma manual_ind_prin_even:
assumes prem: "even z"
shows "P 0 \Longrightarrow (\bigwedgem. Q m \Longrightarrow P (Suc m)) \Longrightarrow (\bigwedgem. P m \Longrightarrow Q (Suc m)) \Longrightarrow P z"
apply(atomize (full))
apply(cut_tac prem)
apply(unfold even_def)
apply(drule spec[where x=P])
apply(drule spec[where x=Q])
apply(assumption)
done
```

The code for automating such induction principles has to accomplish two tasks: constructing the induction principles from the given introduction rules and then automatically generating proofs for them using a tactic.

The tactic will use the following helper function for instantiating universal quantifiers.

```
fun inst_spec ctrm =
  Drule.instantiate' [SOME (ctyp_of_term ctrm)] [NONE, SOME ctrm] @{thm
spec}
```

This helper function instantiates the ?x in the theorem  $\forall x$ . ?P  $x \implies$  ?P ?x with a given *cterm*. We call this helper function in the following tactic..

```
fun inst_spec_tac ctrms =
   EVERY' (map (dtac o inst_spec) ctrms)
```

This tactic expects a list of cterms. It allows us in the proof below to instantiate the three quantifiers in the assumption.

#### lemma

```
fixes P::"nat \Rightarrow nat \Rightarrow nat \Rightarrow bool"

shows "\forall x y z. P x y z \Longrightarrow True"

apply (tactic {*

inst_spec_tac [0{cterm "a::nat"},0{cterm "b::nat"},0{cterm "c::nat"}] 1 *})

We obtain the goal state

goal (1 subgoal):

1. P a b c \Longrightarrow True
```

The complete tactic for proving the induction principles can now be implemented as follows:

We have to give it as arguments the definitions, the premise (a list of formulae) and the instantiations. The premise is even n in lemma manual\_ind\_prin\_even shown above; in our code it will always be a list consisting of a single formula. Compare this tactic with the manual proof for the lemma manual\_ind\_prin\_even: as you can see there is almost a one-to-one correspondence between the apply-script and the ind\_tac. We first rewrite the goal to use only object connectives (Line 2), "cut in" the premise (Line 3), unfold the definitions (Line 4), instantiate the assumptions of the goal (Line 5) and then conclude with assume\_tac.

Two testcases for this tactic are:

```
lemma automatic_ind_prin_even: assumes prem: "even z" shows "P 0 \Longrightarrow (\bigwedgem. Q m \Longrightarrow P (Suc m)) \Longrightarrow (\bigwedgem. P m \Longrightarrow Q (Suc m)) \Longrightarrow P z" by (tactic {* ind_tac eo_defs @{thms prem}}
```

[@{cterm "P::string \Rightarrm \Rightarrow bool"}] \*})

While the tactic for proving the induction principles is relatively simple, it will be a bit more work to construct the goals from the introduction rules the user provides. Therefore let us have a closer look at the first proved theorem:

 $(\land a \ b \ t. \ \llbracket a \neq b; \ P \ a \ t \rrbracket \implies P \ a \ (Lam \ b \ t)) \implies P \ z \ za"$ 

```
thm automatic_ind_prin_even > [even ?z; ?P 0; \mbox{$\backslash$m$}. ?Q m \Longrightarrow ?P (Suc m); \mbox{$\backslash$m$}. ?P m \Longrightarrow ?Q (Suc m)] \Longrightarrow ?P ?z
```

The variables z, P and Q are schematic variables (since they are not quantified in the lemma). These variables must be schematic, otherwise they cannot be instantiated by the user. To generate these schematic variables we use a common trick in Isabelle programming: we first declare them as *free*, *but fixed*, and then use the infrastructure to turn them into schematic variables.

In general we have to construct for each predicate pred a goal of the form

```
pred ?zs \implies rules[preds := ?Ps] \implies ?P ?zs
```

 $(\land a t. P a (Lam a t)) \Longrightarrow$ 

by (tactic {\* ind\_tac O{thms fresh\_def} O{thms prem}

where the predicates *preds* are replaced in *rules* by new distinct variables *?Ps*. We also need to generate fresh arguments *?zs* for the predicate *pred* and the *?P* in the conclusion.

We generate these goals in two steps. The first function, named <code>prove\_ind</code>, expects that the introduction rules are already appropriately substituted. The argument <code>srules</code> stands for these substituted rules; <code>cnewpreds</code> are the certified terms coresponding to the variables <code>?Ps</code>; <code>pred</code> is the predicate for which we prove the induction principle; <code>newpred</code> is its replacement and <code>arg\_tys</code> are the argument types of this predicate.

```
fun prove_ind lthy defs srules cnewpreds ((pred, newpred), arg_tys) =
1
2
     val zs = replicate (length arg_tys) "z"
3
     val (newargnames, lthy') = Variable.variant_fixes zs lthy;
     val newargs = map Free (newargnames ~~ arg_tys)
5
6
     val prem = HOLogic.mk_Trueprop (list_comb (pred, newargs))
7
     val goal = Logic.list_implies
8
            (srules, HOLogic.mk_Trueprop (list_comb (newpred, newargs)))
9
10
     Goal.prove lthy' [] [prem] goal
11
         (fn {prems, ...} => ind_tac defs prems cnewpreds)
12
     /> singleton (ProofContext.export lthy' lthy)
13
   end
14
```

In Line 3 we produce names zs for each type in the argument type list. Line 4 makes these names unique and declares them as free, but fixed, variables in the local theory 1thy. That means they are not schematic variables (yet). In Line 5 we construct the terms corresponding to these variables. The variables are applied to the predicate in Line 7 (this corresponds to the first premise  $pred\ zs$  of the induction principle). In Line 8 and 9, we first construct the term  $P\ zs$  and then add the (substituted) introduction rules as preconditions. In case that no introduction rules are given, the conclusion of this implication needs to be wrapped inside a Trueprop, otherwise the Isabelle's goal mechanism will fail.<sup>2</sup>

In Line 11 we set up the goal to be proved using the function *prove*; in the next line we call the tactic for proving the induction principle. As mentioned before, this tactic expects the definitions, the premise and the (certified) predicates with which the introduction rules have been substituted. The code in these two lines will return a theorem. However, it is a theorem proved inside the local theory <code>lthy</code>, where the variables <code>zs</code> are free, but fixed (see Line 4). By exporting this theorem from <code>lthy</code>, (which contains the <code>zs</code> as free variables) to <code>lthy</code> (which does not), we obtain the desired schematic variables <code>?zs</code>. A testcase for this function is

This prints out the theorem:

```
\llbracket \text{even ?z; P 0; } \land \text{n. } \textit{Q n} \implies \text{P (Suc n); } \land \text{n. P n} \implies \textit{Q (Suc n)} \rrbracket \implies \text{P ?z}
```

The export from 1thy' to 1thy in Line 13 above has correctly turned the free, but fixed, z into a schematic variable ?z; the variables P and Q are not yet schematic.

We still have to produce the new predicates with which the introduction rules are substituted and iterate <code>prove\_ind</code> over all predicates. This is what the second function, named <code>inds</code> does.

```
fun inds rules defs preds arg_tyss lthy =
let
val Ps = replicate (length preds) "P"
val (newprednames, lthy') = Variable.variant_fixes Ps lthy

val thy = ProofContext.theory_of lthy'
```

<sup>&</sup>lt;sup>2</sup>FIXME: check with Stefan...is this so?

```
7
     val tyss' = map (fn tys => tys ---> HOLogic.boolT) arg_tyss
8
     val newpreds = map Free (newprednames ~~ tyss')
9
     val cnewpreds = map (cterm_of thy) newpreds
10
     val srules = map (subst_free (preds ~~ newpreds)) rules
11
12
13
   in
14
     map (prove_ind lthy' defs srules cnewpreds)
           (preds ~~ newpreds ~~ arg_tyss)
15
             /> ProofContext.export 1thy' 1thy
16
   end
17
```

In Line 3, we generate a string "P" for each predicate. In Line 4, we use the same trick as in the previous function, that is making the Ps fresh and declaring them as free, but fixed, in the new local theory 1thy'. From the local theory we extract the ambient theory in Line 6. We need this theory in order to certify the new predicates. In Line 8, we construct the types of these new predicates using the given argument types. Next we turn them into terms and subsequently certify them (Line 9 and 10). We can now produce the substituted introduction rules (Line 11) using the function subst\_free. Line 14 and 15 just iterate the proofs for all predicates. From this we obtain a list of theorems. Finally we need to export the fixed variables Ps to obtain the schematic variables ?Ps (Line 16).

A testcase for this function is

```
local_setup {* fn lthy =>
let
  val ind_thms = inds eo_rules eo_defs eo_preds eo_arg_tyss lthy
in
  tracing (string_of_thms lthy ind_thms); lthy
end *}
```

which prints out

```
even ?z \Longrightarrow ?P1 0 \Longrightarrow (\lambda m. ?P1 m \Longrightarrow ?Pa1 (Suc m)) \Longrightarrow (\lambda m. ?P1 m \Longrightarrow ?Pa1 (Suc m)) \Longrightarrow ?P1 ?z, odd ?z \Longrightarrow ?P1 0 \Longrightarrow (\lambda m. ?Pa1 m \Longrightarrow ?P1 (Suc m)) \Longrightarrow (\lambda m. ?P1 m \Longrightarrow ?Pa1 (Suc m)) \Longrightarrow ?Pa1 ?z
```

Note that now both, the *?Ps* and the *?zs*, are schematic variables. The numbers attached to these variables have been introduced by the pretty-printer and are *not* important for the user.

This completes the code for the induction principles. The final peice of reasoning infrastructure we need are the introduction rules.

#### **Introduction Rules**

Constructing the goals for the introduction rules is easy: they are just the rules given by the user. However, their proofs are quite a bit more involved than the ones for the induction principles. To explain the general method, our running example will be the introduction rule

```
\landa b t. [a \neq b; fresh a t] \implies fresh a (Lam b t)
```

about freshness for lambdas. In order to ease somewhat our work here, we use the following two helper functions.

```
val all_elims = fold (fn ct => fn th => th RS inst_spec ct)
val imp_elims = fold (fn th => fn th' => [th', th] MRS @{thm mp})
```

To see what these functions do, let us suppose we have the following three theorems.

```
lemma all_elims_test:
fixes P::"nat \Rightarrow nat \Rightarrow nat \Rightarrow bool"
shows "\forall x y z. P x y z" sorry

lemma imp_elims_test:
shows "A \longrightarrow B \longrightarrow C" sorry

lemma imp_elims_test':
shows "A" "B" sorry
```

The function  $all_elims$  takes a list of (certified) terms and instantiates theorems of the form  $all_elims_test$ . For example we can instantiate the quantifiers in this theorem with a, b and c as follows:

```
let
  val ctrms = [@{cterm "a::nat"}, @{cterm "b::nat"}, @{cterm "c::nat"}]
  val new_thm = all_elims ctrms @{thm all_elims_test}
in
  tracing (string_of_thm_no_vars @{context} new_thm)
end
> Pabc
```

Note the difference with <code>inst\_spec\_tac</code> from Page 115: <code>inst\_spec\_tac</code> is a tactic which operates on a goal state; in contrast <code>all\_elims</code> operates on theorems. Similarly, the function <code>imp\_elims</code> eliminates preconditions from implications. For example we can eliminate the preconditions <code>A</code> and <code>B</code> from <code>imp\_elims\_test</code>:

```
let
  val res = imp_elims @{thms imp_elims_test'} @{thm imp_elims_test}
in
  tracing (string_of_thm_no_vars @{context} res)
end
> C
```

Now we set up the proof for the introduction rule as follows:

```
 \begin{array}{ll} \textbf{lemma fresh\_Lam:} \\ \textbf{shows "} \land \textbf{a b t. } \llbracket \textbf{a} \neq \textbf{b; fresh a t} \rrbracket \implies \textbf{fresh a (Lam b t)"} \\ \end{array}
```

The first step in the proof will be to expand the definitions of freshness and then introduce quantifiers and implications. For this we will use the tactic

```
fun expand_tac defs =

ObjectLogic.rulify_tac 1

THEN rewrite_goals_tac defs

THEN (REPEAT (resolve_tac [@{thm allI}, @{thm impI}] 1))
```

The function in Line 2 "rulifies" the lemma.<sup>3</sup> This will turn out to be important later on. Applying this tactic in our proof of *fresh\_Lem* 

```
apply(tactic {* expand_tac @{thms fresh_def} *})
gives us the goal state

goal (1 subgoal):
1. \( \) a b t fresh.

\[ [a \neq b; \)
\times fresh.

\[ (\forall a b. a \neq b \rightarrow fresh a (Var b)) \rightarrow

\[ (\forall a t s. fresh a t \rightarrow fresh a s \rightarrow fresh a (App t s)) \rightarrow

\[ (\forall a t. fresh a (Lam a t)) \rightarrow

\[ (\forall a b t. a \neq b \rightarrow fresh a t \rightarrow fresh a (Lam b t)) \rightarrow fresh a t;

\[ \forall a b. a \neq b \rightarrow fresh a s \rightarrow fresh a (App t s);

\[ \forall a t. fresh a (Lam a t);

\[ \forall a b. a \neq b \rightarrow fresh a t \rightarrow fresh a (Lam b t) \]

\[ \infty fresh a (Lam b t) \]

\[ \infty fresh a (Lam b t) \]
```

As you can see, there are parameters (namely a, b and t) which come from the introduction rule and parameters (in the case above only fresh) which come from the universal quantification in the definition fresh a (App t s). Similarly, there are assumptions that come from the premises of the rule (namely the first two) and assumptions from the definition of the predicate (assumption three to six). We need to treat these parameters and assumptions differently. In the code below we will therefore separate them into params1 and params2, respectively prems1 and prems2. To do this separation, it is best to open a subproof with the tactic SUBPROOF, since this tactic provides us with the parameters (as list of cterms) and the assumptions (as list of thms). The problem with SUBPROOF, however, is that it always expects us to completely discharge the goal (see Section 4.2). This is a bit inconvenient for our gradual explanation of the proof here. Therefore we use first the function FOCUS, which does a ame as SUBPROOF but does not require us to completely discharge the goal.

First we calculate the values for params1/2 and prems1/2 from params and prems, respectively. To better see what is going in our example, we will print out these values using the printing function in Figure 5.3. Since FOCUS will supply us the params and prems as lists, we can separate them using the function chop.

<sup>&</sup>lt;sup>3</sup>FIXME: explain this better

```
fun chop_print params1 params2 prems1 prems2 ctxt =
let
  val s = ["Params1 from the rule:", string_of_cterms ctxt params1]
     @ ["Params2 from the predicate:", string_of_cterms ctxt params2]
     @ ["Prems1 from the rule:"] @ (map (string_of_thm ctxt) prems1)
     @ ["Prems2 from the predicate:"] @ (map (string_of_thm ctxt) prems2)

in
     s |> cat_lines
     |> tracing
end
```

Figure 5.3: A helper function that prints out the parameters and premises that need to be treated differently.

```
fun chop_test_tac preds rules =
   Subgoal.FOCUS (fn {params, prems, context, ...} =>
   let
   val cparams = map snd params
   val (params1, params2) = chop (length cparams - length preds) cparams
   val (prems1, prems2) = chop (length prems - length rules) prems
   in
        chop_print params1 params2 prems1 prems2 context; all_tac
   end)
```

For the separation we can rely on the fact that Isabelle deterministically produces parameters and premises in a goal state. The last parameters that were introduced come from the quantifications in the definitions (see the tactic <code>expand\_tac</code>). Therefore we only have to subtract in Line 5 the number of predicates (in this case only 1) from the lenghts of all parameters. Similarly with the <code>prems</code> in line 6: the last premises in the goal state come from unfolding the definition of the predicate in the conclusion. So we can just subtract the number of rules from the number of all premises. To check our calculations we print them out in Line 8 using the function <code>chop\_print</code> from Figure 5.3 and then just do nothing, that is <code>all\_tac</code>. Applying this tactic in our example

```
apply(tactic {* chop_test_tac [fresh_pred] fresh_rules @{context} 1 *}) gives

Params1 from the rule:
a, b, t

Params2 from the predicate:
fresh

Prems1 from the rule:
a \neq b

\forall fresh.

(\forall a b. a \neq b \longrightarrow fresh a (\forall a b)) \longrightarrow

(\forall a t s. fresh a t \longrightarrow fresh a s \longrightarrow fresh a (\forall a p. t s)) \longrightarrow
```

```
 \begin{array}{l} (\forall \texttt{a} \texttt{ t. fresh a } (\texttt{Lam a t})) \longrightarrow \\ (\forall \texttt{a} \texttt{ b} \texttt{ t. a} \neq \texttt{b} \longrightarrow \texttt{fresh a t} \longrightarrow \texttt{fresh a } (\texttt{Lam b t})) \longrightarrow \texttt{fresh a t} \\ \textit{Prems2 from the predicate:} \\ \forall \texttt{a} \texttt{ b. a} \neq \texttt{b} \longrightarrow \texttt{fresh a } (\texttt{Var b}) \\ \forall \texttt{a} \texttt{ t. fresh a t} \longrightarrow \texttt{fresh a s} \longrightarrow \texttt{fresh a } (\texttt{App t s}) \\ \forall \texttt{a t. fresh a } (\texttt{Lam a t}) \\ \forall \texttt{a b t. a} \neq \texttt{b} \longrightarrow \texttt{fresh a t} \longrightarrow \texttt{fresh a } (\texttt{Lam b t}) \\ \end{array}
```

We now have to select from *prems2* the premise that corresponds to the introduction rule we prove, namely:

```
\forall a b t. a \neq b \longrightarrow fresh a t \longrightarrow fresh a (Lam a t)
```

To use this premise with rtac, we need to instantiate its quantifiers (with params1) and transform it into rule format (using rulify). So we can modify the code as follows:

```
fun apply_prem_tac i preds rules =
Subgoal.FOCUS (fn {params, prems, context, ...} =>
let
val cparams = map snd params
val (params1, params2) = chop (length cparams - length preds) cparams
val (prems1, prems2) = chop (length prems - length rules) prems
in
rtac (ObjectLogic.rulify (all_elims params1 (nth prems2 i))) 1
end)
```

The argument *i* corresponds to the number of the introduction we want to prove. We will later on let it range from 0 to the number of *rules* – 1. Below we apply this function with 3, since we are proving the fourth introduction rule.

```
apply(tactic {* apply_prem_tac 3 [fresh_pred] fresh_rules @{context} 1 *})
```

The goal state we obtain is:

```
1. ... \implies a \neq b
2. ... \implies fresh a t
```

As expected there are two subgoals, where the first comes from the non-recursive premise of the introduction rule and the second comes from the recursive one. The first goal can be solved immediately by *prems1*. The second needs more work. It can be solved with the other premise in *prems1*, namely

```
\forall fresh.

(\forall a b. a \neq b \longrightarrow fresh a (\forall a b. b)) \longrightarrow
(\forall a t s. fresh a t \longrightarrow fresh a s \longrightarrow fresh a (\forall a t. fresh a (\forall a b t. a \neq b \longrightarrow fresh a t \longrightarrow fresh a (\forall a b t. a \neq b \longrightarrow fresh a t \longrightarrow fresh a (\forall a b t. a \neq b \longrightarrow fresh a t
```

but we have to instantiate it appropriately. These instantiations come from params1 and prems2. We can determine whether we are in the simple or complicated case by checking whether the topmost connective is an  $\forall$ . The premises in the simple case cannot have such a quantification, since the first step of  $expand_tac$  was to "rulify" the lemma. The premise of the complicated case must have at least one  $\forall$  coming from the quantification over the preds. So we can implement the following function

which either applies the premise outright (the default case) or if it has an outermost universial quantification, instantiates it first with *params1* and then *prems1*. The following tactic will therefore prove the lemma completely.

```
fun prove_intro_tac i preds rules =
   SUBPROOF (fn {params, prems, ...} =>
   let
    val cparams = map snd params
   val (params1, params2) = chop (length cparams - length preds) cparams
   val (prems1, prems2) = chop (length prems - length rules) prems
   in
     rtac (ObjectLogic.rulify (all_elims params1 (nth prems2 i))) 1
     THEN EVERY1 (map (prepare_prem params2 prems2) prems1)
   end)
```

Note that the tactic is now SUBPROOF, not FOCUS anymore. The full proof of the introduction rule is as follows:

```
lemma fresh_Lam: shows "\( a \) b t. \[ a \neq b; fresh a t \] \Longrightarrow fresh a (Lam b t)" apply(tactic {* expand_tac 0{thms fresh_def} *}) apply(tactic {* prove_intro_tac 3 [fresh_pred] fresh_rules 0{context} 1 *}) done
```

Phew!...

Unfortunately, not everything is done yet. If you look closely at the general principle outlined for the introduction rules in Section 5.3, we have not yet dealt with the case where recursive premises have preconditions. The introduction rule of the accessible part is such a rule.

```
lemma accpartI: shows "\A x. (\A y. R y x \Longrightarrow accpart R y) \Longrightarrow accpart R x" apply(tactic {* expand_tac @{thms accpart_def} *}) apply(tactic {* chop_test_tac [acc_pred] acc_rules @{context} 1 *}) apply(tactic {* apply_prem_tac 0 [acc_pred] acc_rules @{context} 1 *})
```

Here chop\_test\_tac prints out the following values for params1/2 and prems1/2

```
Params1 from the rule: x
Params2 from the predicate: P
Prems1 from the rule: R ?y x \Longrightarrow \forall P. (\forall x. (\forall y. R y x \longrightarrow P y) \longrightarrow P x) \longrightarrow P ?y
Prems2 from the predicate: \forall x. (\forall y. R y x \longrightarrow P y) \longrightarrow P x
```

and after application of the introduction rule using apply\_prem\_tac, we are in the goal state

```
1. \bigwedge y. R y x \Longrightarrow P y
```

In order to make progress, we have to use the precondition R y x (in general there can be many of them). The best way to get a handle on these preconditions is to open up another subproof, since the preconditions will then be bound to prems. Therfore we modify the function  $prepare_prem$  as follows

```
fun prepare_prem params2 prems2 ctxt prem =
1
     SUBPROOF (fn {prems, ...} =>
2
3
       val prem' = prems MRS prem
4
5
     in
       rtac (case prop_of prem' of
6
              _ $ (Const (@{const_name All}, _) $ _) =>
7
                    prem' /> all_elims params2
8
                           /> imp_elims prems2
9
            | _ => prem') 1
10
     end) ctxt
11
```

In Line 4 we use the *prems* from the *SUBPROOF* and resolve them with *prem*. In the simple cases, that is where the *prem* comes from a non-recursive premise of the rule, *prems* will be just the empty list and the function *MRS* does nothing. Similarly, in the cases where the recursive premises of the rule do not have preconditions. In case there are preconditions, then Line 4 discharges them. After that we can proceed as before, i.e., check whether the outermost connective is  $\forall$ .

The function <code>prove\_intro\_tac</code> only needs to be changed so that it gives the context to <code>prepare\_prem</code> (Line 8). The modified version is below.

```
fun prove_intro_tac i preds rules =
1
     SUBPROOF (fn {params, prems, context, ...} =>
2
     let.
3
       val cparams = map snd params
       val (params1, params2) = chop (length cparams - length preds) cparams
5
       val (prems1, prems2) = chop (length prems - length rules) prems
6
7
       rtac (ObjectLogic.rulify (all_elims params1 (nth prems2 i))) 1
8
       THEN EVERY1 (map (prepare_prem params2 prems2 context) prems1)
9
     end)
10
```

With these two functions we can now also prove the introduction rule for the accessible part.

```
lemma accpartI: shows "\A x. (\A y. R y x \Longrightarrow accpart R y) \Longrightarrow accpart R x" apply(tactic {* expand_tac @{thms accpart_def} *}) apply(tactic {* prove_intro_tac 0 [acc_pred] acc_rules @{context} 1 *}) done
```

Finally we need two functions that string everything together. The first function is the tactic that performs the proofs.

```
fun intro_tac defs rules preds i ctxt =

EVERY1 [ObjectLogic.rulify_tac,

K (rewrite_goals_tac defs),

REPEAT o (resolve_tac [O{thm allI}, O{thm impI}]),

prove_intro_tac i preds rules ctxt]
```

Lines 2 to 4 in this tactic correspond to the function *expand\_tac*. Some testcases for this tactic are:

```
lemma even0_intro:
shows "even 0"
by (tactic {* intro_tac eo_defs eo_rules eo_preds 0 @{context} *})
lemma evenS_intro:
shows "\m. odd m \improx even (Suc m)"
by (tactic {* intro_tac eo_defs eo_rules eo_preds 1 @{context} *})
lemma fresh_App:
shows "\a t s. [fresh a t; fresh a s] \improx fresh a (App t s)"
by (tactic {*
intro_tac @{thms fresh_def} fresh_rules [fresh_pred] 1 @{context} *})
```

The second function sets up in Line 4 the goals to be proved (this is easy for the introduction rules since they are exactly the rules given by the user) and iterates <code>intro\_tac</code> over all introduction rules.

```
fun intros rules preds defs lthy =
let
fun intros_aux (i, goal) =
Goal.prove lthy [] [] goal
(fn {context, ...} => intro_tac defs rules preds i context)
in
map_index intros_aux rules
end
```

The iteration is done with the function map\_index since we need the introduction rule together with its number (counted from 0). This completes the code for the functions deriving the reasoning infrastructure. It remains to implement some administrative code that strings everything together.

#### **Administrative Functions**

We have produced various theorems (definitions, induction principles and introduction rules), but apart from the definitions, we have not yet registered them with the theorem database. This is what the functions note does.

For convenience, we use the following three wrappers this function:

```
fun note_many qname ((name, attrs), thms) =
  LocalTheory.note Thm.theoremK
      ((Binding.qualify false qname name, attrs), thms)

fun note_single1 qname ((name, attrs), thm) =
  note_many qname ((name, attrs), [thm])

fun note_single2 name attrs (qname, thm) =
  note_many (Binding.name_of qname) ((name, attrs), [thm])
```

The function that "holds everything together" is add\_inductive. Its arguments are the specification of the predicates pred\_specs and the introduction rules rule\_spec.

```
fun add_inductive pred_specs rule_specs lthy =
  let
2
     val mxs = map snd pred_specs
3
     val pred_specs' = map fst pred_specs
4
     val prednames = map fst pred_specs'
5
     val preds = map (fn (p, ty) => Free (Binding.name_of p, ty)) pred_specs'
     val tyss = map (binder_types o fastype_of) preds
8
     val (namesattrs, rules) = split_list rule_specs
10
     val (defs, lthy') = defns rules preds prednames mxs tyss lthy
11
     val ind_prins = inds rules defs preds tyss lthy'
12
     val intro_rules = intros rules preds defs lthy'
13
14
     val mut_name = space_implode "_" (map Binding.name_of prednames)
15
     val case_names = map (Binding.name_of o fst) namesattrs
16
17
     lthy' |> note_many mut_name ((@{binding "intros"}, []), intro_rules)
18
           ||>> note_many mut_name ((@{binding "inducts"}, []), ind_prins)
19
           ||>> fold_map (note_single1 mut_name) (namesattrs ~~ intro_rules)
20
           |/>> fold_map (note_single2 @{binding "induct"}
21
                  [Attrib.internal (K (RuleCases.case_names case_names)),
22
                  Attrib.internal (K (RuleCases.consumes 1)),
23
                  Attrib.internal (K (Induct.induct_pred ""))])
24
                 (prednames ~~ ind_prins)
25
           /> snd
26
   end
27
```

In Line 3 the function extracts the syntax annotations from the predicates. Lines 4 to 6 extract the names of the predicates and generate the variables terms (with

types) corresponding to the predicates. Line 7 produces the argument types for each predicate.

Line 9 extracts the introduction rules from the specifications and stores also in namesattrs the names and attributes the user may have attached to these rules.

Line 11 produces the definitions and also registers the definitions in the local theory lthy'. The next two lines produce the induction principles and the introduction rules (all of them as theorems). Both need the local theory lthy' in which the definitions have been registered.

Lines 15 produces the name that is used to register the introduction rules. It is costum to collect all introduction rules under <code>string.intros</code>, whereby <code>string</code> stands for the "\_"-separated list of predicate names (for example <code>even\_odd</code>. Also by custom, the case names in intuction proofs correspond to the names of the introduction rules. These are generated in Line 16.

Lines 18 and 19 now add to 1thy, all the introduction rules und induction principles under the name mut\_name.intros and mut\_name.inducts, respectively (see previous paragraph).

Line 20 add further every introduction rule under its own name (given by the user). Line 21 registers the induction principles. For this we have to use some specific attributes. The first <code>case\_names</code> corresponds to the case names that are used by Isar to reference the proof obligations in the induction. The second <code>consumes 1</code> indicates that the first premise of the induction principle (namely the predicate over which the induction proceeds) is eliminated.

This completes all the code and fits in with the "front end" described in Section 5.2.5

## 5.5 Extensions of the Package (TBD)

Things to include at the end:

- include the code for the parameters
- say something about add-inductive to return the rules
- say something about the two interfaces for calling packages

**Exercise 5.5.1.** In Section 5.3 we required that introduction rules must be of the form

rule ::= 
$$\land$$
xs. As  $\Longrightarrow$  ( $\land$ ys. Bs  $\Longrightarrow$  pred ss)\*  $\Longrightarrow$  pred ts

where the As and Bs can be any collection of formulae not containing the preds. This requirement is important, because if violated, the theory behind the inductive package does not work and also the proofs break. Write code that tests whether the introduction rules given by the user fit into the scheme described above. Hint: It is not important in which order the premises ar given; the As and ( $\slash$ ys. Bs  $\implies$  pred ss) premises can occur in any order.

<sup>&</sup>lt;sup>4</sup>FIXME: what happens if the user did not give any name.

<sup>&</sup>lt;sup>5</sup>FIXME: Describe Induct.induct\_pred. Why the mut-name? What does Binding.qualify do?

Exercise 5.5.2. If you define even and odd with the standard inductive package

# inductive even\_2 and odd\_2 where even0\_2: "even\_2 0" | evenS\_2: "odd\_2 m ⇒ even\_2 (Suc m)" | oddS\_2: "even\_2 m ⇒ odd\_2 (Suc m)"

you will see that the generated induction principle for even, (namely even\_2\_odd\_2.inducts has the additional assumptions odd\_2 m and even\_2 m in the recursive cases. These additional assumptions can sometimes make "life easier" in proofs. Since more assumptions can be made when proving properties, these induction principles are called strong inductions principles. However, it is the case that the "weak" induction principles imply the "strong" ones. Hint: Prove a property taking a pair (or tuple in case of more than one predicate) as argument: the property that you originally want to prove and the predicate(s) over which the induction proceeds. Write code that automates the derivation of the strong induction principles from the weak ones.

#### Read More

The standard inductive package is based on least fix-points. It allows more general introduction rules that can include any monotone operators and also provides a richer reasoning infrastructure. The code of this package can be found in HOL/Tools/inductive.ML.

## 5.6 Definitional Packages

Type declarations

```
Typedef.add_typedef false NONE (@{binding test},[],NoSyn)
  @{term "{1}::nat set"} NONE (simp_tac @{simpset} 1) @{theory}
```

# Appendix A

# **Recipes**

## Possible topics:

- translations/print translations; ProofContext.print\_syntax
- user space type systems (in the form that already exists)
- unification and typing algorithms (Pure/pattern.ML implements HOPU)
- useful datastructures: discrimination nets, association lists

## A.1 Useful Document Antiquotations

**Problem:** How to keep your ML-code inside a document synchronised with the actual code?

**Solution:** This can be achieved with document antiquotations.

Document antiquotations can be used for ensuring consistent type-setting of various entities in a document. They can also be used for sophisticated Lacking. If you type on the Isabelle level

## print\_antiquotations

you obtain a list of all currently available document antiquotations and their options. Below we will give the code for two additional document antiquotations both of which are intended to typeset ML-code. The crucial point of these document antiquotations is that they not just print the ML-code, but also check whether it compiles. This will provide a sanity check for the code and also allows you to keep documents in sync with other code, for example Isabelle.

We first describe the antiquotation ML\_checked with the syntax:

```
@{ML_checked "a_piece_of_code"}
```

The code is checked by sending the ML-expression "val  $\_$  = a\_piece\_of\_code" to the ML-compiler (i.e. the function  $ML_Context.eval_in$  in Line 4 below). The complete code of the document antiquotation is as follows:

```
fun ml_val code_txt = "val _ = " ^ code_txt

fun output_ml {context = ctxt, ...} code_txt =

(ML_Context.eval_in (SOME ctxt) false Position.none (ml_val code_txt);

ThyOutput.output (map Pretty.str (space_explode "\n" code_txt)))

val _ = ThyOutput.antiquotation "ML_checked" (Scan.lift Args.name) output_ml
```

The parser (Scan.lift Args.name) in Line 7 parses a string, in this case the code, and then calls the function output\_ml. As mentioned before, the parsed code is sent to the ML-compiler in Line 4 using the function ml\_val, which constructs the appropriate ML-expression, and using eval\_in, which calls the compiler. If the code is "approved" by the compiler, then the output function output in the next line pretty prints the code. This function expects that the code is a list of (pretty)strings where each string correspond to a line in the output. Therefore the use of (space\_explode "\n" txt) which produces such a list according to linebreaks. There are a number of options for antiquotations that are observed by the function output when printing the code (including [display] and [quotes]). The function antiquotation in Line 7 sets up the new document antiquotation.

#### Read More

For more information about options of document antiquotations see [Isar Ref. Man., Sec. 4.2]).

Since we used the argument *Position.none*, the compiler cannot give specific information about the line number, in case an error is detected. We can improve the code above slightly by writing

```
fun output_ml {context = ctxt, ...} (code_txt, pos) =
    (ML_Context.eval_in (SOME ctxt) false pos (ml_val code_txt);
    ThyOutput.output (map Pretty.str (space_explode "\n" code_txt)))

val _ = ThyOutput.antiquotation "ML_checked"
    (Scan.lift (OuterParse.position Args.name)) output_ml
```

where in Lines 1 and 2 the positional information is properly treated. The parser <code>OuterParse.position</code> encodes the positional information in the result.

We can now write <code>@{ML\_checked "2 + 3"}</code> in a document in order to obtain 2 + 3 and be sure that this code compiles until somebody changes the definition of addition.

The second document antiquotation we describe extends the first by a pattern that specifies what the result of the ML-code should be and checks the consistency of the actual result with the given pattern. For this we are going to implement the document antiquotation:

```
@{ML_resp "a_piece_of_code" "a_pattern"}
```

To add some convenience and also to deal with large outputs, the user can give a partial specification by using ellipses. For example (..., ...) for specifying a pair.

In order to check consistency between the pattern and the output of the code, we have to change the ML-expression that is sent to the compiler: in ML\_checked we sent the expression "val \_ = a\_piece\_of\_code" to the compiler; now the wildcard \_ must be be replaced by the given pattern. However, we have to remove all ellipses from it and replace them by "\_". The following function will do this:

```
fun ml_pat (code_txt, pat) =
let val pat' =
    implode (map (fn "..." => "_" | s => s) (Symbol.explode pat))
in
    "val " ^ pat' ^ " = " ^ code_txt
end
```

Next we add a response indicator to the result using:

```
fun add_resp pat = map (fn s => "> " ^ s) pat
```

The rest of the code of ML\_resp is:

```
fun output_ml_resp {context = ctxt, ...} ((code_txt, pat), pos) =
     (ML_Context.eval_in (SOME ctxt) false pos (ml_pat (code_txt, pat));
3
        val code_output = space_explode "\n" code_txt
4
        val resp_output = add_resp (space_explode "\n" pat)
5
6
        ThyOutput.output (map Pretty.str (code_output @ resp_output))
8
   val _ = ThyOutput.antiquotation "ML_resp"
10
             (Scan.lift (OuterParse.position (Args.name -- Args.name)))
11
                output_ml_resp
12
```

In comparison with ML\_checked, we only changed the line about the compiler (Line 2), the lines about the output (Lines 4 to 7) and the parser in the setup (Line 11). Now you can write

```
@{ML_resp [display] "true andalso false" "false"}
to obtain

true andalso false
> false

or
@{ML_resp [display] "let val i = 3 in (i * i, "foo") end" "(9, ...)"}
to obtain
```

```
let val i = 3 in (i * i, "foo") end > (9, ...)
```

In both cases, the check by the compiler ensures that code and result match. A limitation of this document antiquotation, however, is that the pattern can only be given for values that can be constructed. This excludes values that are abstract datatypes, like thms and cterms.

## A.2 Restricting the Runtime of a Function

**Problem:** Your tool should run only a specified amount of time.

**Solution:** In PolyML 5.2.1 and later, this can be achieved using the function timeLimit.

Assume you defined the Ackermann function on the ML-level.

```
fun ackermann (0, n) = n + 1

| ackermann (m, 0) = ackermann (m - 1, 1)

| ackermann (m, n) = ackermann (m - 1, ackermann (m, n - 1))
```

Now the call

```
ackermann (4, 12) > ...
```

takes a bit of time before it finishes. To avoid this, the call can be encapsulated in a time limit of five seconds. For this you have to write

```
TimeLimit.timeLimit (Time.fromSeconds 5) ackermann (4, 12)
  handle TimeLimit.TimeOut => ~1
> ~1
```

where TimeOut is the exception raised when the time limit is reached.

Note that timeLimit is only meaningful when you use PolyML 5.2.1 or later, because this version of PolyML has the infrastructure for multithreaded programming on which timeLimit relies.

#### Read More

The function timeLimit is defined in the structure TimeLimit which can be found in the file Pure/ML-Systems/multithreading\_polyml.ML.

## A.3 Measuring Time

**Problem:** You want to measure the running time of a tactic or function.

**Solution:** Time can be measured using the function *start\_timing* and *end\_timing*.

Suppose you defined the Ackermann function on the Isabelle level.

```
ackermann:: "(nat \times nat) \Rightarrow nat"
where
    "ackermann (0, n) = n + 1"
    | "ackermann (m, 0) = ackermann (m - 1, 1)"
    | "ackermann (m, n) = ackermann (m - 1, ackermann (m, n - 1))"
```

You can measure how long the simplifier takes to verify a datapoint of this function. The actual timing is done inside the wrapper function:

```
fun timing_wrapper tac st =
let
val t_start = start_timing ();
val res = tac st;
val t_end = end_timing t_start;
in
(writeln (#message t_end); res)
end
```

Note that this function, in addition to a tactic, also takes a state st as argument and applies this state to the tactic (Line 4). The reason is that tactics are lazy functions and you need to force them to run, otherwise the timing will be meaningless. The simplifier tactic, amongst others, can be forced to run by just applying the state to it. But "fully" lazy tactics, such as resolve\_tac, need even more "standing-on-oneshead" to force them to run.

The time between start and finish of the simplifier will be calculated as the end time minus the start time. An example of the wrapper is the proof

```
lemma "ackermann (3, 4) = 125"
apply(tactic {*
   timing_wrapper (simp_tac (@{simpset} addsimps @{thms "nat_number"}) 1) *})
done
```

where it returns something on the scale of 3 seconds. We chose to return this information as a string, but the timing information is also accessible in number format.

#### Read More

Basic functions regarding timing are defined in Pure/ML-Systems/polyml\_common.ML (for the PolyML compiler). Some more advanced functions are defined in Pure/General/output.ML.

## A.4 Configuration Options

**Problem:** You would like to enhance your tool with options that can be changed by the user without having to resort to the ML-level.

**Solution:** This can be achieved using configuration values.

Assume you want to control three values, say *bval* containing a boolean, *ival* containing an integer and *sval* containing a string. These values can be declared on the ML-level by

```
val (bval, setup_bval) = Attrib.config_bool "bval" false
val (ival, setup_ival) = Attrib.config_int "ival" 0
val (sval, setup_sval) = Attrib.config_string "sval" "some string"
```

where each value needs to be given a default. To enable these values, they need to be set up with

```
setup {* setup_bval *}
setup {* setup_ival *}
```

or on the ML-level with

```
setup_sval @{theory}
```

The user can now manipulate the values from within Isabelle with the command declare [[bval = true, ival = 3]]

On the ML-level these values can be retrieved using the function Config.get:

```
Config.get @{context} bval
> true
```

```
Config.get @{context} ival
> 3
```

The function Config.put manipulates the values. For example

```
Config.put sval "foo" @{context}; Config.get @{context} sval
> foo
```

The same can be achieved using the command **setup**.

```
setup {* Config.put_thy sval "bar" *}
```

Now the retrieval of this value yields:

```
Config.get @{context} sval
> "bar"
```

We can apply a function to a value using <code>Config.map</code>. For example incrementing <code>ival</code> can be done by:

```
let
  val ctxt' = Config.map ival (fn i => i + 1) @{context}
in
  Config.get ctxt' ival
end
> 4
```

#### Read More

For more information see Pure/Isar/attrib.ML and Pure/config.ML.

There are many good reasons to control parameters in this way. One is that no global reference is needed, which would cause many headaches with the multithreaded execution of Isabelle.

## A.5 Storing Data (TBD)

**Problem:** Your tool needs to manage data.

**Solution:** This can be achieved using a generic data slot.

Every generic data slot may keep data of any kind which is stored in the context.

```
structure Data = GenericDataFun
  ( type T = int Symtab.table
    val empty = Symtab.empty
    val extend = I
    fun merge _ = Symtab.merge (K true)
)

in
  val lookup = Symtab.lookup o Data.get
  fun update k v = Data.map (Symtab.update (k, v))
end
```

```
setup {* Context.theory_map (update "foo" 1) *}
```

```
lookup (Context.Proof @{context}) "foo"
> SOME 1
```

alternatives: TheoryDataFun, ProofDataFun Code: Pure/context.ML

## A.6 Executing an External Application (TBD)

Problem: You want to use an external application.

**Solution:** The function <code>system\_out</code> might be the right thing for you.

This function executes an external command as if printed in a shell. It returns the output of the program and its return value.

For example, consider running an ordinary shell commands:

```
system_out "echo Hello world!"
> ("Hello world!\n", 0)
```

Note that it works also fine with timeouts (see Recipe A.2 on Page 132), i.e. external applications are killed properly. For example, the following expression takes only approximately one second:

```
TimeLimit.timeLimit (Time.fromSeconds 1) system_out "sleep 30"
          handle TimeLimit.TimeOut => ("timeout", ~1)
> ("timeout", ~1)
```

The function <code>system\_out</code> can also be used for more reasonable applications, e.g. coupling external solvers with Isabelle. In that case, one has to make sure that Isabelle can find the particular executable. One way to ensure this is by adding a Bash-like variable binding into one of Isabelle's settings file (prefer the user settings file usually to be found at <code>\$HOME/.isabelle/etc/settings</code>).

For example, assume you want to use the application *foo* which is here supposed to be located at /usr/local/bin/. The following line has to be added to one of Isabelle's settings file:

FOO=/usr/local/bin/foo

In Isabelle, this application may now be executed by

```
system_out "$F00"
> ...
```

## A.7 Writing an Oracle (TBD)

**Problem:** You want to use a fast, new decision procedure not based one Isabelle's tactics, and you do not care whether it is sound.

**Solution:** Isabelle provides the oracle mechanisms to bypass the inference kernel. Note that theorems proven by an oracle carry a special mark to inform the user of their potential incorrectness.

#### Read More

A short introduction to oracles can be found in [isar-ref: no suitable label for section 3.11]. A simple example, which we will slightly extend here, is given in FOL/ex/Iff\_Oracle.thy. The raw interface for adding oracles is add\_oracle in Pure/thm.ML.

For our explanation here, we restrict ourselves to decide propositional formulae which consist only of equivalences between propositional variables, i.e. we want to decide whether (P = (Q = P)) = Q is a tautology.

Assume, that we have a decision procedure for such formulae, implemented in ML. Since we do not care how it works, we will use it here as an "external solver":

```
use "external_solver.ML"
```

We do, however, know that the solver provides a function *IffSolver.decide*. It takes a string representation of a formula and returns either *true* if the formula is a tautology or *false* otherwise. The input syntax is specified as follows:

```
formula ::= atom | ( formula <=> formula )
```

and all token are separated by at least one space.

(FIXME: is there a better way for describing the syntax?)

We will proceed in the following way. We start by translating a HOL formula into the string representation expected by the solver. The solver's result is then used to build an oracle, which we will subsequently use as a core for an Isar method to be able to apply the oracle in proving theorems.

Let us start with the translation function from Isabelle propositions into the solver's string representation. To increase efficiency while building the string, we use functions from the *Buffer* module.

```
fun translate t =
let.
  fun trans t =
    (case t of
      Buffer.add " (" #>
       trans t #>
       Buffer.add "<=>" #>
       trans u #>
       Buffer.add ") "
    | Free (n, @{typ bool}) =>
       Buffer.add " " #>
      Buffer.add n #>
      Buffer.add " "
    | _ => error "inacceptable term")
in Buffer.content (trans t Buffer.empty) end
```

Here is the string representation of the term p = (q = p):

```
translate @\{\text{term } "p = (q = p)"\}
> " ( p <=> ( q <=> p ) ) "
```

Let us check, what the solver returns when given a tautology:

```
IffSolver.decide (translate @\{\text{term "p = (q = p) = q"}\}\) > true

And here is what it returns for a formula which is not valid:

IffSolver.decide (translate @\{\text{term "p = (q = p)"}\}\) > false
```

Now, we combine these functions into an oracle. In general, an oracle may be given any input, but it has to return a certified proposition (a special term which is typechecked), out of which Isabelle's inference kernel "magically" makes a theorem.

Here, we take the proposition to be show as input. Note that we have to first extract the term which is then passed to the translation and decision procedure. If the solver finds this term to be valid, we return the given proposition unchanged to be turned then into a theorem:

```
oracle iff_oracle = {* fn ct =>
   if IffSolver.decide (translate (HOLogic.dest_Trueprop (Thm.term_of ct)))
   then ct
   else error "Proof failed."*}
```

Here is what we get when applying the oracle:

```
iff_oracle @{cprop "p = (p::bool)"}
> p = p
```

(FIXME: is there a better way to present the theorem?)

To be able to use our oracle for Isar proofs, we wrap it into a tactic:

```
val iff_oracle_tac =
   CSUBGOAL (fn (goal, i) =>
     (case try iff_oracle goal of
     NONE => no_tac
     | SOME thm => rtac thm i))
```

and create a new method solely based on this tactic:

```
method_setup iff_oracle = {*
    Scan.succeed (K (Method.SIMPLE_METHOD' iff_oracle_tac))
*} "Oracle-based decision procedure for chains of equivalences"
```

Finally, we can test our oracle to prove some theorems:

```
lemma "p = (p::bool)"
    by iff_oracle
lemma "p = (q = p) = q"
    by iff_oracle
```

(FIXME: say something about what the proof of the oracle is ... what do you mean?)

## A.8 SAT Solvers

**Problem:** You like to use a SAT solver to find out whether an Isabelle formula is satisfiable or not.

A.8. SAT SOLVERS 139

**Solution:** Isabelle contains a general interface for a number of external SAT solvers (including ZChaff and Minisat) and also contains a simple internal SAT solver that is based on the DPLL algorithm.

The SAT solvers expect a propositional formula as input and produce a result indicating that the formula is either satisfiable, unsatisfiable or unknown. The type of the propositional formula is <code>PropLogic.prop\_formula</code> with the usual constructors such as <code>And</code>, <code>Or</code> and so on.

The function  $PropLogic.prop\_formula\_of\_term$  translates an Isabelle term into a propositional formula. Let us illustrate this function by translating  $A \land \neg A \lor B$ . The function will return a propositional formula and a table. Suppose

```
val (pform, table) =  PropLogic.prop\_formula\_of\_term @\{term "A \land \neg A \lor B"\} Termtab.empty
```

then the resulting propositional formula pform is

```
Or (And (BoolVar 1, Not (BoolVar 1)), BoolVar 2)
```

where indices are assigned for the variables A and B, respectively. This assignment is recorded in the table that is given to the translation function and also returned (appropriately updated) in the result. In the case above the input table is empty (i.e. Termtab.empty) and the output table is

```
Termtab.dest table
> [(Free ("A", "bool"), 1), (Free ("B", "bool"), 2)]
```

An index is also produced whenever the translation function cannot find an appropriate propositional formula for a term. Attempting to translate  $\forall x$ . P x

```
val (pform', table') =
    PropLogic.prop_formula_of_term @{term "∀x::nat. P x"} Termtab.empty
```

returns BoolVar 1 for pform' and the table table' is:

```
map (apfst (Syntax.string_of_term @\{context\})) (Termtab.dest table') > (\forall x. P x, 1)
```

In the print out of the tabel, we used some pretty printing scaffolding to see better which assignment the table contains.

Having produced a propositional formula, you can now call the SAT solvers with the function SatSolver.invoke\_solver. For example

```
SatSolver.invoke_solver "dpll" pform
> SatSolver.SATISFIABLE assg
```

determines that the formula pform is satisfiable. If we inspect the returned function assg

```
let
  val SatSolver.SATISFIABLE assg = SatSolver.invoke_solver "dpl1" pform
in
  (assg 1, assg 2, assg 3)
end
> (SOME true, SOME true, NONE)
```

we obtain a possible assignment for the variables A and B that makes the formula satisfiable.

Note that we invoked the SAT solver with the string "dpll". This string specifies which SAT solver is invoked (in this case the internal one). If instead you invoke the SAT solver with the string "auto"

```
SatSolver.invoke_solver "auto" pform
```

several external SAT solvers will be tried (assuming they are installed). If no external SAT solver is installed, then the default is "dp11".

There are also two tactics that make use of SAT solvers. One is the tactic <code>sat\_tac</code>. For example

```
lemma "True"
apply(tactic {* sat.sat_tac @{context} 1 *})
done
```

However, for proving anything more exciting using sat\_tac you have to use a SAT solver that can produce a proof. The internal one is not usuable for this.

#### Read More

The interface for the external SAT solvers is implemented in HOL/Tools/sat\_solver.ML. This file contains also a simple SAT solver based on the DPLL algorithm. The tactics for SAT solvers are implemented in HOL/Tools/sat\_funcs.ML. Functions concerning propositional formulas are implemented in HOL/Tools/prop\_logic.ML. The tables used in the translation function are implemented in Pure/General/table.ML.

### A.9 User Space Type-Systems (TBD)

## Appendix B

## **Solutions to Most Exercises**

#### Solution for Exercise 2.6.1.

An alternative solution using the function mk\_binop is:

#### Solution for Exercise 2.6.2.

```
fun make_sum t1 t2 =
    HOLogic.mk_nat (HOLogic.dest_nat t1 + HOLogic.dest_nat t2)
```

#### Solution for Exercise 3.1.1.

```
val any = Scan.one (Symbol.not_eof)

val scan_cmt =
let
   val begin_cmt = Scan.this_string "(*"
   val end_cmt = Scan.this_string "*)"
in
   begin_cmt |-- Scan.repeat (Scan.unless end_cmt any) --| end_cmt
   >> (enclose "(**" "**)" o implode)
end
```

By using #> fst in the last line, the function scan\_all retruns a string, instead of the pair a parser would normally return. For example:

```
let
  val input1 = (explode "foo bar")
  val input2 = (explode "foo (*test*) bar (*test*)")
in
    (scan_all input1, scan_all input2)
end
> ("foo bar", "foo (**test**) bar (**test**)")
```

#### Solution for Exercise 4.5.1.

```
fun dest_sum term =
  case term of
    (\mathbb{Q}\{\text{term "(op +)}:: \text{nat } \Rightarrow \text{nat } \Rightarrow \text{nat"}\}  $ t1 $ t2) =>
         (snd (HOLogic.dest_number t1), snd (HOLogic.dest_number t2))
  | _ => raise TERM ("dest_sum", [term])
fun get_sum_thm ctxt t (n1, n2) =
let
  val sum = HOLogic.mk_number @{typ "nat"} (n1 + n2)
  val goal = Logic.mk_equals (t, sum)
  Goal.prove ctxt [] [] goal (K (Arith_Data.arith_tac ctxt 1))
fun \ add\_sp\_aux \ ss \ t =
let
  val ctxt = Simplifier.the_context ss
  val t' = term_of t
  SOME (get_sum_thm ctxt t' (dest_sum t'))
  handle TERM _ => NONE
```

The setup for the simproc is

```
simproc_setup add_sp ("t1 + t2") = {* K add_sp_aux *}

and a test case is the lemma
lemma "P (Suc (99 + 1)) ((0 + 0)::nat) (Suc (3 + 3 + 3)) (4 + 1)"
   apply(tactic {* simp_tac (HOL_basic_ss addsimprocs [@{simproc add_sp}]) 1 *})
```

```
where the simproc produces the goal state goal (1 subgoal):

1. P (Suc 100) 0 (Suc 9) ((4::'a) + (1::'a))
```

#### Solution for Exercise 4.6.1.

The following code assumes the function dest\_sum from the previous exercise.

```
fun add_simple_conv ctxt ctrm =
let.
  val trm = Thm.term_of ctrm
  get_sum_thm ctxt trm (dest_sum trm)
end
fun add_conv ctxt ctrm =
  (case Thm.term_of ctrm of
     Q\{term "(op +)::nat \Rightarrow nat"\} $ _ $ _ =>
         (Conv.binop_conv (add_conv ctxt)
          then_conv (Conv.try_conv (add_simple_conv ctxt))) ctrm
    / _ $ _ => Conv.combination_conv
                 (add_conv ctxt) (add_conv ctxt) ctrm
    | Abs _ => Conv.abs_conv (fn (_, ctxt) => add_conv ctxt) ctxt ctrm
    | _ => Conv.all_conv ctrm)
fun add_tac ctxt = CSUBGOAL (fn (goal, i) =>
  CONVERSION
    (Conv.params_conv ~1 (fn ctxt =>
       (Conv.prems_conv ~1 (add_conv ctxt) then_conv
          Conv.concl_conv ~1 (add_conv ctxt))) ctxt) i)
```

A test case for this conversion is as follows

```
lemma "P (Suc (99 + 1)) ((0 + 0)::nat) (Suc (3 + 3 + 3)) (4 + 1)"
   apply(tactic {* add_tac @{context} 1 *})?

where it produces the goal state
goal (1 subgoal):
   1. P (Suc 100) 0 (Suc 9) ((4::'a) + (1::'a))
```

#### Solution for Exercise 4.6.1.

We use the timing function <code>timing\_wrapper</code> from Recipe A.3. To measure any difference between the simproc and conversion, we will create mechanically terms involving additions and then set up a goal to be simplified. We have to be careful to set up the goal so that other parts of the simplifier do not interfere. For this we construct an unprovable goal which, after simplification, we are going to "prove" with the help of "sorry", that is the method <code>SkipProof.cheat\_tac</code>

For constructing test cases, we first define a function that returns a complete binary tree whose leaves are numbers and the nodes are additions.

This function generates for example:

```
writeln (Syntax.string_of_term @{context} (term_tree 2))
> (1 + 2) + (3 + 4)
```

The next function constructs a goal of the form P ... with a term produced by  $term\_tree$  filled in.

```
fun goal n = HOLogic.mk_Trueprop (@{term "P::nat⇒ bool"} $ (term_tree n))
```

Note that the goal needs to be wrapped in a *Trueprop*. Next we define two tactics,  $c\_tac$  and  $s\_tac$ , for the conversion and simproc, respectively. The idea is to first apply the conversion (respectively simproc) and then prove the remaining goal using  $cheat\_tac$ .

```
local
   fun mk_tac tac =
        timing_wrapper (EVERY1 [tac, K (SkipProof.cheat_tac @{theory})])
in
val c_tac = mk_tac (add_tac @{context})
val s_tac = mk_tac (simp_tac (HOL_basic_ss addsimprocs [@{simproc add_sp}]))
end
```

This is all we need to let the conversion run against the simproc:

```
val _ = Goal.prove @{context} [] [] (goal 8) (K c_tac)
val _ = Goal.prove @{context} [] [] (goal 8) (K s_tac)
```

If you do the exercise, you can see that both ways of simplifying additions perform relatively similar with perhaps some advantages for the simproc. That means the simplifier, even if much more complicated than conversions, is quite efficient for tasks it is designed for. It usually does not make sense to implement general-purpose rewriting using conversions. Conversions only have clear advantages in special situations: for example if you need to have control over innermost or outermost rewriting, or when rewriting rules are lead to non-termination.

## **Appendix C**

### **Comments for Authors**

• This tutorial can be compiled on the command-line with:

```
$ isabelle make
```

You very likely need a recent snapshot of Isabelle in order to compile the tutorial. Some parts of the tutorial also rely on compilation with PolyML.

• You can include references to other Isabelle manuals using the reference names from those manuals. To do this the following four 上下X commands are defined:

	Chapters	Sections
Implementation Manual		
Isar Reference Manual		

So \ichcite{ch:logic} yields a reference for the chapter about logic in the implementation manual, namely [Impl. Man., Ch. 2].

There are various document antiquotations defined for the tutorial. They allow
to check the written text against the current Isabelle code and also allow to
show responses of the ML-compiler. Therefore authors are strongly encouraged
to use antiquotations wherever appropriate.

The following antiquotations are defined:

@{ML "expr" for vars in structs} should be used for displaying any ML-expression, because the antiquotation checks whether the expression is valid ML-code. The for- and in-arguments are optional. The former is used for evaluating open expressions by giving a list of free variables. The latter is used to indicate in which structure or structures the ML-expression should be evaluated. Examples are:

```
\mathbb{Q}\{ML "1 + 3"\} 1 + 3 \mathbb{Q}\{ML "a + b" \text{ for a } b\} produce a + b \mathbb{Q}\{ML \text{ Ident in OuterLex}\}
```

• <code>@{ML\_response "expr" "pat"}</code> should be used to display ML-expressions and their response. The first expression is checked like in the antiquotation <code>@{ML "expr"}</code>; the second is a pattern that specifies the result the first expression produces. This pattern can contain "..." for parts that you like to omit. The response of the first expression will be checked against this pattern. Examples are:

```
@{ML_response "1+2" "3"}
@{ML_response "(1+2,3)" "(3,...)"}
```

which produce respectively

Note that this antiquotation can only be used when the result can be constructed: it does not work when the code produces an exception or returns an abstract datatype (like thm or cterm).

• @{ML\_response\_fake "expr" "pat"} works just like the antiquotation @{ML\_response "expr" "pat"} above, except that the result-specification is not checked. Use this antiquotation when the result cannot be constructed or the code generates an exception. Examples are:

which produce respectively

```
cterm_of @{theory} @{term "a + b = c"}
> a + b = c
($$ "x") (explode "world")
> Exception FAIL raised
```

This output mimics to some extend what the user sees when running the code.

• @{ML\_response\_fake\_both "expr" "pat"} can be used to show erroneous code. Neither the code nor the response will be checked. An example is:

• *Q{ML\_file "name"}* should be used when referring to a file. It checks whether the file exists. An example is

```
@{ML_file "Pure/General/basics.ML"}
```

The listed antiquotations honour options including [display] and [quotes]. For example

$$@{ML [quotes] "\"foo\" ^ \"bar\""} produces "foobar"$$

whereas

$$@{ML "\"foo\" ^ \"bar\""} produces only foobar$$

- Functions and value bindings cannot be defined inside antiquotations; they need to be included inside ML {\* ... \*} environments. In this way they are also checked by the compiler. Some Lack-hack in the tutorial, however, ensures that the environment markers are not printed.
- Line numbers can be printed using ML %linenos {\* ... \*} for ML-code or lemma %linenos ... for proofs. The tag is %linenosgray when the numbered text should be gray.

## **Bibliography**

- [1] R. Bornat. In Defence of Programming. Available online via http://www.cs.mdx. ac.uk/staffpages/r\_bornat/lectures/revisedinauguraltext.pdf, April 2005. Corrected and revised version of inaugural lecture, delivered on 22nd January 2004 at the School of Computing Science, Middlesex University.
- [2] M. Gordon, R. Milner, and C. P. Wadsworth. *Edinburgh LCF*, volume 78 of *Lecture Notes in Computer Science*. Springer, 1979.
- [3] T. Nipkow, L. C. Paulson, and M. Wenzel. *Isabelle/HOL: A Proof Assistant for Higher-Order Logic*. Springer, 2002. LNCS Tutorial 2283.
- [4] D. C. Oppen. Pretty printing. *ACM Transactions on Programming Languages and Systems*, pages 465–483, 1980.
- [5] L. C. Paulson. *ML for the Working Programmer*. Cambridge University Press, 2nd edition, 1996.

# Index

46 52	f + 01	
, 46, 53	asm_full_simp_tac, 81	
>, 27	asm_lr_simp_tac, 81	
1, 47	asm_simp_tac, 81	
>, 27	atac, 66, 70	
#->, 16	atomize_term (in ObjectLogic), 114	
##>, 16	axiomK (in Thm), 111	
##>>, 16	(' ) 06	
<b>#&gt;</b> , 14, 16	base_name (in Long_Name), 26	
#>>, 16	beta_conversion (in Thm), 92	
!!! (in OuterParse), 54	binder_types, 13	
!!, 48	blk (in Pretty), 40	
1, 47	breaks (in Pretty), 40	
/->, 15, 16		
11, 47	case_names (in RuleCases), 127	
//>, 15, 16	cat_lines, 11	
//>>, 15, 16	CHANGED, 78	
/>, 12, 16	cheat_tac (in SkipProof), 70	
/>>, 15, 16	chop, 120	
\$\$\$ (in OuterParse), 53	comb_conv (in Conv), 95	
<b>\$\$</b> , 45	combination_conv (in Conv), 95	
>>, 50	Command (in OuterLex), 52	
·, 14	commas (in Pretty), 41	
, - 1	commas, 10	
abs_conv (in Conv), 95	concl_conv (in Conv), 97	
add_consts_i (in Sign), 32	COND, 80	
add_del (in Attrib), 35	CONVERSION, 96	
add_tfrees (in Term), 27	crep_thm, 10	
add_thm (in Thm), 35, 36	CSUBGOAL, 75	
add_thms_dynamic (in PureThy), 38	cterm (type), 28	
add_tvars (in Term), 27	cterm_of, 28	
addcongs, 82		
addsimprocs, 82, 89	curry, 14	
addsimps, 82	cut_facts_tac, 71	
ahead (in Scan), 51	debug (in Toplevel), 10	
	declaration_attribute (in Thm), 35, 36	
all_conv (in Conv), 92		
all_tac, 67	define (in LocalTheory), 18, 111	
ALLGOALS, 77	definitionK (in Thm), 111	
arg_conv (in Conv), 94	del_thm (in Thm), 35	

INDEX 151

delcongs, 82	is_proper (in OuterLex), $52$	
delsimprocs, 82		
delsimps, 82	К, 12	
dest_number (in HOLogic), 90	Keyword (in OuterLex), 54	
dest_ss (in MetaSimplifier), 17	keyword (in OuterKeyword), 52	
dest_ss (in Simplifier), 83		
DETERM, 80	lambda, 23	
dresolve_tac, 71	lift (in Scan), 51	
dtac, 70	$list\_comb, 14, 23$	
dummyT, 29	local_theory (in OuterSyntax), 49, 59	
(dammy 1, 2)	106	
else_conv, 94	(1, -, -, -, -)	
empty_binding (in Attrib), 111	makestring (in PolyML), 9	
empty_ss, 82	map_index, 125	
enum (in Pretty), 41	map_types, 27	
enum (in OuterParse), 54	Mixfix, 58	
enum1 (in OuterParse), 54	mk_binop (in HOLogic), 141	
eq_thm_prop (in Thm), 35	$mk_eq$ (in $HOLogic$ ), 24	
eresolve_tac, 71	MRL, 72	
error (in Scan), 48	MRS, 72, 124	
error, 9	(1	
etac, 66, 70	no_conv (in Conv), 92	
EVERY', 76	no_tac, 67	
	NoSyn, 58	
EVERY1, 77	not_eof (in Symbol), 50	
explode (in Symbol), 46	note (in LocalTheory), 63, 126	
explode, 46	('	
extern_const (in Sign), 26	one (in Scan), 46, 49	
fastype_of, 13, 29	opt_thm_name (in SpecParse), 58	
fconv_rule (in Conv), 96	option (in Scan), 48	
finite (in Scan), 49	optional (in Scan), 47	
	ORELSE', 77	
FIRST', 77	OuterLex (structure), 52	
FIRST1, 77	(in a ) 0(	
fixes (in OuterParse), 57	params_conv (in Conv), 96	
FOCUS (in Subgoal), 73, 120	parse (in YXML), 56	
ftac, 70	prems_conv (in Conv), 97	
full_simp_tac, 81	Pretty.T (type), 39	
fun_conv (in Conv), 94	pretty_term (in Syntax), 42	
HOL_basic_ss, 83, 84	pretty_typ (in Syntax), 42	
	print_depth, 53	
HOL_ss, 84	print_tac, 70	
I, 12	profiling (in Toplevel), 10	
Ident (in OuterLex), 52	prop (in OuterParse), 56, 58	
import (in Variable), 11	prove (in Goal), 66, 117	
<del>-</del>	C - 10	
indent (in Pretty), 40	quote (in Pretty), 42	
inline (in ML_Antiquote), 18	mond nmon (in Compton) 60	
internalK (in Thm), 111	read_prop (in Syntax), 62	

152 INDEX

```
read_spec (in Specification), 107
                                        TRYALL, 80
read_term_pattern (in ProofContext),
                                        type_of, 29
        18
                                        unless (in Scan), 50
REPEAT, 78
repeat (in Scan), 49, 50
                                         variant_frees (in Variable), 14
REPEAT1, 79
repeat1 (in Scan), 49
                                        where_alt_specs (in SpecParse), 58
REPEAT_ALL_NEW, 79
                                        writeln, 8, 10
reserved (in OuterParse), 53
resolve_tac, 71
rewr_conv (in Conv), 93
rewrite_goals_tac, 84
RS, 33
RS, 72
RSN, 72
rtac, 66, 70, 71
rule_attribute (in Thm), 33, 34
rulify (in ObjectLogic), 31, 122
scan (in OuterSyntax), 52
setup (in Attrib), 33
simp_tac, 81
SIMPLE_METHOD, 63
stopper (in Symbol), 49
string_of (in Pretty), 39
string_of_term (in Syntax), 10
SUBGOAL, 75
SUBPROOF, 73, 120
subst_free, 23, 118
succeed (in Scan), 61
tap, 14
term (in OuterParse), 55
term (type), 19, 28
term_of, 10
THEN, 66
THEN', 67
then_conv, 93
theorem_i (in Proof), 62
theory_name (in Context), 16
theory_of (in ProofContext), 114
this_string (in Scan), 47
thm_name (in SpecParse), 58
thy_decl (in OuterKeyword), 62, 106
thy_goal (in OuterKeyword), 62
tracing, 9, 10
TRY, 78, 80
try_conv (in Conv), 94
```